

# NAG Library Routine Document

## F08VCF (DGGSSVD3)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08VCF (DGGSSVD3) computes the generalized singular value decomposition (GSVD) of an  $m$  by  $n$  real matrix  $A$  and a  $p$  by  $n$  real matrix  $B$ .

### 2 Specification

```
SUBROUTINE F08VCF (JOBV, JOBV, JOBQ, M, N, P, K, L, A, LDA, B, LDB,      &
                  ALPHA, BETA, U, LDU, V, LDV, Q, LDQ, WORK, LWORK,      &
                  IWORK, INFO)
INTEGER                M, N, P, K, L, LDA, LDB, LDU, LDV, LDQ, LWORK,      &
                      IWORK(N), INFO
REAL (KIND=nag_wp)    A(LDA,*), B(LDB,*), ALPHA(N), BETA(N), U(LDU,*),      &
                      V(LDV,*), Q(LDQ,*), WORK(max(1,LWORK))
CHARACTER(1)          JOBV, JOBV, JOBQ
```

The routine may be called by its LAPACK name ***dggssvd3***.

### 3 Description

Given an  $m$  by  $n$  real matrix  $A$  and a  $p$  by  $n$  real matrix  $B$ , the generalized singular value decomposition is given by

$$U^T A Q = D_1 \begin{pmatrix} 0 & R \end{pmatrix}, \quad V^T B Q = D_2 \begin{pmatrix} 0 & R \end{pmatrix},$$

where  $U$ ,  $V$  and  $Q$  are orthogonal matrices. Let  $l$  be the effective numerical rank of  $B$  and  $(k+l)$  be the effective numerical rank of the matrix  $\begin{pmatrix} A \\ B \end{pmatrix}$ , then the first  $k$  generalized singular values are infinite and the remaining  $l$  are finite.  $R$  is a  $(k+l)$  by  $(k+l)$  nonsingular upper triangular matrix,  $D_1$  and  $D_2$  are  $m$  by  $(k+l)$  and  $p$  by  $(k+l)$  ‘diagonal’ matrices structured as follows:

if  $m - k - l \geq 0$ ,

$$D_1 = \begin{matrix} & k & l \\ & \begin{pmatrix} I & 0 \\ 0 & C \end{pmatrix} \\ m-k-l & \begin{pmatrix} 0 & 0 \end{pmatrix} \end{matrix}$$

$$D_2 = \begin{matrix} & k & l \\ & \begin{pmatrix} 0 & S \\ 0 & 0 \end{pmatrix} \\ p-l & \begin{pmatrix} 0 & 0 \end{pmatrix} \end{matrix}$$

$$\begin{pmatrix} 0 & R \end{pmatrix} = \begin{matrix} & n-k-l & k & l \\ k & \begin{pmatrix} 0 & R_{11} & R_{12} \\ 0 & 0 & R_{22} \end{pmatrix} \\ l & \end{matrix}$$

where

$$C = \text{diag}(\alpha_{k+1}, \dots, \alpha_{k+l}),$$

$$S = \text{diag}(\beta_{k+1}, \dots, \beta_{k+l}),$$

and

$$C^2 + S^2 = I.$$

$R$  is stored as a submatrix of  $A$  with elements  $R_{ij}$  stored as  $A_{i,n-k-l+j}$  on exit.

If  $m - k - l < 0$ ,

$$D_1 = \begin{matrix} & k & m-k & k+l-m \\ m-k & \begin{pmatrix} I & 0 & 0 \\ 0 & C & 0 \end{pmatrix} \end{matrix}$$

$$D_2 = \begin{matrix} & k & m-k & k+l-m \\ m-k & \begin{pmatrix} 0 & S & 0 \\ 0 & 0 & I \\ p-l & 0 & 0 \end{pmatrix} \end{matrix}$$

$$(0 \quad R) = \begin{matrix} & n-k-l & k & m-k & k+l-m \\ k & \begin{pmatrix} 0 & R_{11} & R_{12} & R_{13} \\ 0 & 0 & R_{22} & R_{23} \\ m-k & 0 & 0 & R_{33} \end{pmatrix} \end{matrix}$$

where

$$C = \text{diag}(\alpha_{k+1}, \dots, \alpha_m),$$

$$S = \text{diag}(\beta_{k+1}, \dots, \beta_m),$$

and

$$C^2 + S^2 = I.$$

$\begin{pmatrix} R_{11} & R_{12} & R_{13} \\ 0 & R_{22} & R_{23} \end{pmatrix}$  is stored as a submatrix of  $A$  with  $R_{ij}$  stored as  $A_{i,n-k-l+j}$ , and  $R_{33}$  is stored as a submatrix of  $B$  with  $(R_{33})_{ij}$  stored as  $B_{m-k+i,n+m-k-l+j}$ .

The routine computes  $C$ ,  $S$ ,  $R$  and, optionally, the orthogonal transformation matrices  $U$ ,  $V$  and  $Q$ .

In particular, if  $B$  is an  $n$  by  $n$  nonsingular matrix, then the GSVD of  $A$  and  $B$  implicitly gives the SVD of  $AB^{-1}$ :

$$AB^{-1} = U(D_1 D_2^{-1})V^T.$$

If  $\begin{pmatrix} A \\ B \end{pmatrix}$  has orthonormal columns, then the GSVD of  $A$  and  $B$  is also equal to the CS decomposition of  $A$  and  $B$ . Furthermore, the GSVD can be used to derive the solution of the eigenvalue problem:

$$A^T A x = \lambda B^T B x.$$

In some literature, the GSVD of  $A$  and  $B$  is presented in the form

$$U^T A X = (0 \quad D_1), \quad V^T B X = (0 \quad D_2),$$

where  $U$  and  $V$  are orthogonal and  $X$  is nonsingular, and  $D_1$  and  $D_2$  are ‘diagonal’. The former GSVD form can be converted to the latter form by setting

$$X = Q \begin{pmatrix} I & 0 \\ 0 & R^{-1} \end{pmatrix}.$$

A two stage process is used to compute the GSVD of the matrix pair  $(A, B)$ . The pair is first reduced to upper triangular form by orthogonal transformations using F08VGF (DGGSP3). The GSVD of the resulting upper triangular matrix pair is then performed by F08YEF (DTGSJA) which uses a variant of the Kogbetliantz algorithm (a cyclic Jacobi method).

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (2012) *Matrix Computations* (4th Edition) Johns Hopkins University Press, Baltimore

## 5 Arguments

- 1:    JOBU – CHARACTER(1) *Input*  
*On entry:* if JOBU = 'U', the orthogonal matrix  $U$  is computed.  
If JOBU = 'N',  $U$  is not computed.  
*Constraint:* JOBU = 'U' or 'N'.
- 2:    JOBV – CHARACTER(1) *Input*  
*On entry:* if JOBV = 'V', the orthogonal matrix  $V$  is computed.  
If JOBV = 'N',  $V$  is not computed.  
*Constraint:* JOBV = 'V' or 'N'.
- 3:    JOBQ – CHARACTER(1) *Input*  
*On entry:* if JOBQ = 'Q', the orthogonal matrix  $Q$  is computed.  
If JOBQ = 'N',  $Q$  is not computed.  
*Constraint:* JOBQ = 'Q' or 'N'.
- 4:    M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
*Constraint:*  $M \geq 0$ .
- 5:    N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrices  $A$  and  $B$ .  
*Constraint:*  $N \geq 0$ .
- 6:    P – INTEGER *Input*  
*On entry:*  $p$ , the number of rows of the matrix  $B$ .  
*Constraint:*  $P \geq 0$ .
- 7:    K – INTEGER *Output*
- 8:    L – INTEGER *Output*  
*On exit:* K and L specify the dimension of the subblocks  $k$  and  $l$  as described in Section 3;  
 $(k + l)$  is the effective numerical rank of  $\begin{pmatrix} A \\ B \end{pmatrix}$ .
- 9:    A(LDA,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $m$  by  $n$  matrix  $A$ .  
*On exit:* contains the triangular matrix  $R$ , or part of  $R$ . See Section 3 for details.

- 10: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08VCF (DGGSD3) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .
- 11: B(LDB,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, N)$ .  
*On entry:* the  $p$  by  $n$  matrix  $B$ .  
*On exit:* contains the triangular matrix  $R$  if  $m - k - l < 0$ . See Section 3 for details.
- 12: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F08VCF (DGGSD3) is called.  
*Constraint:*  $LDB \geq \max(1, P)$ .
- 13: ALPHA(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* see the description of BETA.
- 14: BETA(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* ALPHA and BETA contain the generalized singular value pairs of  $A$  and  $B$ ,  $\alpha_i$  and  $\beta_i$ ;  
 $ALPHA(1 : K) = 1$ ,  
 $BETA(1 : K) = 0$ ,  
and if  $m - k - l \geq 0$ ,  
 $ALPHA(K + 1 : K + L) = C$ ,  
 $BETA(K + 1 : K + L) = S$ ,  
or if  $m - k - l < 0$ ,  
 $ALPHA(K + 1 : M) = C$ ,  
 $ALPHA(M + 1 : K + L) = 0$ ,  
 $BETA(K + 1 : M) = S$ ,  
 $BETA(M + 1 : K + L) = 1$ , and  
 $ALPHA(K + L + 1 : N) = 0$ ,  
 $BETA(K + L + 1 : N) = 0$ .  
The notation  $ALPHA(K : N)$  above refers to consecutive elements  $ALPHA(i)$ , for  $i = K, \dots, N$ .
- 15: U(LDU,\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array U must be at least  $\max(1, M)$  if  $JOB_U = 'U'$ , and at least 1 otherwise.  
*On exit:* if  $JOB_U = 'U'$ , U contains the  $m$  by  $m$  orthogonal matrix  $U$ .  
If  $JOB_U = 'N'$ , U is not referenced.
- 16: LDU – INTEGER *Input*  
*On entry:* the first dimension of the array U as declared in the (sub)program from which F08VCF (DGGSD3) is called.

*Constraints:*

if  $\text{JOBV} = 'U'$ ,  $\text{LDU} \geq \max(1, M)$ ;  
 otherwise  $\text{LDU} \geq 1$ .

- 17:  $V(\text{LDV}, *)$  – REAL (KIND=nag\_wp) array *Output*

**Note:** the second dimension of the array  $V$  must be at least  $\max(1, P)$  if  $\text{JOBV} = 'V'$ , and at least 1 otherwise.

*On exit:* if  $\text{JOBV} = 'V'$ ,  $V$  contains the  $p$  by  $p$  orthogonal matrix  $V$ .

If  $\text{JOBV} = 'N'$ ,  $V$  is not referenced.

- 18:  $\text{LDV}$  – INTEGER *Input*

*On entry:* the first dimension of the array  $V$  as declared in the (sub)program from which F08VCF (DGGSD3) is called.

*Constraints:*

if  $\text{JOBV} = 'V'$ ,  $\text{LDV} \geq \max(1, P)$ ;  
 otherwise  $\text{LDV} \geq 1$ .

- 19:  $Q(\text{LDQ}, *)$  – REAL (KIND=nag\_wp) array *Output*

**Note:** the second dimension of the array  $Q$  must be at least  $\max(1, N)$  if  $\text{JOBQ} = 'Q'$ , and at least 1 otherwise.

*On exit:* if  $\text{JOBQ} = 'Q'$ ,  $Q$  contains the  $n$  by  $n$  orthogonal matrix  $Q$ .

If  $\text{JOBQ} = 'N'$ ,  $Q$  is not referenced.

- 20:  $\text{LDQ}$  – INTEGER *Input*

*On entry:* the first dimension of the array  $Q$  as declared in the (sub)program from which F08VCF (DGGSD3) is called.

*Constraints:*

if  $\text{JOBQ} = 'Q'$ ,  $\text{LDQ} \geq \max(1, N)$ ;  
 otherwise  $\text{LDQ} \geq 1$ .

- 21:  $\text{WORK}(\max(1, \text{LWORK}))$  – REAL (KIND=nag\_wp) array *Workspace*

*On exit:* if  $\text{INFO} = 0$ ,  $\text{WORK}(1)$  contains the minimum value of  $\text{LWORK}$  required for optimal performance.

- 22:  $\text{LWORK}$  – INTEGER *Input*

*On entry:* the dimension of the array  $\text{WORK}$  as declared in the (sub)routine from which F08VCF (DGGSD3) is called.

If  $\text{LWORK} = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the  $\text{WORK}$  array, returns this value as the first entry of the  $\text{WORK}$  array, and no error message related to  $\text{LWORK}$  is issued.

*Suggested value:* for optimal performance,  $\text{LWORK}$  must generally be larger than the minimum; increase workspace by, say,  $nb \times (N + 1)$ , where  $nb$  is the optimal **block size**.

*Constraints:*

if  $\text{JOBV} = 'V'$ ,  $\text{LWORK} \geq \max(3 * N + 1, M, P)$ ;  
 if  $\text{JOBV} = 'N'$ ,  $\text{LWORK} \geq \max(3 * N + 1, M)$ .

23: IWORK(N) – INTEGER array Output

*On exit:* stores the sorting information. More precisely, if  $I$  is the ordered set of indices of ALPHA containing  $C$  (denote as  $\text{ALPHA}(I)$ , see BETA), then the corresponding elements  $\text{IWORK}(I)$  contain the swap pivots,  $J$ , that sorts  $I$  such that  $\text{ALPHA}(I)$  is in descending numerical order.

The following pseudocode sorts the set  $I$ :

```

for  $i \in I$ 
     $j = J_i$ 
    swap  $I_i$  and  $I_j$ 
end

```

24: INFO – INTEGER Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $\text{INFO} = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

The Jacobi-type procedure failed to converge.

## 7 Accuracy

The computed generalized singular value decomposition is nearly the exact generalized singular value decomposition for nearby matrices  $(A + E)$  and  $(B + F)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2 \text{ and } \|F\|_2 = O(\epsilon)\|B\|_2,$$

and  $\epsilon$  is the *machine precision*. See Section 4.12 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08VCF (DGGSVD3) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08VCF (DGGSVD3) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

This routine replaces the deprecated routine F08VAF (DGGSVD) which used an unblocked algorithm and therefore did not make best use of level 3 BLAS routines.

The complex analogue of this routine is F08VQF (ZGGSVD3).

## 10 Example

This example finds the generalized singular value decomposition

$$A = U\Sigma_1(0 \ R)Q^T, \quad B = V\Sigma_2(0 \ R)Q^T,$$

where

$$A = \begin{pmatrix} 1 & 2 & 3 \\ 3 & 2 & 1 \\ 4 & 5 & 6 \\ 7 & 8 & 8 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -2 & -3 & 3 \\ 4 & 6 & 5 \end{pmatrix},$$

together with estimates for the condition number of  $R$  and the error bound for the computed generalized singular values.

The example program assumes that  $m \geq n$ , and would need slight modification if this is not the case.

### 10.1 Program Text

Program f08vcfe

```
!      F08VCF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
      Use nag_library, Only: dggsvd3, dtrcon, nag_wp, x02ajf, x04cbf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: eps, rcond, serrbd
      Integer                     :: i, ifail, info, irank, j, k, l, lda, &
                                   ldb, ldq, ldu, ldv, lwork, m, n, p
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:,,:), alpha(:), b(:,,:), beta(:), &
                                   q(:,,:), u(:,,:), v(:,,:), work(:)
      Real (Kind=nag_wp)            :: wdum(1)
      Integer, Allocatable           :: iwork(:)
      Character (1)                 :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                     :: nint
!      .. Executable Statements ..
      Write (nout,*) 'F08VCF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, p
      lda = m
      ldb = p
      ldq = n
      ldu = m
      ldv = p
      Allocate (a(lda,n),alpha(n),b(ldb,n),beta(n),q(ldq,n),u(ldu,m),v(ldv,p), &
               iwork(n))

!      Perform workspace query to get optimal size of work
!      The NAG name equivalent of dggsvd3 is f08vcf
      lwork = -1
      Call dggsvd3('U','V','Q',m,n,p,k,l,a,lda,b,ldb,alpha,beta,u,ldu,v,ldv,q, &
                  ldq,wdum,lwork,iwork,info)
      lwork = nint(wdum(1))
      Allocate (work(lwork))

!      Read the m by n matrix A and p by n matrix B from data file
```

```

Read (nin,*)(a(i,1:n),i=1,m)
Read (nin,*)(b(i,1:n),i=1,p)

!      Compute the generalized singular value decomposition of (A, B)
!      (A = U*D1*(O R)*(Q**T), B = V*D2*(O R)*(Q**T), m>=n)
!      The NAG name equivalent of dggsvd3 is f08vcf
Call dggsvd3('U','V','Q',m,n,p,k,l,a,lda,b,ldb,alpha,beta,u,ldu,v,ldv,q, &
    ldq,work,lwork,iwork,info)

If (info==0) Then

!      Print solution

    irank = k + 1
    Write (nout,*) 'Number of infinite generalized singular values (K)'
    Write (nout,99999) k
    Write (nout,*) 'Number of finite generalized singular values (L)'
    Write (nout,99999) l
    Write (nout,*) 'Numerical rank of (A**T B**T)**T (K+L)'
    Write (nout,99999) irank
    Write (nout,*)
    Write (nout,*) 'Finite generalized singular values'
    Write (nout,99998)(alpha(j)/beta(j),j=k+1,irank)

    Write (nout,*)
    Flush (nout)

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
    ifail = 0
    Call x04cbf('General',' ',m,m,u,ldu,'1P,E12.4','Orthogonal matrix U', &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

    Write (nout,*)
    Flush (nout)

    Call x04cbf('General',' ',p,p,v,ldv,'1P,E12.4','Orthogonal matrix V', &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

    Write (nout,*)
    Flush (nout)

    Call x04cbf('General',' ',n,n,q,ldq,'1P,E12.4','Orthogonal matrix Q', &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

    Write (nout,*)
    Flush (nout)

    Call x04cbf('Upper triangular','Non-unit',irank,irank,a(1,n-irank+1), &
        lda,'1P,E12.4','Nonsingular upper triangular matrix R','Integer', &
        rlabs,'Integer',clabs,80,0,ifail)

!      Call DTRCON (F07TGF) to estimate the reciprocal condition
!      number of R

    Call dtrcon('Infinity-norm','Upper','Non-unit',irank,a(1,n-irank+1), &
        lda,rcond,work,iwork,info)

    Write (nout,*)
    Write (nout,*) 'Estimate of reciprocal condition number for R'
    Write (nout,99997) rcond
    Write (nout,*)

!      So long as irank = n, get the machine precision, eps, and
!      compute the approximate error bound for the computed
!      generalized singular values

    If (irank==n) Then
        eps = x02ajf()
        serrbd = eps/rcond
        Write (nout,*) 'Error estimate for the generalized singular values'

```

```

        Write (nout,99997) serrbd
      Else
        Write (nout,*) '(A**T B**T)**T is not of full rank'
      End If
    Else
      Write (nout,99996) 'Failure in DGGSD3. INFO =', info
    End If

99999 Format (1X,I5)
99998 Format (3X,8(1P,E12.4))
99997 Format (1X,1P,E11.1)
99996 Format (1X,A,I4)
      End Program f08vcfe

```

## 10.2 Program Data

F08VCF Example Program Data

```

  4      3      2      :Values of M, N and P

  1.0  2.0  3.0
  3.0  2.0  1.0
  4.0  5.0  6.0
  7.0  8.0  8.0 :End of matrix A

 -2.0 -3.0  3.0
  4.0  6.0  5.0 :End of matrix B

```

## 10.3 Program Results

F08VCF Example Program Results

Number of infinite generalized singular values (K)

1

Number of finite generalized singular values (L)

2

Numerical rank of (A\*\*T B\*\*T)\*\*T (K+L)

3

Finite generalized singular values

1.3151E+00 8.0185E-02

Orthogonal matrix U

	1	2	3	4
1	-1.3484E-01	5.2524E-01	-2.0924E-01	8.1373E-01
2	6.7420E-01	-5.2213E-01	-3.8886E-01	3.4874E-01
3	2.6968E-01	5.2757E-01	-6.5782E-01	-4.6499E-01
4	6.7420E-01	4.1615E-01	6.1014E-01	1.5127E-15

Orthogonal matrix V

	1	2
1	3.5539E-01	-9.3472E-01
2	9.3472E-01	3.5539E-01

Orthogonal matrix Q

	1	2	3
1	-8.3205E-01	-9.4633E-02	-5.4657E-01
2	5.5470E-01	-1.4195E-01	-8.1985E-01
3	0.0000E+00	-9.8534E-01	1.7060E-01

Nonsingular upper triangular matrix R

	1	2	3
1	-2.0569E+00	-9.0121E+00	-9.3705E+00
2		-1.0882E+01	-7.2688E+00
3			-6.0405E+00

Estimate of reciprocal condition number for R  
4.2E-02

Error estimate for the generalized singular values  
2.6E-15

---