
NAG Library Routine Document

F08XPF (ZGGESX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F08XPF (ZGGESX) computes the generalized eigenvalues, the generalized Schur form S; Tð Þ and,
optionally, the left and/or right generalized Schur vectors for a pair of n by n complex nonsymmetric
matrices A;Bð Þ.
Estimates of condition numbers for selected generalized eigenvalue clusters and Schur vectors are also
computed.

2 Specification

SUBROUTINE F08XPF (JOBVSL, JOBVSR, SORT, SELCTG, SENSE, N, A, LDA, B,
LDB, SDIM, ALPHA, BETA, VSL, LDVSL, VSR, LDVSR,
RCONDE, RCONDV, WORK, LWORK, RWORK, IWORK, LIWORK,
BWORK, INFO)

&
&
&

INTEGER N, LDA, LDB, SDIM, LDVSL, LDVSR, LWORK,
IWORK(max(1,LIWORK)), LIWORK, INFO

&

REAL (KIND=nag_wp) RCONDE(2), RCONDV(2), RWORK(max(1,8*N))
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), ALPHA(N), BETA(N),

VSL(LDVSL,*), VSR(LDVSR,*), WORK(max(1,LWORK))
&

LOGICAL SELCTG, BWORK(*)
CHARACTER(1) JOBVSL, JOBVSR, SORT, SENSE
EXTERNAL SELCTG

The routine may be called by its LAPACK name zggesx.

3 Description

The generalized Schur factorization for a pair of complex matrices A;Bð Þ is given by

A ¼ QSZH; B ¼ QTZH;

where Q and Z are unitary, T and S are upper triangular. The generalized eigenvalues, �, of A;Bð Þ are
computed from the diagonals of T and S and satisfy

Az ¼ �Bz;

where z is the corresponding generalized eigenvector. � is actually returned as the pair �; �ð Þ such that

� ¼ �=�

since �, or even both � and � can be zero. The columns of Q and Z are the left and right generalized
Schur vectors of A;Bð Þ.
Optionally, F08XPF (ZGGESX) can order the generalized eigenvalues on the diagonals of S; Tð Þ so that
selected eigenvalues are at the top left. The leading columns of Q and Z then form an orthonormal
basis for the corresponding eigenspaces, the deflating subspaces.

F08XPF (ZGGESX) computes T to have real non-negative diagonal entries. The generalized Schur
factorization, before reordering, is computed by the QZ algorithm.

The reciprocals of the condition estimates, the reciprocal values of the left and right projection norms,
are returned in RCONDEð1Þ and RCONDEð2Þ respectively, for the selected generalized eigenvalues,
together with reciprocal condition estimates for the corresponding left and right deflating subspaces, in
RCONDVð1Þ and RCONDVð2Þ. See Section 4.11 of Anderson et al. (1999) for further information.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users' Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University
Press, Baltimore

5 Arguments

1: JOBVSL – CHARACTER(1) Input

On entry: if JOBVSL ¼ N , do not compute the left Schur vectors.

If JOBVSL ¼ V , compute the left Schur vectors.

Constraint: JOBVSL ¼ N or V .

2: JOBVSR – CHARACTER(1) Input

On entry: if JOBVSR ¼ N , do not compute the right Schur vectors.

If JOBVSR ¼ V , compute the right Schur vectors.

Constraint: JOBVSR ¼ N or V .

3: SORT – CHARACTER(1) Input

On entry: specifies whether or not to order the eigenvalues on the diagonal of the generalized
Schur form.

SORT ¼ N
Eigenvalues are not ordered.

SORT ¼ S
Eigenvalues are ordered (see SELCTG).

Constraint: SORT ¼ N or S .

4: SELCTG – LOGICAL FUNCTION, supplied by the user. External Procedure

If SORT ¼ S , SELCTG is used to select generalized eigenvalues to be moved to the top left of
the generalized Schur form.

If SORT ¼ N , SELCTG is not referenced by F08XPF (ZGGESX), and may be called with the
dummy function F08XNZ.

The specification of SELCTG is:

FUNCTION SELCTG (A, B)
LOGICAL SELCTG

COMPLEX (KIND=nag_wp) A, B

1: A – COMPLEX (KIND=nag_wp) Input
2: B – COMPLEX (KIND=nag_wp) Input

On entry: an eigenvalue AðjÞ=BðjÞ is selected if SELCTG AðjÞ;BðjÞð Þ is .TRUE..

Note that in the ill-conditioned case, a selected generalized eigenvalue may no longer
satisfy SELCTG AðjÞ;BðjÞð Þ ¼ :TRUE: after ordering. INFO ¼ Nþ 2 in this case.

SELCTG must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which F08XPF (ZGGESX) is called. Arguments denoted as Input must not be
changed by this procedure.

5: SENSE – CHARACTER(1) Input

On entry: determines which reciprocal condition numbers are computed.

SENSE ¼ N
None are computed.

SENSE ¼ E
Computed for average of selected eigenvalues only.

SENSE ¼ V
Computed for selected deflating subspaces only.

SENSE ¼ B
Computed for both.

If SENSE ¼ E , V or B , SORT ¼ S .

Constraint: SENSE ¼ N , E , V or B .

6: N – INTEGER Input

On entry: n, the order of the matrices A and B.

Constraint: N � 0.

7: AðLDA; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the first of the pair of matrices, A.

On exit: A has been overwritten by its generalized Schur form S.

8: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08XPF
(ZGGESX) is called.

Constraint: LDA � max 1;Nð Þ.

9: BðLDB; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;Nð Þ.
On entry: the second of the pair of matrices, B.

On exit: B has been overwritten by its generalized Schur form T .

10: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F08XPF
(ZGGESX) is called.

Constraint: LDB � max 1;Nð Þ.

11: SDIM – INTEGER Output

On exit: if SORT ¼ N , SDIM ¼ 0.

If SORT ¼ S , SDIM ¼ number of eigenvalues (after sorting) for which SELCTG is .TRUE..

12: ALPHAðNÞ – COMPLEX (KIND=nag_wp) array Output

On exit: see the description of BETA.

13: BETAðNÞ – COMPLEX (KIND=nag_wp) array Output

On exit: ALPHAðjÞ=BETAðjÞ, for j ¼ 1; 2; . . . ;N, will be the generalized eigenvalues.
ALPHAðjÞ and BETAðjÞ; j ¼ 1; 2; . . . ;N are the diagonals of the complex Schur form S; Tð Þ.
BETAðjÞ will be non-negative real.

Note: the quotients ALPHAðjÞ=BETAðjÞ may easily overflow or underflow, and BETAðjÞ may
even be zero. Thus, you should avoid naively computing the ratio �=�. However, ALPHA will
always be less than and usually comparable with Ak k in magnitude, and BETA will always be
less than and usually comparable with Bk k.

14: VSLðLDVSL; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array VSL must be at least max 1;Nð Þ if JOBVSL ¼ V , and at
least 1 otherwise.

On exit: if JOBVSL ¼ V , VSL will contain the left Schur vectors, Q.

If JOBVSL ¼ N , VSL is not referenced.

15: LDVSL – INTEGER Input

On entry: the first dimension of the array VSL as declared in the (sub)program from which
F08XPF (ZGGESX) is called.

Constraints:

if JOBVSL ¼ V , LDVSL � max 1;Nð Þ;
otherwise LDVSL � 1.

16: VSRðLDVSR; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array VSR must be at least max 1;Nð Þ if JOBVSR ¼ V , and
at least 1 otherwise.

On exit: if JOBVSR ¼ V , VSR will contain the right Schur vectors, Z.

If JOBVSR ¼ N , VSR is not referenced.

17: LDVSR – INTEGER Input

On entry: the first dimension of the array VSR as declared in the (sub)program from which
F08XPF (ZGGESX) is called.

Constraints:

if JOBVSR ¼ V , LDVSR � max 1;Nð Þ;
otherwise LDVSR � 1.

18: RCONDEð2Þ – REAL (KIND=nag_wp) array Output

On exit: if SENSE ¼ E or B , RCONDEð1Þ and RCONDEð2Þ contain the reciprocal condition
numbers for the average of the selected eigenvalues.

If SENSE ¼ N or V , RCONDE is not referenced.

19: RCONDVð2Þ – REAL (KIND=nag_wp) array Output

On exit: if SENSE ¼ V or B , RCONDVð1Þ and RCONDVð2Þ contain the reciprocal condition
numbers for the selected deflating subspaces.

if SENSE ¼ N or E , RCONDV is not referenced.

20: WORKðmax 1;LWORKð ÞÞ – COMPLEX (KIND=nag_wp) array Workspace

On exit: if INFO ¼ 0, the real part of WORKð1Þ contains a bound on the value of LWORK
required for optimal performance.

21: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which
F08XPF (ZGGESX) is called.

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the bound on the
optimal size of the WORK array and the minimum size of the IWORK array, returns these values
as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or
LIWORK is issued.

Constraints:

if N ¼ 0, LWORK � 1;
if SENSE ¼ E , V or B , LWORK � max 1; 2� N; 2� SDIM� N� SDIMð Þð Þ;
otherwise LWORK � max 1; 2� Nð Þ.

Note: 2� SDIM� N� SDIMð Þ � N� N=2. Note also that an error is only returned if
LWORK < max 1; 2� Nð Þ, but if SENSE ¼ E , V or B this may not be large enough. Consider
increasing LWORK by nb, where nb is the optimal block size.

22: RWORKðmax 1; 8� Nð ÞÞ – REAL (KIND=nag_wp) array Workspace

Real workspace.

23: IWORKðmax 1;LIWORKð ÞÞ – INTEGER array Workspace

On exit: if INFO ¼ 0, IWORKð1Þ returns the minimum LIWORK.

24: LIWORK – INTEGER Input

On entry: the dimension of the array IWORK as declared in the (sub)program from which
F08XPF (ZGGESX) is called.

If LIWORK ¼ �1, a workspace query is assumed; the routine only calculates the bound on the
optimal size of the WORK array and the minimum size of the IWORK array, returns these values
as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or
LIWORK is issued.

Constraints:

if SENSE ¼ N or N ¼ 0, LIWORK � 1;
otherwise LIWORK � Nþ 2.

25: BWORKð�Þ – LOGICAL array Workspace

Note: the dimension of the array BWORK must be at least 1 if SORT ¼ N , and at least
max 1;Nð Þ otherwise.

If SORT ¼ N , BWORK is not referenced.

26: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

INFO ¼ 1 to N

The QZ iteration failed. A;Bð Þ are not in Schur form, but ALPHAðjÞ and BETAðjÞ should be
correct for j ¼ INFOþ 1; . . . ;N.

INFO ¼ Nþ 1

Unexpected error returned from F08XSF (ZHGEQZ).

INFO ¼ Nþ 2

After reordering, roundoff changed values of some complex eigenvalues so that leading
eigenvalues in the generalized Schur form no longer satisfy SELCTG ¼ :TRUE:. This could also
be caused by underflow due to scaling.

INFO ¼ Nþ 3

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the
problem is very ill-conditioned).

7 Accuracy

The computed generalized Schur factorization satisfies

Aþ E ¼ QSZT; Bþ F ¼ QTZT;

where

E;Fð Þk kF ¼ O �ð Þ A;Bð Þk kF
and � is the machine precision. See Section 4.11 of Anderson et al. (1999) for further details.

8 Parallelism and Performance

F08XPF (ZGGESX) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F08XPF (ZGGESX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n3.

The real analogue of this routine is F08XBF (DGGESX).

10 Example

This example finds the generalized Schur factorization of the matrix pair A;Bð Þ, where

A ¼
�21:10� 22:50i 53:50� 50:50i �34:50þ 127:50i 7:50þ 0:50i
�0:46� 7:78i �3:50� 37:50i �15:50þ 58:50i �10:50� 1:50i
4:30� 5:50i 39:70� 17:10i �68:50þ 12:50i �7:50� 3:50i
5:50þ 4:40i 14:40þ 43:30i �32:50� 46:00i �19:00� 32:50i

0
B@

1
CA

and

B ¼
1:00� 5:00i 1:60þ 1:20i �3:00þ 0:00i 0:00� 1:00i
0:80� 0:60i 3:00� 5:00i �4:00þ 3:00i �2:40� 3:20i
1:00þ 0:00i 2:40þ 1:80i �4:00� 5:00i 0:00� 3:00i
0:00þ 1:00i �1:80þ 2:40i 0:00� 4:00i 4:00� 5:00i

0
B@

1
CA;

such that the eigenvalues of A;Bð Þ for which �j j < 6 correspond to the top left diagonal elements of the

generalized Schur form, S; Tð Þ. Estimates of the condition numbers for the selected eigenvalue cluster
and corresponding deflating subspaces are also returned.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem,
but should be suitable for large problems.

10.1 Program Text

! F08XPF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module f08xpfe_mod

! F08XPF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: selctg

! .. Parameters ..
Integer, Parameter, Public :: nb = 64, nin = 5, nout = 6
Logical, Parameter, Public :: chkfac = .False., prcond = .False., &

prmat = .False.
Contains

Function selctg(a,b)

! Logical function selctg for use with ZGGESX (F08XPF)
! Returns the value .TRUE. if the absolute value of the eigenvalue
! a/b < 6.0

! .. Function Return Value ..
Logical :: selctg

! .. Scalar Arguments ..
Complex (Kind=nag_wp), Intent (In) :: a, b

! .. Intrinsic Procedures ..
Intrinsic :: abs

! .. Executable Statements ..
selctg = (abs(a)<6.0_nag_wp*abs(b))
Return

End Function selctg
End Module f08xpfe_mod
Program f08xpfe

! F08XPF Example Main Program

! .. Use Statements ..
Use nag_library, Only: f06bnf, m01daf, m01edf, nag_wp, x02ajf, x04dbf, &

zgemm, zggesx, zlange => f06uaf
Use f08xpfe_mod, Only: chkfac, nb, nin, nout, prcond, prmat, selctg

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Complex (Kind=nag_wp) :: alph, bet
Real (Kind=nag_wp) :: abnorm, anorm, bnorm, eps, normd, &

norme, tol
Integer :: i, ifail, info, lda, ldb, ldc, ldd, &

lde, ldvsl, ldvsr, liwork, lwork, n, &
sdim

Logical :: factor
! .. Local Arrays ..

Complex (Kind=nag_wp), Allocatable :: a(:,:), alpha(:), b(:,:), beta(:), &
c(:,:), d(:,:), e(:,:), vsl(:,:), &
vsr(:,:), work(:)

Complex (Kind=nag_wp) :: dummy(1)
Real (Kind=nag_wp) :: rconde(2), rcondv(2)
Real (Kind=nag_wp), Allocatable :: rwork(:)

Integer :: idum(1)
Integer, Allocatable :: iwork(:)
Logical, Allocatable :: bwork(:)
Character (1) :: clabs(1), rlabs(1)

! .. Intrinsic Procedures ..
Intrinsic :: abs, cmplx, max, nint, real

! .. Executable Statements ..
Write (nout,*) ’F08XPF Example Program Results’
Write (nout,*)
Flush (nout)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
ldb = n
ldc = n
ldd = n
lde = n
ldvsl = n
ldvsr = n
Allocate (a(lda,n),alpha(n),b(ldb,n),beta(n),c(ldc,n),d(ldd,n),e(lde,n), &

vsl(ldvsl,n),vsr(ldvsr,n),rwork(8*n),bwork(n))

! Use routine workspace query to get optimal workspace.
lwork = -1
liwork = -1

! The NAG name equivalent of zggesx is f08xpf
Call zggesx(’Vectors (left)’,’Vectors (right)’,’Sort’,selctg, &

’Both reciprocal condition numbers’,n,a,lda,b,ldb,sdim,alpha,beta,vsl, &
ldvsl,vsr,ldvsr,rconde,rcondv,dummy,lwork,rwork,idum,liwork,bwork, &
info)

! Make sure that there is enough workspace for block size nb.
lwork = max(n*nb+n*n/2,nint(real(dummy(1))))
liwork = max(n+2,idum(1))
Allocate (work(lwork),iwork(liwork))

! Read in the matrices A and B
Read (nin,*)(a(i,1:n),i=1,n)
Read (nin,*)(b(i,1:n),i=1,n)

If (chkfac) Then
! Copy A and B into D and E respectively

d(1:n,1:n) = a(1:n,1:n)
e(1:n,1:n) = b(1:n,1:n)

End If

! Find the Frobenius norms of A and B
! The NAG name equivalent of the LAPACK auxiliary zlange is f06uaf

anorm = zlange(’Frobenius’,n,n,a,lda,rwork)
bnorm = zlange(’Frobenius’,n,n,b,ldb,rwork)

If (prmat) Then
! Print matrices A and B
! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04dbf(’General’,’ ’,n,n,a,lda,’Bracketed’,’F8.4’,’Matrix A’, &

’Integer’,rlabs,’Integer’,clabs,80,0,ifail)
Write (nout,*)
Flush (nout)

ifail = 0
Call x04dbf(’General’,’ ’,n,n,b,ldb,’Bracketed’,’F8.4’,’Matrix B’, &

’Integer’,rlabs,’Integer’,clabs,80,0,ifail)
Write (nout,*)
Flush (nout)

End If

factor = .True.
! Find the generalized Schur form

! The NAG name equivalent of zggesx is f08xpf
Call zggesx(’Vectors (left)’,’Vectors (right)’,’Sort’,selctg, &

’Both reciprocal condition numbers’,n,a,lda,b,ldb,sdim,alpha,beta,vsl, &
ldvsl,vsr,ldvsr,rconde,rcondv,work,lwork,rwork,iwork,liwork,bwork, &
info)

If (info/=0 .And. info/=(n+2)) Then
Write (nout,99999) ’Failure in ZGGESX. INFO =’, info
factor = .False.

Else If (chkfac) Then
! Compute A - Q*S*Z^H from the factorization of (A,B) and store in
! matrix D
! The NAG name equivalent of zgemm is f06zaf

alph = cmplx(1,kind=nag_wp)
bet = cmplx(0,kind=nag_wp)
Call zgemm(’N’,’N’,n,n,n,alph,vsl,ldvsl,a,lda,bet,c,ldc)
alph = cmplx(-1,kind=nag_wp)
bet = cmplx(1,kind=nag_wp)
Call zgemm(’N’,’C’,n,n,n,alph,c,ldc,vsr,ldvsr,bet,d,ldd)

! Compute B - Q*T*Z^H from the factorization of (A,B) and store in
! matrix E

alph = cmplx(1,kind=nag_wp)
bet = cmplx(0,kind=nag_wp)
Call zgemm(’N’,’N’,n,n,n,alph,vsl,ldvsl,b,ldb,bet,c,ldc)
alph = cmplx(-1,kind=nag_wp)
bet = cmplx(1,kind=nag_wp)
Call zgemm(’N’,’C’,n,n,n,alph,c,ldc,vsr,ldvsr,bet,e,lde)

! Find norms of matrices D and E and warn if either is too large
normd = zlange(’O’,ldd,n,d,ldd,rwork)
If (normd>x02ajf()**0.75_nag_wp) Then

Write (nout,*) ’Norm of A-(Q*S*Z^T) is much greater than 0.’
factor = .False.
Write (nout,*) ’Schur factorization has failed.’

End If
norme = zlange(’O’,lde,n,e,lde,rwork)
If (norme>x02ajf()**0.75_nag_wp) Then

Write (nout,*) ’Norm of B-(Q*T*Z^T) is much greater than 0.’
factor = .False.

End If
End If

If (factor) Then
! Print eigenvalue details

Write (nout,99999) ’Number of eigenvalues for which SELCTG is true = ’ &
, sdim, ’(dimension of deflating subspaces)’

Write (nout,*)
! Print selected (finite) generalized eigenvalues

Write (nout,*) ’Selected generalized eigenvalues’

! Store absolute values of eigenvalues for ranking
work(1:n) = alpha(1:n)/beta(1:n)
rwork(1:n) = abs(work(1:n))

! Rank eigenvalues
ifail = 0
Call m01daf(rwork,1,sdim,’Descending’,iwork,ifail)

! Sort eigenvalues in work(1:n)
Call m01edf(work,1,sdim,iwork,ifail)
Do i = 1, sdim

Write (nout,99998) i, work(i)
End Do

If (info==(n+2)) Then
Write (nout,99997) ’*** Note that rounding errors mean ’, &

’that leading eigenvalues in the’, &
’generalized Schur form no longer satisfy SELCTG = .TRUE.’

Write (nout,*)

End If
Flush (nout)

If (prcond) Then
! Compute the machine precision and sqrt(anorm**2+bnorm**2)

eps = x02ajf()
abnorm = f06bnf(anorm,bnorm)
tol = eps*abnorm

! Print out the reciprocal condition numbers and error bound for
! selected eigenvalues

Write (nout,*)
Write (nout,99996) &

’Reciprocal condition numbers for the average of the’, &
’selected eigenvalues and their asymptotic error bound’, &
’rcond-left = ’, rconde(1), ’, rcond-right = ’, rconde(2), &
’, error = ’, tol/rconde(1)

Write (nout,*)
Write (nout,99996) &

’Reciprocal condition numbers for the deflating subspaces’, &
’and their approximate asymptotic error bound’, ’rcond-left = ’, &
rcondv(1), ’, rcond-right = ’, rcondv(2), ’, error = ’, &
tol/rcondv(2)

End If

Else
Write (nout,*) ’Schur factorization has failed.’

End If

99999 Format (1X,A,I4,/,1X,A)
99998 Format (1X,I2,1X,’(’,F6.2,’,’,F6.2,’)’)
99997 Format (1X,2A,/,1X,A)
99996 Format (1X,A,/,1X,A,/,1X,3(A,1P,E8.1))

End Program f08xpfe

10.2 Program Data

F08XPF Example Program Data
4 : Value of N
(-21.10,-22.50) ( 53.50,-50.50) (-34.50,127.50) ( 7.50, 0.50)
( -0.46, -7.78) ( -3.50,-37.50) (-15.50, 58.50) (-10.50, -1.50)
( 4.30, -5.50) ( 39.70,-17.10) (-68.50, 12.50) ( -7.50, -3.50)
( 5.50, 4.40) ( 14.40, 43.30) (-32.50,-46.00) (-19.00,-32.50) : End of A
( 1.00, -5.00) ( 1.60, 1.20) ( -3.00, 0.00) ( 0.00, -1.00)
( 0.80, -0.60) ( 3.00, -5.00) ( -4.00, 3.00) ( -2.40, -3.20)
( 1.00, 0.00) ( 2.40, 1.80) ( -4.00, -5.00) ( 0.00, -3.00)
( 0.00, 1.00) ( -1.80, 2.40) ( 0.00, -4.00) ( 4.00, -5.00) : End of B

10.3 Program Results

F08XPF Example Program Results

Number of eigenvalues for which SELCTG is true = 2
(dimension of deflating subspaces)

Selected generalized eigenvalues
1 ( 2.00, -5.00)
2 ( 3.00, -1.00)

