

# NAG Library Routine Document

## F08YVF (ZTGSYL)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08YVF (ZTGSYL) solves the generalized complex triangular Sylvester equations.

### 2 Specification

```
SUBROUTINE F08YVF (TRANS, IJOB, M, N, A, LDA, B, LDB, C, LDC, D, LDD, E,      &
                  LDE, F, LDF, SCALE, DIF, WORK, LWORK, IWORK, INFO)
INTEGER          IJOB, M, N, LDA, LDB, LDC, LDD, LDE, LDF, LWORK,      &
                  IWORK(M+N+2), INFO
REAL (KIND=nag_wp) SCALE, DIF
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), C(LDC,*), D(LDD,*),      &
                     E(LDE,*), F(LDF,*), WORK(max(1,LWORK))
CHARACTER(1)     TRANS
```

The routine may be called by its LAPACK name ***ztgsyl***.

### 3 Description

F08YVF (ZTGSYL) solves either the generalized complex Sylvester equations

$$\begin{aligned} AR - LB &= \alpha C \\ DR - LE &= \alpha F, \end{aligned} \quad (1)$$

or the equations

$$\begin{aligned} A^H R + D^H L &= \alpha C \\ RB^H + LE^H &= -\alpha F, \end{aligned} \quad (2)$$

where the pair  $(A, D)$  are given  $m$  by  $m$  matrices in generalized Schur form,  $(B, E)$  are given  $n$  by  $n$  matrices in generalized Schur form and  $(C, F)$  are given  $m$  by  $n$  matrices. The pair  $(R, L)$  are the  $m$  by  $n$  solution matrices, and  $\alpha$  is an output scaling factor determined by the routine to avoid overflow in computing  $(R, L)$ .

Equations (1) are equivalent to equations of the form

$$Zx = \alpha b,$$

where

$$Z = \begin{pmatrix} I \otimes A - B^H \otimes I \\ I \otimes D - E^H \otimes I \end{pmatrix}$$

and  $\otimes$  is the Kronecker product. Equations (2) are then equivalent to

$$Z^H y = \alpha b.$$

The pair  $(S, T)$  are in generalized Schur form if  $S$  and  $T$  are upper triangular as returned, for example, by F08XNF (ZGGES), or F08XSF (ZHGEQZ) with JOB = 'S'.

Optionally, the routine estimates  $\text{Dif}[(A, D), (B, E)]$ , the separation between the matrix pairs  $(A, D)$  and  $(B, E)$ , which is the smallest singular value of  $Z$ . The estimate can be based on either the Frobenius norm, or the 1-norm. The 1-norm estimate can be three to ten times more expensive than the Frobenius norm estimate, but makes the condition estimation uniform with the nonsymmetric eigenproblem. The Frobenius norm estimate provides a low cost, but equally reliable estimate. For more

information see Sections 2.4.8.3 and 4.11.1.3 of Anderson *et al.* (1999) and K  gstr  m and Poromaa (1996).

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

K  gstr  m B (1994) A perturbation analysis of the generalized Sylvester equation  $(AR - LB, DR - LE) = (c, F)$  *SIAM J. Matrix Anal. Appl.* **15** 1045–1060

K  gstr  m B and Poromaa P (1996) LAPACK-style algorithms and software for solving the generalized Sylvester equation and estimating the separation between regular matrix pairs *ACM Trans. Math. Software* **22** 78–103

## 5 Arguments

- 1: TRANS – CHARACTER(1) *Input*  
*On entry:* if TRANS = 'N', solve the generalized Sylvester equation (1).  
 If TRANS = 'C', solve the ‘conjugate transposed’ system (2).  
*Constraint:* TRANS = 'N' or 'C'.
- 2: IJOB – INTEGER *Input*  
*On entry:* specifies what kind of functionality is to be performed when TRANS = 'N'.  
 IJOB = 0  
     Solve (1) only.  
 IJOB = 1  
     The functionality of IJOB = 0 and 3.  
 IJOB = 2  
     The functionality of IJOB = 0 and 4.  
 IJOB = 3  
     Only an estimate of  $\text{Dif}[(A, D), (B, E)]$  is computed based on the Frobenius norm.  
 IJOB = 4  
     Only an estimate of  $\text{Dif}[(A, D), (B, E)]$  is computed based on the 1-norm.  
 If TRANS = 'C', IJOB is not referenced.  
*Constraint:* if TRANS = 'N',  $0 \leq \text{IJOB} \leq 4$ .
- 3: M – INTEGER *Input*  
*On entry:*  $m$ , the order of the matrices  $A$  and  $D$ , and the row dimension of the matrices  $C$ ,  $F$ ,  $R$  and  $L$ .  
*Constraint:*  $M \geq 0$ .
- 4: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrices  $B$  and  $E$ , and the column dimension of the matrices  $C$ ,  $F$ ,  $R$  and  $L$ .  
*Constraint:*  $N \geq 0$ .

- 5: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array A must be at least  $\max(1, M)$ .  
*On entry:* the upper triangular matrix  $A$ .
- 6: LDA – INTEGER Input  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .
- 7: B(LDB,\*) – COMPLEX (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array B must be at least  $\max(1, N)$ .  
*On entry:* the upper triangular matrix  $B$ .
- 8: LDB – INTEGER Input  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 9: C(LDC,\*) – COMPLEX (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array C must be at least  $\max(1, N)$ .  
*On entry:* contains the right-hand-side matrix  $C$ .  
*On exit:* if IJOB = 0, 1 or 2, C is overwritten by the solution matrix  $R$ .  
If TRANS = 'N' and IJOB = 3 or 4, C holds  $R$ , the solution achieved during the computation of the Dif estimate.
- 10: LDC – INTEGER Input  
*On entry:* the first dimension of the array C as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDC \geq \max(1, M)$ .
- 11: D(LDD,\*) – COMPLEX (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array D must be at least  $\max(1, M)$ .  
*On entry:* the upper triangular matrix  $D$ .
- 12: LDD – INTEGER Input  
*On entry:* the first dimension of the array D as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDD \geq \max(1, M)$ .
- 13: E(LDE,\*) – COMPLEX (KIND=nag\_wp) array Input  
**Note:** the second dimension of the array E must be at least  $\max(1, N)$ .  
*On entry:* the upper triangular matrix  $E$ .
- 14: LDE – INTEGER Input  
*On entry:* the first dimension of the array E as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDE \geq \max(1, N)$ .

- 15: F(LDF,\*) – COMPLEX (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array F must be at least  $\max(1, N)$ .  
*On entry:* contains the right-hand side matrix  $F$ .  
*On exit:* if IJOB = 0, 1 or 2, F is overwritten by the solution matrix  $L$ .  
 If TRANS = 'N' and IJOB = 3 or 4, F holds  $L$ , the solution achieved during the computation of the Dif estimate.
- 16: LDF – INTEGER Input  
*On entry:* the first dimension of the array F as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
*Constraint:*  $LDF \geq \max(1, M)$ .
- 17: SCALE – REAL (KIND=nag\_wp) Output  
*On exit:*  $\alpha$ , the scaling factor in (1) or (2).  
 If  $0 < SCALE < 1$ , C and F hold the solutions  $R$  and  $L$ , respectively, to a slightly perturbed system but the input arrays A, B, D and E have not been changed.  
 If SCALE = 0, C and F hold the solutions  $R$  and  $L$ , respectively, to the homogeneous system with  $C = F = 0$ . In this case DIF is not referenced.  
 Normally, SCALE = 1.
- 18: DIF – REAL (KIND=nag\_wp) Output  
*On exit:* the estimate of Dif. If IJOB = 0, DIF is not referenced.
- 19: WORK(max(1,LWORK)) – COMPLEX (KIND=nag\_wp) array Workspace  
*On exit:* if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 20: LWORK – INTEGER Input  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08YVF (ZTGSYL) is called.  
 If LWORK = -1, a workspace query is assumed; the routine only calculates the minimum size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.  
*Constraints:* if LWORK  $\neq -1$ ,  
     if TRANS = 'N' and IJOB = 1 or 2,  $LWORK \geq 2 \times M \times N$ ;  
     otherwise  $LWORK \geq 1$ .
- 21: IWORK(M + N + 2) – INTEGER array Workspace
- 22: INFO – INTEGER Output  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

$(A, D)$  and  $(B, E)$  have common or close eigenvalues and so no solution could be computed.

## 7 Accuracy

See Kågström (1994) for a perturbation analysis of the generalized Sylvester equation.

## 8 Parallelism and Performance

F08YVF (ZTGSYL) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations needed to solve the generalized Sylvester equations is approximately  $8mn(n + m)$ . The Frobenius norm estimate of Dif does not require additional significant computation, but the 1-norm estimate is typically five times more expensive.

The real analogue of this routine is F08YHF (DTGSYL).

## 10 Example

This example solves the generalized Sylvester equations

$$\begin{aligned} AR - LB &= \alpha C \\ DR - LE &= \alpha F, \end{aligned}$$

where

$$\begin{aligned} A &= \begin{pmatrix} 4.0 + 4.0i & 1.0 + 1.0i & 1.0 + 1.0i & 2.0 - 1.0i \\ 0 & 2.0 + 1.0i & 1.0 + 1.0i & 1.0 + 1.0i \\ 0 & 0 & 2.0 - 1.0i & 1.0 + 1.0i \\ 0 & 0 & 0 & 6.0 - 2.0i \end{pmatrix}, \\ B &= \begin{pmatrix} 2.0 & 1.0 + 1.0i & 1.0 + 1.0i & 3.0 - 1.0i \\ 0 & 1.0 & 2.0 + 1.0i & 1.0 + 1.0i \\ 0 & 0 & 1.0 & 1.0 + 1.0i \\ 0 & 0 & 0 & 2.0 \end{pmatrix}, \\ D &= \begin{pmatrix} 1.0 + 1.0i & 1.0 - 1.0i & 1.0 + 1.0i & 1.0 - 1.0i \\ 0 & 6.0 - 4.0i & 1.0 - 1.0i & 1.0 + 1.0i \\ 0 & 0 & 2.0 + 4.0i & 1.0 - 1.0i \\ 0 & 0 & 0 & 2.0 + 3.0i \end{pmatrix}, \\ E &= \begin{pmatrix} 1.0 & 1.0 + 1.0i & 1.0 - 1.0i & 1.0 + 1.0i \\ 0 & 2.0 & 1.0 + 1.0i & 1.0 - 1.0i \\ 0 & 0 & 2.0 & 1.0 + 1.0i \\ 0 & 0 & 0 & 1.0 \end{pmatrix}, \\ C &= \begin{pmatrix} -13.0 + 9.0i & 2.0 + 8.0i & -2.0 + 8.0i & -2.0 + 5.0i \\ -9.0 - 1.0i & 0.0 + 5.0i & -7.0 - 3.0i & -6.0 - 0.0i \\ -1.0 + 1.0i & 4.0 + 2.0i & 4.0 - 5.0i & 9.0 - 5.0i \\ -6.0 + 6.0i & 9.0 + 1.0i & -2.0 + 4.0i & 22.0 - 8.0i \end{pmatrix} \end{aligned}$$

and

$$F = \begin{pmatrix} -6.0 + 5.0i & 4.0 - 4.0i & -3.0 + 11.0i & 3.0 - 7.0i \\ -5.0 + 11.0i & 12.0 - 4.0i & -2.0 + 2.0i & 0.0 + 14.0i \\ -5.0 - 1.0i & 0.0 + 4.0i & -2.0 + 10.0i & 3.0 - 1.0i \\ -6.0 - 2.0i & 1.0 + 1.0i & -7.0 - 3.0i & 4.0 + 7.0i \end{pmatrix}.$$

## 10.1 Program Text

Program f08yvfe

```
!      F08YVF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, ztg syl
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: dif, scale
      Integer                     :: i, ifail, ijob, info, lda, ldb, ldc, &
                                   ldd, lde, ldf, lwork, m, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), c(:,:), d(:,:),      &
                                   e(:,:), f(:,:), work(:)
      Integer, Allocatable          :: iwork(:)
      Character (1)                 :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F08YVF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      ldb = n
      ldc = m
      ldd = m
      lde = n
      ldf = m
      lwork = 1
      Allocate (a(lda,m),b(ldb,n),c(ldc,n),d(ldd,m),e(lde,n),f(ldf,n),      &
               work(lwork),iwork(m+n+2))
!
!      Read A, B, D, E, C and F from data file
!
      Read (nin,*)(a(i,1:m),i=1,m)
      Read (nin,*)(b(i,1:n),i=1,n)
      Read (nin,*)(d(i,1:m),i=1,m)
      Read (nin,*)(e(i,1:n),i=1,n)
      Read (nin,*)(c(i,1:n),i=1,m)
      Read (nin,*)(f(i,1:n),i=1,m)
!
!      Solve the Sylvester equations
!      A*R - L*B = scale*C and D*R - L*E = scale*F
!      for R and L.
!
      ijob = 0
!
!      The NAG name equivalent of ztg syl is f08yvfe
      Call ztg syl('No transpose',ijob,m,n,a,lda,b,ldb,c,ldc,d,ldd,e,lde,f,ldf, &
                  scale,dif,work,lwork,iwork,info)
!
      If (info>=1) Then
        Write (nout,99999)
        Write (nout,*)
        Flush (nout)
```

```

      End If

!      Print the solution matrices R and L

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04dbf('General',' ',m,n,c,ldc,'Bracketed','F7.4',           &
        'Solution matrix R','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Write (nout,*)
      Flush (nout)

      ifail = 0
      Call x04dbf('General',' ',m,n,f,ldf,'Bracketed','F7.4',           &
        'Solution matrix L','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Write (nout,*)
      Write (nout,99998) 'SCALE = ', scale

99999 Format (/, ' (A,D) and (B,E) have common or very close eigenval','ues.', &
  /, ' Perturbed values were used to solve the equations')
99998 Format (1X,A,1P,E10.2)
      End Program f08yvfe

```

## 10.2 Program Data

F08YVF Example Program Data

```

4 4                                     :Values of M and N
( 4.0, 4.0) ( 1.0, 1.0) ( 1.0, 1.0) ( 2.0, -1.0)
( 0.0, 0.0) ( 2.0, 1.0) ( 1.0, 1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, -1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 6.0, -2.0)      :End of matrix A
( 2.0, 0.0) ( 1.0, 1.0) ( 1.0, 1.0) ( 3.0, -1.0)
( 0.0, 0.0) ( 1.0, 0.0) ( 2.0, 1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, 0.0)      :End of matrix B
( 1.0, 1.0) ( 1.0, -1.0) ( 1.0, 1.0) ( 1.0, -1.0)
( 0.0, 0.0) ( 6.0, -4.0) ( 1.0, -1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, 4.0) ( 1.0, -1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, 3.0)      :End of matrix D
( 1.0, 0.0) ( 1.0, 1.0) ( 1.0, -1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 2.0, 0.0) ( 1.0, 1.0) ( 1.0, -1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, 0.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0)      :End of matrix E
(-13.0, 9.0) ( 2.0, 8.0) (-2.0, 8.0) (-2.0, 5.0)
(-9.0, -1.0) ( 0.0, 5.0) (-7.0, -3.0) (-6.0, 0.0)
(-1.0, 1.0) ( 4.0, 2.0) ( 4.0, -5.0) ( 9.0, -5.0)
(-6.0, 6.0) ( 9.0, 1.0) (-2.0, 4.0) (22.0, -8.0)      :End of matrix C
(-6.0, 5.0) ( 4.0, -4.0) (-3.0, 11.0) ( 3.0, -7.0)
(-5.0, 11.0) (12.0, -4.0) (-2.0, 2.0) ( 0.0, 14.0)
(-5.0, -1.0) ( 0.0, 4.0) (-2.0, 10.0) ( 3.0, -1.0)
(-6.0, -2.0) ( 1.0, 1.0) (-7.0, -3.0) ( 4.0, 7.0)      :End of matrix F

```

## 10.3 Program Results

F08YVF Example Program Results

Solution matrix R

```

1 2 3 4
1 ( 1.0000, 1.0000) ( 1.0000, 1.0000) ( 1.0000, 1.0000) ( 1.0000, 1.0000)
2 (-1.0000, 1.0000) ( 2.0000, 1.0000) (-1.0000, 1.0000) (-1.0000, 1.0000)
3 (-1.0000, 1.0000) ( 1.0000, 1.0000) ( 3.0000, 1.0000) ( 1.0000, 1.0000)
4 (-1.0000, 1.0000) ( 1.0000, 1.0000) (-1.0000, 1.0000) ( 4.0000, 1.0000)

```

Solution matrix L

```

1 2 3 4
1 ( 4.0000, 1.0000) (-1.0000, 1.0000) ( 1.0000, 1.0000) (-1.0000, 1.0000)

```

```
2  ( 1.0000, 1.0000) ( 3.0000, 1.0000) (-1.0000, 1.0000) ( 1.0000, 1.0000)
3  (-1.0000, 1.0000) ( 1.0000, 1.0000) ( 2.0000, 1.0000) (-1.0000, 1.0000)
4  ( 1.0000, 1.0000) (-1.0000, 1.0000) ( 1.0000, 1.0000) ( 1.0000, 1.0000)
```

```
SCALE = 1.00E+00
```

---