

NAG Library Routine Document

F11XAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F11XAF computes a matrix-vector or transposed matrix-vector product involving a real sparse nonsymmetric matrix stored in coordinate storage format.

2 Specification

```
SUBROUTINE F11XAF (TRANS, N, NNZ, A, IROW, ICOL, CHECK, X, Y, IFAIL)
  INTEGER          N, NNZ, IROW(NNZ), ICOL(NNZ), IFAIL
  REAL (KIND=nag_wp) A(NNZ), X(N), Y(N)
  CHARACTER(1)     TRANS, CHECK
```

3 Description

F11XAF computes either the matrix-vector product $y = Ax$, or the transposed matrix-vector product $y = A^T x$, according to the value of the argument TRANS, where A is an n by n sparse nonsymmetric matrix, of arbitrary sparsity pattern. The matrix A is stored in coordinate storage (CS) format (see Section 2.1.1 in the F11 Chapter Introduction). The array A stores all nonzero elements of A , while arrays IROW and ICOL store the corresponding row and column indices respectively.

It is envisaged that a common use of F11XAF will be to compute the matrix-vector product required in the application of F11BEF to sparse linear systems. An illustration of this usage appears in Section 10 in F11DDF.

4 References

None.

5 Arguments

- | | | |
|----|---|--------------|
| 1: | TRANS – CHARACTER(1) | <i>Input</i> |
| | <i>On entry:</i> specifies whether or not the matrix A is transposed. | |
| | TRANS = 'N'
$y = Ax$ is computed. | |
| | TRANS = 'T'
$y = A^T x$ is computed. | |
| | <i>Constraint:</i> TRANS = 'N' or 'T'. | |
| 2: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the order of the matrix A . | |
| | <i>Constraint:</i> $N \geq 1$. | |
| 3: | NNZ – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of nonzero elements in the matrix A . | |
| | <i>Constraint:</i> $1 \leq \text{NNZ} \leq N^2$. | |

- 4: $A(NNZ)$ – REAL (KIND=nag_wp) array *Input*
On entry: the nonzero elements in the matrix A , ordered by increasing row index, and by increasing column index within each row. Multiple entries for the same row and column indices are not permitted. The routine F11ZAF may be used to order the elements in this way.
- 5: $IROW(NNZ)$ – INTEGER array *Input*
 6: $ICOL(NNZ)$ – INTEGER array *Input*
On entry: the row and column indices of the nonzero elements supplied in array A .
Constraints:
 $IROW$ and $ICOL$ must satisfy the following constraints (which may be imposed by a call to F11ZAF):
- $$1 \leq IROW(i) \leq N \text{ and } 1 \leq ICOL(i) \leq N, \text{ for } i = 1, 2, \dots, NNZ;$$
- $$IROW(i-1) < IROW(i) \text{ or } IROW(i-1) = IROW(i) \text{ and } ICOL(i-1) < ICOL(i), \text{ for } i = 2, 3, \dots, NNZ.$$
- 7: $CHECK$ – CHARACTER(1) *Input*
On entry: specifies whether or not the CS representation of the matrix A , values of N , NNZ , $IROW$ and $ICOL$ should be checked.
 $CHECK = 'C'$
 Checks are carried on the values of N , NNZ , $IROW$ and $ICOL$.
 $CHECK = 'N'$
 None of these checks are carried out.
 See also Section 9.2.
Constraint: $CHECK = 'C'$ or $'N'$.
- 8: $X(N)$ – REAL (KIND=nag_wp) array *Input*
On entry: the vector x .
- 9: $Y(N)$ – REAL (KIND=nag_wp) array *Output*
On exit: the vector y .
- 10: $IFAIL$ – INTEGER *Input/Output*
On entry: $IFAIL$ must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of $IFAIL$ on exit.**
On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $TRANS \neq 'N'$ or $'T'$,
or $CHECK \neq 'C'$ or $'N'$.

$IFAIL = 2$

On entry, $N < 1$,
or $NNZ < 1$,
or $NNZ > N^2$.

$IFAIL = 3$

On entry, the arrays IROW and ICOL fail to satisfy the following constraints:

$1 \leq IROW(i) \leq N$ and $1 \leq ICOL(i) \leq N$, for $i = 1, 2, \dots, NNZ$;

$IROW(i-1) < IROW(i)$, or $IROW(i-1) = IROW(i)$ and $ICOL(i-1) < ICOL(i)$, for $i = 2, 3, \dots, NNZ$.

Therefore a nonzero element has been supplied which does not lie within the matrix A , is out of order, or has duplicate row and column indices. Call F11ZAF to reorder and sum or remove duplicates.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed vector y satisfies the error bound:

$\|y - Ax\|_\infty \leq c(n)\epsilon\|A\|_\infty\|x\|_\infty$, if $TRANS = 'N'$, or

$\|y - A^T x\|_\infty \leq c(n)\epsilon\|A^T\|_\infty\|x\|_\infty$, if $TRANS = 'T'$,

where $c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

8 Parallelism and Performance

F11XAF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F11XAF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

9.1 Timing

The time taken for a call to F11XAF is proportional to NNZ.

9.2 Use of CHECK

It is expected that a common use of F11XAF will be to compute the matrix-vector product required in the application of F11BEF to sparse linear systems. In this situation F11XAF is likely to be called many times with the same matrix A . In the interests of both reliability and efficiency you are recommended to set CHECK = 'C' for the first of such calls, and to set CHECK = 'N' for all subsequent calls.

10 Example

This example reads in a sparse matrix A and a vector x . It then calls F11XAF to compute the matrix-vector product $y = Ax$ and the transposed matrix-vector product $y = A^T x$.

10.1 Program Text

```

Program f11xafe

!      F11XAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f11xaf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, n, nnz
      Character (1)               :: check, trans
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:), x(:), y(:)
      Integer, Allocatable        :: icol(:), irow(:)
!      .. Executable Statements ..
      Write (nout,*) 'F11XAF Example Program Results'
!      Skip heading in data file
      Read (nin,*)

!      Read order of matrix and number of nonzero entries

      Read (nin,*) n
      Read (nin,*) nnz

      Allocate (a(nnz),x(n),y(n),icol(nnz),irow(nnz))

!      Read the matrix A

      Do i = 1, nnz
         Read (nin,*) a(i), irow(i), icol(i)
      End Do

!      Read the vector x

      Read (nin,*) x(1:n)

!      Calculate matrix-vector product

```

```

      trans = 'N'
      check = 'C'

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call f11xaf(trans,n,nnz,a,irow,icol,check,x,y,ifail)

!      Output results

      Write (nout,*)
      Write (nout,*) ' Matrix-vector product'
      Write (nout,'(E16.4)') y(1:n)

!      Calculate transposed matrix-vector product

      trans = 'T'
      check = 'N'

      ifail = 0
      Call f11xaf(trans,n,nnz,a,irow,icol,check,x,y,ifail)

!      Output results

      Write (nout,*)
      Write (nout,*) ' Transposed matrix-vector product'
      Write (nout,'(E16.4)') y(1:n)

      End Program f11xaf

```

10.2 Program Data

F11XAF Example Program Data

```

5          N
11         NNZ
2.   1     1
1.   1     2
1.   2     3
-1.  2     4
4.   3     1
1.   3     3
1.   3     5
1.   4     4
2.   4     5
-2.  5     2
3.   5     5      A(I), IROW(I), ICOL(I), I=1,...,NNZ
0.70 0.16 0.52
0.77 0.28      X(I), I=1,...,N

```

10.3 Program Results

F11XAF Example Program Results

```

Matrix-vector product
0.1560E+01
-0.2500E+00
0.3600E+01
0.1330E+01
0.5200E+00

Transposed matrix-vector product
0.3480E+01
0.1400E+00
0.6800E+00
0.6100E+00
0.2900E+01

```
