
F12APF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

Note: this routine uses optional parameters to define choices in the problem specification. If you wish
to use default settings for all of the optional parameters, then the option setting routine F12ARF need
not be called. If, however, you wish to reset some or all of the settings please refer to Section 11 in
F12ARF for a detailed description of the specification of the optional parameters.

1 Purpose

F12APF is an iterative solver in a suite of routines consisting of F12ANF, F12APF, F12AQF, F12ARF
and F12ASF. It is used to find some of the eigenvalues (and optionally the corresponding eigenvectors)
of a standard or generalized eigenvalue problem defined by complex nonsymmetric matrices.

2 Specification

SUBROUTINE F12APF (IREVCM, RESID, V, LDV, X, MX, NSHIFT, COMM, ICOMM,
IFAIL)

&

INTEGER IREVCM, LDV, NSHIFT, ICOMM(*), IFAIL
COMPLEX (KIND=nag_wp) RESID(*), V(LDV,*), X(*), MX(*), COMM(*)

3 Description

The suite of routines is designed to calculate some of the eigenvalues, �, (and optionally the
corresponding eigenvectors, x) of a standard eigenvalue problem Ax ¼ �x, or of a generalized
eigenvalue problem Ax ¼ �Bx of order n, where n is large and the coefficient matrices A and B are
sparse, complex and nonsymmetric. The suite can also be used to find selected eigenvalues/eigenvectors
of smaller scale dense, complex and nonsymmetric problems.

F12APF is a reverse communication routine, based on the ARPACK routine znaupd, using the
Implicitly Restarted Arnoldi iteration method. The method is described in Lehoucq and Sorensen (1996)
and Lehoucq (2001) while its use within the ARPACK software is described in great detail in Lehoucq
et al. (1998). An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices is
provided in Lehoucq and Scott (1996). This suite of routines offers the same functionality as the
ARPACK software for complex nonsymmetric problems, but the interface design is quite different in
order to make the option setting clearer and to simplify the interface of F12APF.

The setup routine F12ANF must be called before F12APF, the reverse communication iterative solver.
Options may be set for F12APF by prior calls to the option setting routine F12ARF and a post-
processing routine F12AQF must be called following a successful final exit from F12APF. F12ASF may
be called following certain flagged intermediate exits from F12APF to provide additional monitoring
information about the computation.

F12APF uses reverse communication, i.e., it returns repeatedly to the calling program with the
argument IREVCM (see Section 5) set to specified values which require the calling program to carry
out one of the following tasks:

– compute the matrix-vector product y ¼ OPx, where OP is defined by the computational mode;

– compute the matrix-vector product y ¼ Bx;

– notify the completion of the computation;

– allow the calling program to monitor the solution.

The problem type to be solved (standard or generalized), the spectrum of eigenvalues of interest, the
mode used (regular, regular inverse, shifted inverse, shifted real or shifted imaginary) and other options

can all be set using the option setting routine F12ARF (see Section 11.1 in F12ARF for details on
setting options and of the default settings).

4 References

Lehoucq R B (2001) Implicitly restarted Arnoldi methods and subspace iteration SIAM Journal on
Matrix Analysis and Applications 23 551–562

Lehoucq R B and Scott J A (1996) An evaluation of software for computing eigenvalues of sparse
nonsymmetric matrices Preprint MCS-P547-1195 Argonne National Laboratory

Lehoucq R B and Sorensen D C (1996) Deflation techniques for an implicitly restarted Arnoldi iteration
SIAM Journal on Matrix Analysis and Applications 17 789–821

Lehoucq R B, Sorensen D C and Yang C (1998) ARPACK Users' Guide: Solution of Large-scale
Eigenvalue Problems with Implicitly Restarted Arnoldi Methods SIAM, Philidelphia

5 Arguments

Note: this routine uses reverse communication. Its use involves an initial entry, intermediate exits and
re-entries, and a final exit, as indicated by the argument IREVCM. Between intermediate exits and re-
entries, all arguments other than X, MX and COMM must remain unchanged.

1: IREVCM – INTEGER Input/Output

On initial entry: IREVCM ¼ 0, otherwise an error condition will be raised.

On intermediate re-entry: must be unchanged from its previous exit value. Changing IREVCM to
any other value between calls will result in an error.

On intermediate exit: has the following meanings.

IREVCM ¼ �1
The calling program must compute the matrix-vector product y ¼ OPx, where x is stored
in X (by default) or in the array COMM (starting from the location given by the first
element of ICOMM) when the option Pointers ¼ YES is set in a prior call to F12ARF.
The result y is returned in X (by default) or in the array COMM (starting from the location
given by the second element of ICOMM) when the option Pointers ¼ YES is set.

IREVCM ¼ 1
The calling program must compute the matrix-vector product y ¼ OPx. This is similar to
the case IREVCM ¼ �1 except that the result of the matrix-vector product Bx (as required
in some computational modes) has already been computed and is available in MX (by
default) or in the array COMM (starting from the location given by the third element of
ICOMM) when the option Pointers ¼ YES is set.

IREVCM ¼ 2
The calling program must compute the matrix-vector product y ¼ Bx, where x is stored in
X and y is returned in MX (by default) or in the array COMM (starting from the location
given by the second element of ICOMM) when the option Pointers ¼ YES is set.

IREVCM ¼ 3
Compute the NSHIFT complex shifts. This value of IREVCM will only arise if the
optional parameter Supplied Shifts is set in a prior call to F12ARF which is intended for
experienced users only; the default and recommended option is to use exact shifts (see
Lehoucq et al. (1998) for details).

IREVCM ¼ 4
Monitoring step: a call to F12ASF can now be made to return the number of Arnoldi
iterations, the number of converged Ritz values, the array of converged values, and the
corresponding Ritz estimates.

On final exit: IREVCM ¼ 5: F12APF has completed its tasks. The value of IFAIL determines
whether the iteration has been successfully completed, or whether errors have been detected. On

successful completion F12AQF must be called to return the requested eigenvalues and
eigenvectors (and/or Schur vectors).

Constraint: on initial entry, IREVCM ¼ 0; on re-entry IREVCM must remain unchanged.

2: RESIDð�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the dimension of the array RESID must be at least N (see F12ANF).

On initial entry: need not be set unless the option Initial Residual has been set in a prior call to
F12ARF in which case RESID should contain an initial residual vector, possibly from a previous
run.

On intermediate re-entry: must be unchanged from its previous exit. Changing RESID to any
other value between calls may result in an error exit.

On intermediate exit: contains the current residual vector.

On final exit: contains the final residual vector.

3: VðLDV; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array V must be at least max 1;NCVð Þ (see F12ANF).

On initial entry: need not be set.

On intermediate re-entry: must be unchanged from its previous exit.

On intermediate exit: contains the current set of Arnoldi basis vectors.

On final exit: contains the final set of Arnoldi basis vectors.

4: LDV – INTEGER Input

On entry: the first dimension of the array V as declared in the (sub)program from which F12APF
is called.

Constraint: LDV � N.

5: Xð�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the dimension of the array X must be at least N if Pointers ¼ NO (default) and at least 1 if
Pointers ¼ YES (see F12ANF).

On initial entry: need not be set, it is used as a convenient mechanism for accessing elements of
COMM.

On intermediate re-entry: if Pointers ¼ YES, X need not be set.

If Pointers ¼ NO, X must contain the result of y ¼ OPx when IREVCM returns the value �1 or
þ1. It must return the computed shifts when IREVCM returns the value 3.

On intermediate exit: if Pointers ¼ YES, X is not referenced.

If Pointers ¼ NO, X contains the vector x when IREVCM returns the value �1 or þ1.

On final exit: does not contain useful data.

6: MXð�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the dimension of the array MX must be at least N if Pointers ¼ NO (default) and at least 1
if Pointers ¼ YES (see F12ANF).

On initial entry: need not be set, it is used as a convenient mechanism for accessing elements of
COMM.

On intermediate re-entry: if Pointers ¼ YES, MX need not be set.

If Pointers ¼ NO, MX must contain the result of y ¼ Bx when IREVCM returns the value 2.

On intermediate exit: if Pointers ¼ YES, MX is not referenced.

If Pointers ¼ NO, MX contains the vector Bx when IREVCM returns the value þ1.

On final exit: does not contain any useful data.

7: NSHIFT – INTEGER Output

On intermediate exit: if the option Supplied Shifts is set and IREVCM returns a value of 3,
NSHIFT returns the number of complex shifts required.

8: COMMð�Þ – COMPLEX (KIND=nag_wp) array Communication Array

Note: the dimension of the array COMM must be at least max 1;LCOMMð Þ (see F12ANF).

On initial entry: must remain unchanged following a call to the setup routine F12ANF.

On exit: contains data defining the current state of the iterative process.

9: ICOMMð�Þ – INTEGER array Communication Array

Note: the dimension of the array ICOMM must be at least max 1;LICOMMð Þ (see F12ANF).

On initial entry: must remain unchanged following a call to the setup routine F12ANF.

On exit: contains data defining the current state of the iterative process.

10: IFAIL – INTEGER Input/Output

On initial entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you
should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On intermediate exit: the value of IFAIL is meaningless and should be ignored.

On final exit: (i.e., when IREVCM ¼ 5) IFAIL ¼ 0, unless the routine detects an error or a
warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On initial entry, the maximum number of iterations � 0, the option Iteration Limit has been set
to a non-positive value.

IFAIL ¼ 2

The options Generalized and Regular are incompatible.

IFAIL ¼ 3

The option Initial Residual was selected but the starting vector held in RESID is zero.

IFAIL ¼ 4

The maximum number of iterations has been reached. Some Ritz values may have converged; a
subsequent call to F12AQF will return the number of converged values and the converged values.

IFAIL ¼ 5

No shifts could be applied during a cycle of the implicitly restarted Arnoldi iteration. One
possibility is to increase the size of NCV relative to NEV (see Section 5 in F12ANF for details of
these arguments).

IFAIL ¼ 6

Could not build an Arnoldi factorization. Consider changing NCV or NEV in the initialization
routine (see Section 5 in F12ANF for details of these arguments).

IFAIL ¼ 7

Unexpected error in internal call to compute eigenvalues and corresponding error bounds of the
current upper Hessenberg matrix. Please contact NAG.

IFAIL ¼ 8

Either the initialization routine F12ANF has not been called prior to the first call of this routine
or a communication array has become corrupted.

IFAIL ¼ 9

An unexpected error has occurred. Please contact NAG.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The relative accuracy of a Ritz value, �, is considered acceptable if its Ritz estimate
� Tolerance� �j j. The default Tolerance used is the machine precision given by X02AJF.

8 Parallelism and Performance

F12APF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

F12APF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example solves Ax ¼ �x in shift-invert mode, where A is obtained from the standard central

difference discretization of the convection-diffusion operator @2u
@x2

þ @2u
@y2

þ �
@u

@x
on the unit square, with

zero Dirichlet boundary conditions. The shift used is a complex number.

10.1 Program Text

Program f12apfe

! F12APF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: dznrm2, f12anf, f12apf, f12aqf, f12arf, f12asf, &

nag_wp, zgttrf, zgttrs
! .. Implicit None Statement ..

Implicit None
! .. Parameters ..

Complex (Kind=nag_wp), Parameter :: one = (1.0_nag_wp,0.0_nag_wp)
Complex (Kind=nag_wp), Parameter :: two = (2.0_nag_wp,0.0_nag_wp)
Integer, Parameter :: imon = 0, nerr = 6, nin = 5, &

nout = 6
! .. Local Scalars ..

Complex (Kind=nag_wp) :: h, h2, rho, s, s1, s2, s3, sigma
Integer :: ifail, info, irevcm, j, lcomm, ldv, &

licomm, n, nconv, ncv, nev, niter, &
nshift, nx

! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: comm(:), d(:,:), dd(:), dl(:), &

du(:), du2(:), mx(:), resid(:), &
v(:,:), x(:)

Integer, Allocatable :: icomm(:), ipiv(:)
! .. Intrinsic Procedures ..

Intrinsic :: cmplx, int, max
! .. Executable Statements ..

Write (nout,*) ’F12APF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)

Read (nin,*) nx, nev, ncv
n = nx*nx

! Initialize communication arrays.
! Query the required sizes of the communication arrays.

licomm = -1
lcomm = -1
Allocate (icomm(max(1,licomm)),comm(max(1,lcomm)))

ifail = 0
Call f12anf(n,nev,ncv,icomm,licomm,comm,lcomm,ifail)

licomm = icomm(1)
lcomm = int(comm(1))
Deallocate (icomm,comm)
Allocate (icomm(max(1,licomm)),comm(max(1,lcomm)))

ifail = 0
Call f12anf(n,nev,ncv,icomm,licomm,comm,lcomm,ifail)

! Set the mode.

ifail = 0
Call f12arf(’SHIFTED INVERSE’,icomm,comm,ifail)

sigma = cmplx(0,kind=nag_wp)
rho = (10.0_nag_wp,0.0_nag_wp)
h = one/cmplx(n+1,kind=nag_wp)
h2 = h*h
s = rho/two

s1 = -one/h2 - s/h
s2 = two/h2 - sigma
s3 = -one/h2 + s/h

Allocate (dl(n-1),dd(n),du(n-1),du2(n-2),ipiv(n))
dl(1:n-1) = s1
dd(1:n-1) = s2
du(1:n-1) = s3
dd(n) = s2

! The NAG name equivalent of zgttrf is f07crf
Call zgttrf(n,dl,dd,du,du2,ipiv,info)

If (info/=0) Then
Write (nerr,99999) info
Go To 110

End If

ldv = n
Allocate (resid(n),v(ldv,ncv),x(n),mx(n),d(ncv,2))

irevcm = 0
ifail = -1

100 Continue

Call f12apf(irevcm,resid,v,ldv,x,mx,nshift,comm,icomm,ifail)

If (irevcm/=5) Then

If (irevcm==-1 .Or. irevcm==1) Then

! Perform x <--- OP*x = inv[A-SIGMA*I]*x

! The NAG name equivalent of zgttrs is f07csf
Call zgttrs(’N’,n,1,dl,dd,du,du2,ipiv,x,n,info)

If (info/=0) Then
Write (nerr,99998) info
Go To 110

End If

Else If (irevcm==4 .And. imon/=0) Then

! Output monitoring information

Call f12asf(niter,nconv,d,d(1,2),icomm,comm)

! The NAG name equivalent of dznrm2 is f06jjf
Write (6,99997) niter, nconv, dznrm2(nev,d(1,2),1)

End If

Go To 100

End If

If (ifail==0) Then

! Post-Process using F12AQF to compute eigenvalues/vectors.

ifail = 0
Call f12aqf(nconv,d,v,ldv,sigma,resid,v,ldv,comm,icomm,ifail)

Write (nout,99996) nconv

Do j = 1, nconv
Write (nout,99995) j, d(j,1)

End Do

End If

110 Continue

99999 Format (1X,’** Error status returned by ZGTTRF, INFO =’,I12)
99998 Format (1X,’** Error status returned by ZGTTRS, INFO =’,I12)
99997 Format (1X,’Iteration’,1X,I3,’, No. converged =’,1X,I3,’, norm o’, &

’f estimates =’,E16.8)
99996 Format (1X,/,’ The ’,I4,’ Ritz values of smallest magnitude are:’,/)
99995 Format (1X,I8,5X,’( ’,F12.4,’ , ’,F12.4,’ )’)

End Program f12apfe

10.2 Program Data

F12APF Example Program Data
10 4 20 : Vaues for NX NEV and NCV

10.3 Program Results

F12APF Example Program Results

The 4 Ritz values of smallest magnitude are:

1 ( 34.8720 , 0.0000 )
2 ( 64.4326 , -0.0000 )
3 ( 113.6685 , 0.0000 )
4 ( 182.5320 , 0.0000 )

