NAG Library Routine Document

F16DPF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F16DPF computes the smallest component of an integer vector, along with the index of that component.

2 Specification

```
SUBROUTINE F16DPF (N, X, INCX, K, I)
INTEGER N, X(1+(N-1)*ABS(INCX)), INCX, K, I
```

3 Description

F16DPF computes the smallest component, i, of an n-element integer vector x, and determines the smallest index, k, such that

$$i = x_k = \min_j x_j.$$

4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee http://www.netlib.org/blas/blast-forum/blas-report.pdf

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of elements in x.

2: X$(1 + (\mathrm{N} - 1) \times |\mathrm{INCX}|)$ – INTEGER array *Input*

On entry: the n-element vector x.

If INCX > 0, x_i must be stored in X$((i-1) \times \mathrm{INCX} + 1)$, for $i = 1, 2, \ldots, \mathrm{N}$.

If INCX < 0, x_i must be stored in X$((\mathrm{N} - i) \times |\mathrm{INCX}| + 1)$, for $i = 1, 2, \ldots, \mathrm{N}$.

Intermediate elements of X are not referenced. If N $= 0$, X is not referenced.

3: INCX – INTEGER *Input*

On entry: the increment in the subscripts of X between successive elements of x.

Constraint: INCX $\neq 0$.

4: K – INTEGER *Output*

On exit: k, the index, from the set $\{1, 2, \ldots, \mathrm{N}\}$, of the smallest component of x. If N ≤ 0 on input then K is returned as 0.

5: I – INTEGER *Output*

On exit: i, the smallest component of x. If N ≤ 0 on input then I is returned as 0.

6 Error Indicators and Warnings

If INCX $= 0$, an error message is printed and program execution is terminated.

7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

8 Parallelism and Performance

F16DPF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example computes the smallest component and index of that component for the vector

$$x = (1, 10, 11, -2, 9)^{\mathrm{T}}.$$

10.1 Program Text

```
    Program f16dpfe

!     F16DPF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: f16dpf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, incx, j, jx, k, n
!     .. Local Arrays ..
      Integer, Allocatable             :: x(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: abs
!     .. Executable Statements ..
      Write (nout,*) 'F16DPF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n, incx
      Allocate (x(1+(n-1)*abs(incx)))

!     Read the vector x and store forwards or backwards
!     as determined by incx.
      If (incx>0) Then
        jx = 1
      Else
        jx = 1 - (n-1)*incx
      End If

      Do j = 1, n
        Read (nin,*) x(jx)
        jx = jx + incx
      End Do
```

```
!     Find k = argmin(x) and i = min(x).

      Call f16dpf(n,x,incx,k,i)

      Write (nout,*)
      Write (nout,99999) k
      Write (nout,99998) i

99999 Format (1X,'Index of smallest component of x is',I3)
99998 Format (1X,'Smallest component of x is',I12)
    End Program f16dpfe
```

10.2 Program Data

```
F16DPF Example Program Data
  5   1                                                 : n and incx
  1
  10
  11
  -2
  9                                                     : Vector x
```

10.3 Program Results

```
 F16DPF Example Program Results

 Index of smallest component of x is  4
 Smallest component of x is          -2
```