

# NAG Library Routine Document

## F16ELF (BLAS\_DSUM)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F16ELF (BLAS\_DSUM) sums the elements of a real vector.

### 2 Specification

```
FUNCTION F16ELF (N, X, INCX)
REAL (KIND=nag_wp) F16ELF
INTEGER                N, INCX
REAL (KIND=nag_wp) X(1+(N-1)*ABS(INCX))
```

The routine may be called by its BLAST name *blas\_dsum*.

### 3 Description

F16ELF (BLAS\_DSUM) returns the sum

$$x_1 + x_2 + \cdots + x_n$$

of the elements of an  $n$ -element real vector  $x$ , via the function name.

If  $N \leq 0$  on entry, F16ELF (BLAS\_DSUM) returns the value 0.

### 4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee <http://www.netlib.org/blas/blast-forum/blas-report.pdf>

### 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the number of elements in  $x$ .
- 2: X(1 + (N – 1) × |INCX|) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* the  $n$ -element vector  $x$ .  
 If  $\text{INCX} > 0$ ,  $x_i$  must be stored in  $X((i - 1) \times \text{INCX} + 1)$ , for  $i = 1, 2, \dots, N$ .  
 If  $\text{INCX} < 0$ ,  $x_i$  must be stored in  $X((N - i) \times |\text{INCX}| + 1)$ , for  $i = 1, 2, \dots, N$ .  
 Intermediate elements of  $X$  are not referenced. If  $N = 0$ ,  $X$  is not referenced.
- 3: INCX – INTEGER *Input*  
*On entry:* the increment in the subscripts of  $X$  between successive elements of  $x$ .  
*Constraint:*  $\text{INCX} \neq 0$ .

### 6 Error Indicators and Warnings

If  $\text{INCX} = 0$ , an error message is printed and program execution is terminated.

## 7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

## 8 Parallelism and Performance

F16ELF (BLAS\_DSUM) is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

This example computes the sum of the elements of

$$x = (1.1, 10.2, 11.5, -2.7, 9.2)^T.$$

### 10.1 Program Text

```

Program f16elfe

!      F16ELF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: blas_dsum, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: sumval
      Integer                     :: i, incx, ix, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: x(:)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: abs
!      .. Executable Statements ..
      Write (nout,*) 'F16ELF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Read (nin,*) n, incx
      Allocate (x(1+(n-1)*abs(incx)))

!      Read the vector x and store forwards or backwards
!      as determined by incx.
      If (incx>0) Then
         ix = 1
      Else
         ix = 1 - (n-1)*incx
      End If

      Do i = 1, n
         Read (nin,*) x(ix)
         ix = ix + incx
      End Do

!      Sum the elements of x

      sumval = blas_dsum(n,x,incx)

```

```
      Write (nout,*)  
      Write (nout,99999) sumval  
  
99999 Format (1X,'Sum of elements of x is',F9.5)  
      End Program f16elfe
```

## 10.2 Program Data

F16ELF Example Program Data

```
5      1                                : n and incx  
1.1  
10.2  
11.5  
-2.7  
9.2                                : Vector x
```

## 10.3 Program Results

F16ELF Example Program Results

Sum of elements of x is 29.30000

---