

NAG Library Routine Document

F16UBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F16UBF calculates the value of the 1-norm, the ∞ -norm, the Frobenius norm or the maximum absolute value of the elements of a complex m by n band matrix stored in banded packed form.

It can also be used to compute the value of the 2-norm of a row n -vector or a column m -vector.

2 Specification

```
FUNCTION F16UBF (INORM, M, N, KL, KU, AB, LDAB)
REAL (KIND=nag_wp) F16UBF
INTEGER                INORM, M, N, KL, KU, LDAB
COMPLEX (KIND=nag_wp) AB(LDAB,*)
```

3 Description

Given a complex m by n band matrix, A , F16UBF calculates one of the values given by

$$\begin{aligned} \|A\|_1 &= \max_j \sum_{i=1}^m |a_{ij}| && \text{(the 1-norm of } A), \\ \|A\|_\infty &= \max_i \sum_{j=1}^n |a_{ij}| && \text{(the } \infty\text{-norm of } A), \\ \|A\|_F &= \left(\sum_{i=1}^m \sum_{j=1}^n |a_{ij}|^2 \right)^{1/2} && \text{(the Frobenius norm of } A), \quad \text{or} \\ &\max_{i,j} |a_{ij}| && \text{(the maximum absolute element value of } A). \end{aligned}$$

If m or n is 1 then additionally F16UBF can calculate the value $\|A\|_2 = \sqrt{\sum |a_i|^2}$ (the 2-norm of A).

4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee <http://www.netlib.org/blas/blast-forum/blas-report.pdf>

5 Arguments

- 1: INORM – INTEGER *Input*
- On entry:* specifies the value to be returned. The integer codes shown below can be replaced by the equivalent named constants of the form NAG_?_NORM. These named constants are available via the **nag_library** module and are also used in the example program for clarity.
- INORM = 171 (NAG_ONE_NORM)
The 1-norm.

INORM = 173 (NAG_TWO_NORM)
The 2-norm of a row or column vector.

INORM = 174 (NAG_FROBENIUS_NORM)
The Frobenius (or Euclidean) norm.

INORM = 175 (NAG_INF_NORM)
The ∞ -norm.

INORM = 177 (NAG_MAX_NORM)
The value $\max_{i,j} |a_{ij}|$ (not a norm).

Constraints:

INORM = 171, 173, 174, 175 or 177;
if INORM = 173, M = 1 or N = 1.

- 2: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A . If $M \leq 0$ on input, F16UBF returns 0.
- 3: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A . If $N \leq 0$ on input, F16UBF returns 0.
- 4: KL – INTEGER *Input*
On entry: k_l , the number of subdiagonals within the band of A . If $KL \leq 0$ on input, F16UBF returns 0.
- 5: KU – INTEGER *Input*
On entry: k_u , the number of superdiagonals within the band of A . If $KU \leq 0$ on input, F16UBF returns 0.
- 6: AB(LDAB,*) – COMPLEX (KIND=nag_wp) array *Input*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the m by n band matrix A .
The matrix is stored in rows 1 to $k_l + k_u + 1$, more precisely, the element A_{ij} must be stored in
$$AB(k_u + 1 + i - j, j) \quad \text{for } \max(1, j - k_u) \leq i \leq \min(m, j + k_l).$$
- 7: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F16UBF is called.
Constraint: $LDAB \geq KL + KU + 1$.

6 Error Indicators and Warnings

If any constraint on an input parameter is violated, an error message is printed and program execution is terminated.

7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

8 Parallelism and Performance

F16UBF is not threaded in any implementation.

9 Further Comments

None.

10 Example

Reads in a 6 by 4 banded complex matrix A with two subdiagonals and one superdiagonal, and prints the four norms of A .

10.1 Program Text

```

Program f16ubfe

!      F16UBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f01zdf, f16ubf, nag_frobenius_norm, nag_inf_norm, &
                             nag_max_norm, nag_one_norm, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: r_fro, r_inf, r_max, r_one
      Integer                     :: i, ifail, j, kl, ku, lda, ldab, m, n
      Character (1)                :: job
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,,:), ab(:,,:)
!      .. Intrinsic Procedures ..
      Intrinsic                    :: max, min
!      .. Executable Statements ..
      Write (nout,*) 'F16UBF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Read (nin,*) m, n, kl, ku
      lda = m
      ldab = kl + ku + 1
      Allocate (a(lda,n),ab(ldab,n))

!      Read A from data file into rectangular storage

      Do i = 1, m
         Read (nin,*)(a(i,j),j=max(1,i-kl),min(n,i+ku))
      End Do

!      Convert A to packed storage

      job = 'P'

      ifail = 0
      Call f01zdf(job,m,n,kl,ku,a,lda,ab,ldab,ifail)

      Write (nout,*)
      Write (nout,99999) 'Norms of banded matrix AB:'
      Write (nout,*)

      r_one = f16ubf(nag_one_norm,m,n,kl,ku,ab,ldab)
      Write (nout,99998) 'One norm'           = ', r_one

```

```

      r_inf = f16ubf(nag_inf_norm,m,n,kl,ku,ab,ldab)
      Write (nout,99998) 'Infinity norm      = ', r_inf

      r_fro = f16ubf(nag_frobenius_norm,m,n,kl,ku,ab,ldab)
      Write (nout,99998) 'Frobenius norm     = ', r_fro

      r_max = f16ubf(nag_max_norm,m,n,kl,ku,ab,ldab)
      Write (nout,99998) 'Maximum norm      = ', r_max

99999 Format (1X,A)
99998 Format (1X,A,F9.4)
      End Program f16ubfe

```

10.2 Program Data

F16UBF Example Program Data

```

6 4 2 1                                : M, N, KL, KU
( 1.0, 1.0) ( 1.0, 2.0)
( 2.0, 1.0) ( 2.0, 2.0) ( 2.0, 3.0)
( 3.0, 1.0) ( 3.0, 2.0) ( 3.0, 3.0) ( 3.0, 4.0)
                ( 4.0, 2.0) ( 4.0, 3.0) ( 4.0, 4.0)
                        ( 5.0, 3.0) ( 5.0, 4.0)
                                ( 6.0, 4.0) : AB

```

10.3 Program Results

F16UBF Example Program Results

Norms of banded matrix AB:

```

One norm           = 24.2711
Infinity norm      = 16.0105
Frobenius norm     = 17.4069
Maximum norm       = 7.2111

```
