
NAG Library Routine Document

G01AUF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

G01AUF combines sets of summaries produced by G01ATF.

2 Specification

SUBROUTINE G01AUF (B, MRCOMM, PN, XMEAN, XSD, XSKEW, XKURT, XMIN, XMAX,
RCOMM, IFAIL)

&

INTEGER B, PN, IFAIL
REAL (KIND=nag_wp) MRCOMM(20,B), XMEAN, XSD, XSKEW, XKURT, XMIN, XMAX,

RCOMM(20)
&

3 Description

Assume a dataset containing n observations, denoted by x ¼ xi : i ¼ 1; 2; . . . ; nf g and a set of weights,
w ¼ wi : i ¼ 1; 2; . . . ; nf g, has been split into b blocks, and each block summarised via a call to
G01ATF. Then G01AUF takes the b communication arrays returned by G01ATF and returns the mean
(�x), standard deviation (s2), coefficients of skewness (s3) and kurtosis (s4), and the maximum and
minimum values for the whole dataset.

For a definition of �x; s2; s3 and s4 see Section 3 in G01ATF.

4 References

West D H D (1979) Updating mean and variance estimates: An improved method Comm. ACM 22 532–
555

5 Arguments

1: B – INTEGER Input

On entry: b, the number of blocks the full dataset was split into.

Constraint: B � 1.

2: MRCOMMð20;BÞ – REAL (KIND=nag_wp) array Communication Array

On entry: the jth column of MRCOMM must contain the information returned in RCOMM from
one of the runs of G01ATF.

3: PN – INTEGER Output

On exit: the number of valid observations, that is the number of observations with wi > 0, for
i ¼ 1; 2; . . . ; n.

4: XMEAN – REAL (KIND=nag_wp) Output

On exit: �x, the mean.

5: XSD – REAL (KIND=nag_wp) Output

On exit: s2, the standard deviation.

6: XSKEW – REAL (KIND=nag_wp) Output

On exit: s3, the coefficient of skewness.

7: XKURT – REAL (KIND=nag_wp) Output

On exit: s4, the coefficient of kurtosis.

8: XMIN – REAL (KIND=nag_wp) Output

On exit: the smallest value.

9: XMAX – REAL (KIND=nag_wp) Output

On exit: the largest value.

10: RCOMMð20Þ – REAL (KIND=nag_wp) array Communication Array

On exit: an amalgamation of the information held in MRCOMM. This is in the same format as
RCOMM from G01ATF.

11: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 11

On entry, B ¼ valueh i.
Constraint: B � 1.

IFAIL ¼ 21

On entry, MRCOMM is not in the expected format.

IFAIL ¼ 31

On entry, the number of valid observations is zero.

IFAIL ¼ 51

On exit we were unable to calculate XSKEW or XKURT. A value of 0 has been returned.

IFAIL ¼ 52

On exit we were unable to calculate XSD, XSKEW or XKURT. A value of 0 has been returned.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G01AUF is not threaded in any implementation.

9 Further Comments

The order that the b communication arrays are stored in MRCOMM is arbitrary. Different orders can
lead to slightly different results due to numerical accuracy of floating-point calculations.

Both G01AUF and G01ATF consolidate results from multiple summaries. Whereas the former can only
be used to combine summaries calculated sequentially, the latter combines summaries calculated in an
arbitrary order allowing, for example, summaries calculated on different processing units to be
combined.

10 Example

This example summarises some simulated data. The data is supplied in three blocks, the first consisting
of 21 observations, the second 51 observations and the last 28 observations. Summaries are produced
for each block of data separately and then an overall summary is produced.

10.1 Program Text

Program g01aufe
! G01AUF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: g01atf, g01auf, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: xkurt, xmax, xmean, xmin, xsd, xskew
Integer :: b, i, ifail, iwt, j, nb, pn

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: mrcomm(:,:), wt(:), x(:)
Real (Kind=nag_wp) :: rcomm(20)

! .. Executable Statements ..
Write (nout,*) ’G01AUF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)

! Read in the number of block of data we have
Read (nin,*) b

Allocate (mrcomm(20,b))

! Loop over each block of data
Do i = 1, b

! Read in the number of observations in this block and the weight flag
Read (nin,*) nb, iwt

! Allocate X to the required size
Allocate (x(nb))

! Read in the data for this block
If (iwt==0) Then

Allocate (wt(0))
Read (nin,*) x(1:nb)

Else
Allocate (wt(nb))
Read (nin,*)(x(j),wt(j),j=1,nb)

End If

! IFAIL = 53, 71 or 72 are warnings and so we don’t want to terminate
! on any nonzero IFAIL. Therefore we set the flag for a quiet exit

ifail = 1

! Summarise this block of data
pn = 0
Call g01atf(nb,x,iwt,wt,pn,xmean,xsd,xskew,xkurt,xmin,xmax, &

mrcomm(1:20,i),ifail)
If (ifail/=0 .And. ifail/=71 .And. ifail/=72 .And. ifail/=53) Then

Write (nout,*) ’G01ATF failed with IFAIL = ’, ifail
Exit

End If

! Display the results for this block
Write (nout,99999) ’Summary for block ’, i
If (ifail==53) Then

Write (nout,*) &
’No valid observations supplied. All weights are zero.’

Else
Write (nout,99997) pn, ’valid observations’
Write (nout,99998) ’ Mean ’, xmean
If (ifail==72) Then

Write (nout,*) &
’ Unable to calculate the standard deviation, skewness or ’, &
’kurtosis’

Else
Write (nout,99998) ’ Std devn ’, xsd
If (ifail==71) Then

Write (nout,*) ’ Unable to calculate the skewness or kurtosis’
Else

Write (nout,99998) ’ Skewness ’, xskew
Write (nout,99998) ’ Kurtosis ’, xkurt

End If
End If
Write (nout,99998) ’ Minimum ’, xmin
Write (nout,99998) ’ Maximum ’, xmax

End If
Write (nout,*)

Deallocate (x,wt)
End Do

If (ifail==0 .Or. ifail==71 .Or. ifail==72 .Or. ifail==53) Then
! Combine the summaries across all the blocks

Call g01auf(b,mrcomm,pn,xmean,xsd,xskew,xkurt,xmin,xmax,rcomm,ifail)

! Display the combined results
Write (nout,99999) ’Summary for the combined data’
If (ifail==53) Then

Write (nout,*) &
’No valid observations supplied. All weights are zero.’

Else
Write (nout,99997) pn, ’valid observations’
Write (nout,99998) ’ Mean ’, xmean
If (ifail==72) Then

Write (nout,*) &
’ Unable to calculate the standard deviation, skewness and ’, &
’kurtosis’

Else
Write (nout,99998) ’ Std devn ’, xsd
If (ifail==71) Then

Write (nout,*) ’ Unable to calculate the skewness and kurtosis’
Else

Write (nout,99998) ’ Skewness ’, xskew
Write (nout,99998) ’ Kurtosis ’, xkurt

End If
End If
Write (nout,99998) ’ Minimum ’, xmin
Write (nout,99998) ’ Maximum ’, xmax

End If
End If

99999 Format (1X,A,I0,A)
99998 Format (1X,A,F13.2)
99997 Format (1X,I0,1X,A)

End Program g01aufe

10.2 Program Data

G01AUF Example Program Data
3 :: B
21 1 :: NB,IWT (1st block)
-0.62 4.91 -1.92 0.25
-1.72 3.90 -6.35 3.75
2.00 1.17 7.65 3.19
6.15 2.66 3.81 0.02
4.87 3.59 -0.51 3.63
6.88 4.83 -5.85 3.72

-0.72 1.72 0.66 0.78
2.23 4.74 -1.61 1.72

-0.15 3.94 -1.15 1.33
-8.74 0.51 -3.94 2.40
3.61 3.90 :: End of X,WT for 1st block
51 0 :: NB,IWT (2nd block)

-0.66 -2.39 -6.25 1.23 2.27 -2.27
10.12 8.29 -2.99 8.71 -0.74 0.02
1.22 1.70 4.30 2.99 -0.83 -1.00
6.57 2.32 -3.47 -1.41 -5.26 0.53
1.80 4.79 -3.04 1.20 -3.21 -3.75
0.86 1.27 -5.95 -5.27 1.63 3.59

-0.01 -1.38 -4.71 -4.82 3.55 0.46
2.57 1.76 -4.05 1.23 -1.99 3.20

-0.65 8.42 -6.01 :: End of X for 2nd block
28 0 :: NB,IWT (3rd block)
1.13 -8.86 5.92 -1.71 -3.99 6.57

-2.01 -2.29 -1.11 7.14 4.84 -4.44
-3.32 10.25 -2.11 8.02 -7.31 2.80
-1.20 1.01 1.37 -2.28 1.28 -3.95
3.43 -0.61 4.85 -0.11 :: End of X for 3rd block

10.3 Program Results

G01AUF Example Program Results

Summary for block 1
21 valid observations

Mean 0.73
Std devn 4.40
Skewness -0.05
Kurtosis -1.00
Minimum -8.74
Maximum 7.65

Summary for block 2
51 valid observations

Mean 0.28
Std devn 3.96
Skewness 0.46
Kurtosis -0.16
Minimum -6.25
Maximum 10.12

Summary for block 3
28 valid observations

Mean 0.48
Std devn 4.65
Skewness 0.19
Kurtosis -0.58
Minimum -8.86
Maximum 10.25

Summary for the combined data
100 valid observations

Mean 0.51
Std devn 4.24
Skewness 0.18
Kurtosis -0.59
Minimum -8.86
Maximum 10.25

