

NAG Library Routine Document

G01EBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01EBF returns the lower tail, upper tail or two tail probability for the Student's t -distribution with real degrees of freedom, via the routine name.

2 Specification

```
FUNCTION G01EBF (TAIL, T, DF, IFAIL)
REAL (KIND=nag_wp) G01EBF
INTEGER                IFAIL
REAL (KIND=nag_wp) T, DF
CHARACTER(1)           TAIL
```

3 Description

The lower tail probability for the Student's t -distribution with ν degrees of freedom, $P(T \leq t : \nu)$ is defined by:

$$P(T \leq t : \nu) = \frac{\Gamma((\nu+1)/2)}{\sqrt{\pi\nu}\Gamma(\nu/2)} \int_{-\infty}^t \left[1 + \frac{T^2}{\nu}\right]^{-(\nu+1)/2} dT, \quad \nu \geq 1.$$

Computationally, there are two situations:

- (i) when $\nu < 20$, a transformation of the beta distribution, $P_\beta(B \leq \beta : a, b)$ is used

$$P(T \leq t : \nu) = \frac{1}{2}P_\beta\left(B \leq \frac{\nu}{\nu + t^2} : \nu/2, \frac{1}{2}\right) \quad \text{when } t < 0.0$$

or

$$P(T \leq t : \nu) = \frac{1}{2} + \frac{1}{2}P_\beta\left(B \geq \frac{\nu}{\nu + t^2} : \nu/2, \frac{1}{2}\right) \quad \text{when } t > 0.0;$$

- (ii) when $\nu \geq 20$, an asymptotic normalizing expansion of the Cornish–Fisher type is used to evaluate the probability, see Hill (1970).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Hill G W (1970) Student's t -distribution *Comm. ACM* **13**(10) 617–619

5 Arguments

1: TAIL – CHARACTER(1)

Input

On entry: indicates which tail the returned probability should represent.

TAIL = 'U'

The upper tail probability is returned, i.e., $P(T \geq t : \nu)$.

TAIL = 'S'

The two tail (significance level) probability is returned,
i.e., $P(T \geq |t| : \nu) + P(T \leq -|t| : \nu)$.

TAIL = 'C'

The two tail (confidence interval) probability is returned,
i.e., $P(T \leq |t| : \nu) - P(T \leq -|t| : \nu)$.

TAIL = 'L'

The lower tail probability is returned, i.e., $P(T \leq t : \nu)$.

Constraint: TAIL = 'U', 'S', 'C' or 'L'.

2: T – REAL (KIND=nag_wp) *Input*

On entry: t , the value of the Student's t variate.

3: DF – REAL (KIND=nag_wp) *Input*

On entry: ν , the degrees of freedom of the Student's t -distribution.

Constraint: $DF \geq 1.0$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If IFAIL \neq 0, then G01EBF returns 0.0.

IFAIL = 1

On entry, TAIL \neq 'U', 'S', 'C' or 'L'.

IFAIL = 2

On entry, $DF < 1.0$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed probability should be accurate to five significant places for reasonable probabilities but there will be some loss of accuracy for very low probabilities (less than 10^{-10}), see Hastings and Peacock (1975).

8 Parallelism and Performance

G01EBF is not threaded in any implementation.

9 Further Comments

The probabilities could also be obtained by using the appropriate transformation to a beta distribution (see Abramowitz and Stegun (1972)) and using G01EEF. This routine allows you to set the required accuracy.

10 Example

This example reads values from, and degrees of freedom for Student's t -distributions along with the required tail. The probabilities are calculated and printed until the end of data is reached.

10.1 Program Text

```

Program g01ebfe

!      G01EBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g01ebf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: df, prob, t
      Integer                     :: ifail
      Character (1)                :: tail
!      .. Executable Statements ..
      Write (nout,*) 'G01EBF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '  TAIL      T      DF      Probability'
      Write (nout,*)

d_lp: Do
      Read (nin,*,Iostat=ifail) tail, t, df
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate probability
      ifail = 0
      prob = g01ebf(tail,t,df,ifail)

```

```
!      Display results
      Write (nout,99999) tail, t, df, prob
      End Do d_lp

99999 Format (3X,A1,4X,F6.3,F8.1,7X,F7.4)
      End Program g01ebfe
```

10.2 Program Data

```
G01EBF Example Program Data
'L' 0.85 20.0 : TAIL T DF
'S' 0.85 20.0
'C' 0.85 20.0
'U' 0.85 20.0
```

10.3 Program Results

G01EBF Example Program Results

TAIL	T	DF	Probability
L	0.850	20.0	0.7973
S	0.850	20.0	0.4054
C	0.850	20.0	0.5946
U	0.850	20.0	0.2027
