

NAG Library Routine Document

G01FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01FCF returns the deviate associated with the given lower tail probability of the χ^2 -distribution with real degrees of freedom, via the routine name.

2 Specification

```
FUNCTION G01FCF (P, DF, IFAIL)
  REAL (KIND=nag_wp) G01FCF
  INTEGER IFAIL
  REAL (KIND=nag_wp) P, DF
```

3 Description

The deviate, x_p , associated with the lower tail probability p of the χ^2 -distribution with ν degrees of freedom is defined as the solution to

$$P(X \leq x_p : \nu) = p = \frac{1}{2^{\nu/2} \Gamma(\nu/2)} \int_0^{x_p} e^{-X/2} X^{\nu/2-1} dX, \quad 0 \leq x_p < \infty; \nu > 0.$$

The required x_p is found by using the relationship between a χ^2 -distribution and a gamma distribution, i.e., a χ^2 -distribution with ν degrees of freedom is equal to a gamma distribution with scale parameter 2 and shape parameter $\nu/2$.

For very large values of ν , greater than 10^5 , Wilson and Hilferty's normal approximation to the χ^2 is used; see Kendall and Stuart (1969).

4 References

Best D J and Roberts D E (1975) Algorithm AS 91. The percentage points of the χ^2 distribution *Appl. Statist.* **24** 385–388

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

5 Arguments

- | | | |
|----|---|--------------|
| 1: | P – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> p , the lower tail probability from the required χ^2 -distribution. | |
| | <i>Constraint:</i> $0.0 \leq P < 1.0$. | |
| 2: | DF – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> ν , the degrees of freedom of the χ^2 -distribution. | |
| | <i>Constraint:</i> DF > 0.0. | |

3: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $IFAIL \neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01FCF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If IFAIL = 1, 2, 3 or 5 on exit, then G01FCF returns 0.0.

IFAIL = 1

On entry, $P < 0.0$,
or $P \geq 1.0$.

IFAIL = 2

On entry, $DF \leq 0.0$.

IFAIL = 3

P is too close to 0 or 1 for the result to be calculated.

IFAIL = 4

The solution has failed to converge. The result should be a reasonable approximation.

IFAIL = 5

The series used to calculate the gamma function has failed to converge. This is an unlikely error exit.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The results should be accurate to five significant digits for most argument values. Some accuracy is lost for p close to 0.0.

8 Parallelism and Performance

G01FCF is not threaded in any implementation.

9 Further Comments

For higher accuracy the relationship described in Section 3 may be used and a direct call to G01FFF made.

10 Example

This example reads lower tail probabilities for several χ^2 -distributions, and calculates and prints the corresponding deviates until the end of data is reached.

10.1 Program Text

```

Program g01fcfe

!      G01FCF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g01fcf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: df, p, x
      Integer                     :: ifail
!      .. Executable Statements ..
      Write (nout,*) 'G01FCF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '          P          DF          X'
      Write (nout,*)

d_lp: Do
      Read (nin,*,Iostat=ifail) p, df
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate deviates (inverse CDF)
      ifail = -1
      x = g01fcf(p,df,ifail)
      If (ifail/=0) Then
         If (ifail/=4 .And. ifail/=5) Then
            Exit d_lp
         End If
      End If

!      Display results

```

```
      Write (nout,99999) p, df, x
      End Do d_lp

99999 Format (1X,3F8.3,A,I1)
      End Program g01fcfe
```

10.2 Program Data

G01FCF Example Program Data

0.0100	20.0		:P DF
0.4279	7.5		:P DF
0.8694	45.0		:P DF

10.3 Program Results

G01FCF Example Program Results

P	DF	X
0.010	20.000	8.260
0.428	7.500	6.200
0.869	45.000	55.759
