

NAG Library Routine Document

G01GCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01GCF returns the probability associated with the lower tail of the noncentral χ^2 -distribution via the routine name.

2 Specification

```
FUNCTION G01GCF (X, DF, RLAMDA, TOL, MAXIT, IFAIL)
REAL (KIND=nag_wp) G01GCF
INTEGER                MAXIT, IFAIL
REAL (KIND=nag_wp) X, DF, RLAMDA, TOL
```

3 Description

The lower tail probability of the noncentral χ^2 -distribution with ν degrees of freedom and noncentrality parameter λ , $P(X \leq x : \nu; \lambda)$, is defined by

$$P(X \leq x : \nu; \lambda) = \sum_{j=0}^{\infty} e^{-\lambda/2} \frac{(\lambda/2)^j}{j!} P(X \leq x : \nu + 2j; 0), \quad (1)$$

where $P(X \leq x : \nu + 2j; 0)$ is a central χ^2 -distribution with $\nu + 2j$ degrees of freedom.

The value of j at which the Poisson weight, $e^{-\lambda/2} \frac{(\lambda/2)^j}{j!}$, is greatest is determined and the summation (1) is made forward and backward from that value of j .

The recursive relationship:

$$P(X \leq x : a + 2; 0) = P(X \leq x : a; 0) - \frac{(x^a/2)e^{-x/2}}{\Gamma(a+1)} \quad (2)$$

is used during the summation in (1).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Arguments

- 1: X – REAL (KIND=nag_wp) *Input*
On entry: the deviate from the noncentral χ^2 -distribution with ν degrees of freedom and noncentrality parameter λ .
Constraint: $X \geq 0.0$.
- 2: DF – REAL (KIND=nag_wp) *Input*
On entry: ν , the degrees of freedom of the noncentral χ^2 -distribution.
Constraint: $DF \geq 0.0$.

- 3: RLAMDA – REAL (KIND=nag_wp) Input
On entry: λ , the noncentrality parameter of the noncentral χ^2 -distribution.
Constraint: RLAMDA ≥ 0.0 if DF > 0.0 or RLAMDA > 0.0 if DF = 0.0.
- 4: TOL – REAL (KIND=nag_wp) Input
On entry: the required accuracy of the solution. If G01GCF is entered with TOL greater than or equal to 1.0 or less than $10 \times \text{machine precision}$ (see X02AJF), then the value of $10 \times \text{machine precision}$ is used instead.
- 5: MAXIT – INTEGER Input
On entry: the maximum number of iterations to be performed.
Suggested value: 100. See Section 9 for further discussion.
Constraint: MAXIT ≥ 1 .
- 6: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01GCF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit IFAIL = 1, 2, 4 or 5, then G01GCF returns 0.0.

IFAIL = 1

On entry, DF < 0.0 ,
 or RLAMDA < 0.0 ,
 or DF = 0.0 and RLAMDA = 0.0,
 or X < 0.0 ,
 or MAXIT < 1 .

IFAIL = 2

The initial value of the Poisson weight used in the summation (1) was too small to be calculated. The value of $P(X \leq x : \nu; \lambda)$ is likely to be zero.

IFAIL = 3

The solution has failed to converge in MAXIT iterations.

IFAIL = 4

The value of a term required in (2) is too large to be evaluated accurately. The most likely cause of this error is both X and RLAMDA being very large.

IFAIL = 5

The calculations for the central χ^2 probability has failed to converge. This is an unlikely error exit. A larger value of TOL should be used.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The summations described in Section 3 are made until an upper bound on the truncation error relative to the current summation value is less than TOL.

8 Parallelism and Performance

G01GCF is not threaded in any implementation.

9 Further Comments

The number of terms in (1) required for a given accuracy will depend on the following factors:

- (i) The rate at which the Poisson weights tend to zero. This will be slower for larger values of λ .
- (ii) The rate at which the central χ^2 probabilities tend to zero. This will be slower for larger values of ν and x .

10 Example

This example reads values from various noncentral χ^2 -distributions, calculates the lower tail probabilities and prints all these values until the end of data is reached.

10.1 Program Text

```

Program g01gcfe

!      G01GCF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g01gcf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6

```

```

!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: df, prob, rlamda, tol, x
      Integer                     :: ifail, maxit
!      .. Executable Statements ..
      Write (nout,*) 'G01GCF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '          X          DF    RLAMDA    PROB'
      Write (nout,*)

!      Use default tolerance and suggested number of iterations
      tol = 0.0E0_nag_wp
      maxit = 100

d_lp: Do
      Read (nin,*,Iostat=ifail) x, df, rlamda
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate probability
      ifail = 0
      prob = g01gcf(x,df,rlamda,tol,maxit,ifail)

!      Display results
      Write (nout,99999) x, df, rlamda, prob
End Do d_lp

99999 Format (1X,3F8.3,F8.4,A,I1)
End Program g01gcfe

```

10.2 Program Data

G01GCF Example Program Data

8.26	20.0	3.5	:X DF RLAMDA
6.2	7.5	2.0	:X DF RLAMDA
55.76	45.0	1.0	:X DF RLAMDA

10.3 Program Results

G01GCF Example Program Results

X	DF	RLAMDA	PROB
8.260	20.000	3.500	0.0032
6.200	7.500	2.000	0.2699
55.760	45.000	1.000	0.8443
