NAG Library Routine Document

G01GDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01GDF returns the probability associated with the lower tail of the noncentral F or variance-ratio distribution, via the routine name.

2 Specification

```
FUNCTION G01GDF (F, DF1, DF2, RLAMDA, TOL, MAXIT, IFAIL)
REAL (KIND=nag_wp) G01GDF
INTEGER            MAXIT, IFAIL
REAL (KIND=nag_wp) F, DF1, DF2, RLAMDA, TOL
```

3 Description

The lower tail probability of the noncentral F-distribution with ν_1 and ν_2 degrees of freedom and noncentrality parameter λ, $P(F \leq f : \nu_1, \nu_2; \lambda)$, is defined by

$$P(F \leq f : \nu_1, \nu_2; \lambda) = \int_0^x p(F : \nu_1, \nu_2; \lambda)\, dF,$$

where

$$P(F : \nu_1, \nu_2; \lambda) = \sum_{j=0}^{\infty} e^{-\lambda/2} \frac{(\lambda/2)^j}{j!} \times \frac{(\nu_1 + 2j)^{(\nu_1+2j)/2} \nu_2^{\nu_2/2}}{B((\nu_1 + 2j)/2, \nu_2/2)}$$

$$\times u^{(\nu_1+2j-2)/2} [\nu_2 + (\nu_1 + 2j)u]^{-(\nu_1+2j+\nu_2)/2}$$

and $B(\cdot, \cdot)$ is the beta function.

The probability is computed by means of a transformation to a noncentral beta distribution:

$$P(F \leq f : \nu_1, \nu_2; \lambda) = P_\beta(X \leq x : a, b; \lambda),$$

where $x = \dfrac{\nu_1 f}{\nu_1 f + \nu_2}$ and $P_\beta(X \leq x : a, b; \lambda)$ is the lower tail probability integral of the noncentral beta distribution with parameters a, b, and λ.

If ν_2 is very large, greater than 10^6, then a χ^2 approximation is used.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Arguments

1: F – REAL (KIND=nag_wp) *Input*

On entry: f, the deviate from the noncentral F-distribution.

Constraint: F > 0.0.

2: DF1 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the numerator variance, ν_1.

Constraint: $0.0 < \text{DF1} \leq 10^6$.

3: DF2 – REAL (KIND=nag_wp) *Input*

On entry: the degrees of freedom of the denominator variance, ν_2.

Constraint: $\text{DF2} > 0.0$.

4: RLAMDA – REAL (KIND=nag_wp) *Input*

On entry: λ, the noncentrality parameter.

Constraint: $0.0 \leq \text{RLAMDA} \leq -2.0\log(U)$ where U is the safe range parameter as defined by X02AMF.

5: TOL – REAL (KIND=nag_wp) *Input*

On entry: the relative accuracy required by you in the results. If G01GDF is entered with TOL greater than or equal to 1.0 or less than $10 \times$ ***machine precision*** (see X02AJF), then the value of $10 \times$ ***machine precision*** is used instead.

6: MAXIT – INTEGER *Input*

On entry: the maximum number of iterations to be used.

Suggested value: 500. See G01GCF and G01GEF for further details.

Constraint: $\text{MAXIT} \geq 1$.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $\text{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01GDF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit $\text{IFAIL} = 1$ or 3, then G01GDF returns 0.0.

$\text{IFAIL} = 1$

On entry, $\text{DF1} \leq 0.0$,
or $\text{DF1} > 10^6$,
or $\text{DF2} \leq 0.0$,
or $\text{F} \leq 0.0$,
or $\text{RLAMDA} < 0.0$,

or MAXIT < 1,
or RLAMDA $> -2.0\log(U)$, where $U =$ safe range argument as defined by X02AMF.

IFAIL $= 2$

The solution has failed to converge in MAXIT iterations. You should try a larger value of MAXIT or TOL.

IFAIL $= 3$

The required probability cannot be computed accurately. This may happen if the result would be very close to 0.0 or 1.0. Alternatively the values of DF1 and F may be too large. In the latter case you could try using a normal approximation; see Abramowitz and Stegun (1972).

IFAIL $= 4$

The required accuracy was not achieved when calculating the initial value of the central F (or χ^2) probability. You should try a larger value of TOL. If the χ^2 approximation is being used then G01GDF returns zero otherwise the value returned should be an approximation to the correct value.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The relative accuracy should be as specified by TOL. For further details see G01GCF and G01GEF.

8 Parallelism and Performance

G01GDF is not threaded in any implementation.

9 Further Comments

When both ν_1 and ν_2 are large a Normal approximation may be used and when only ν_1 is large a χ^2 approximation may be used. In both cases λ is required to be of the same order as ν_1. See Abramowitz and Stegun (1972) for further details.

10 Example

This example reads values from, and degrees of freedom for, F-distributions, computes the lower tail probabilities and prints all these values until the end of data is reached.

10.1 Program Text

```
    Program g01gdfe

!     G01GDF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g01gdf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: df1, df2, f, prob, rlamda, tol
      Integer                          :: ifail, maxit
!     .. Executable Statements ..
      Write (nout,*) 'G01GDF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) '     F      DF1     DF2   RLAMDA   PROB'
      Write (nout,*)

!     Use default tolerance and suggested number of iterations
      tol = 0.0E0_nag_wp
      maxit = 100

d_lp: Do
        Read (nin,*,Iostat=ifail) f, df1, df2, rlamda
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Calculate probability
        ifail = -1
        prob = g01gdf(f,df1,df2,rlamda,tol,maxit,ifail)
        If (ifail/=0) Then
          If (ifail<3) Then
            Exit d_lp
          End If
        End If

!       Display results
        Write (nout,99999) f, df1, df2, rlamda, prob
      End Do d_lp

99999 Format (1X,4F8.3,F8.4,A,I1)
    End Program g01gdfe
```

10.2 Program Data

```
G01GDF Example Program Data
  5.5   1.5    25.5   3.0          :F DF1 DF2 RLAMDA
 39.9   1.0     1.0   2.0          :F DF1 DF2 RLAMDA
  2.5  20.25    1.0   0.0          :F DF1 DF2 RLAMDA
```

10.3 Program Results

```
 G01GDF Example Program Results

      F       DF1      DF2    RLAMDA    PROB

   5.500    1.500   25.500    3.000   0.8214
  39.900    1.000    1.000    2.000   0.8160
   2.500   20.250    1.000    0.000   0.5342
```