

NAG Library Routine Document

G01JCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01JCF returns the lower tail probability of a distribution of a positive linear combination of χ^2 random variables.

2 Specification

```
SUBROUTINE G01JCF (A, MULT, RLAMDA, N, C, P, PDF, TOL, MAXIT, WRK,      &
                  IFAIL)
INTEGER                MULT(N), N, MAXIT, IFAIL
REAL (KIND=nag_wp) A(N), RLAMDA(N), C, P, PDF, TOL, WRK(N+2*MAXIT)
```

3 Description

For a linear combination of noncentral χ^2 random variables with integer degrees of freedom the lower tail probability is

$$P\left(\sum_{j=1}^n a_j \chi^2(m_j, \lambda_j) \leq c\right), \quad (1)$$

where a_j and c are positive constants and where $\chi^2(m_j, \lambda_j)$ represents an independent χ^2 random variable with m_j degrees of freedom and noncentrality argument λ_j . The linear combination may arise from considering a quadratic form in Normal variables.

Ruben's method as described in Farebrother (1984) is used. Ruben has shown that (1) may be expanded as an infinite series of the form

$$\sum_{k=0}^{\infty} d_k F(m + 2k, c/\beta), \quad (2)$$

where $F(m + 2k, c/\beta) = P(\chi^2(m + 2k) < c/\beta)$, i.e., the probability that a central χ^2 is less than c/β .

The value of β is set at

$$\beta = \beta_B = \frac{2}{(1/a_{\min} + 1/a_{\max})}$$

unless $\beta_B > 1.8a_{\min}$, in which case

$$\beta = \beta_A = a_{\min}$$

is used, where $a_{\min} = \min\{a_j\}$ and $a_{\max} = \max\{a_j\}$, for $j = 1, 2, \dots, n$.

4 References

Farebrother R W (1984) The distribution of a positive linear combination of χ^2 random variables *Appl. Statist.* **33**(3)

5 Arguments

- 1: A(N) – REAL (KIND=nag_wp) array *Input*
On entry: the weights, a_1, a_2, \dots, a_n .
Constraint: $A(i) > 0.0$, for $i = 1, 2, \dots, n$.

- 2: MULT(N) – INTEGER array *Input*
On entry: the degrees of freedom, m_1, m_2, \dots, m_n .
Constraint: $MULT(i) \geq 1$, for $i = 1, 2, \dots, N$.

- 3: RLAMDA(N) – REAL (KIND=nag_wp) array *Input*
On entry: the noncentrality parameters, $\lambda_1, \lambda_2, \dots, \lambda_n$.
Constraint: $RLAMDA(i) \geq 0.0$, for $i = 1, 2, \dots, n$.

- 4: N – INTEGER *Input*
On entry: n , the number of χ^2 random variables in the combination, i.e., the number of terms in equation (1).
Constraint: $N \geq 1$.

- 5: C – REAL (KIND=nag_wp) *Input*
On entry: c , the point for which the lower tail probability is to be evaluated.
Constraint: $C \geq 0.0$.

- 6: P – REAL (KIND=nag_wp) *Output*
On exit: the lower tail probability associated with the linear combination of n χ^2 random variables with m_j degrees of freedom, and noncentrality arguments λ_j , for $j = 1, 2, \dots, n$.

- 7: PDF – REAL (KIND=nag_wp) *Output*
On exit: the value of the probability density function of the linear combination of χ^2 variables.

- 8: TOL – REAL (KIND=nag_wp) *Input*
On entry: the relative accuracy required by you in the results. If G01JCF is entered with TOL greater than or equal to 1.0 or less than $10 \times \textit{machine precision}$ (see X02AJF), then the value of $10 \times \textit{machine precision}$ is used instead.

- 9: MAXIT – INTEGER *Input*
On entry: the maximum number of terms that should be used during the summation.
Suggested value: 500.
Constraint: $MAXIT \geq 1$.

- 10: WRK(N + 2 × MAXIT) – REAL (KIND=nag_wp) array *Workspace*

- 11: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output

arguments may be useful even if $IFAIL \neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of $IFAIL$ on exit.**

On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01JCF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If on exit $IFAIL = 1$ or 2 , then G01JCF returns 0.0 .

$IFAIL = 1$

On entry, $N < 1$,
or $MAXIT < 1$,
or $C < 0.0$.

$IFAIL = 2$

On entry, A has an element ≤ 0.0 ,
or $MULT$ has an element < 1 ,
or $RLAMDA$ has an element < 0.0 .

$IFAIL = 3$

The central χ^2 calculation has failed to converge. This is an unlikely exit. A larger value of TOL should be tried.

$IFAIL = 4$

The solution has failed to converge within MAXIT iterations. A larger value of MAXIT or TOL should be used. The returned value should be a reasonable approximation to the correct value.

$IFAIL = 5$

The solution appears to be too close to 0 or 1 for accurate calculation. The value returned is 0 or 1 as appropriate.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The series (2) is summed until a bound on the truncation error is less than TOL. See Farebrother (1984) for further discussion.

8 Parallelism and Performance

G01JCF is not threaded in any implementation.

9 Further Comments

None.

10 Example

The number of χ^2 variables is read along with their coefficients, degrees of freedom and noncentrality arguments. The lower tail probability is then computed and printed.

10.1 Program Text

```

Program g01jcfe

!      G01JCF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g01jcf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: c, p, pdf, tol
      Integer                     :: i, ifail, lwrk, maxit, n, pn
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:), rlamda(:), wrk(:)
      Integer, Allocatable          :: mult(:)
!      .. Executable Statements ..
      Write (nout,*) 'G01JCF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '          A      MULT  RLAMDA'
      Write (nout,*)

!      Use default tolerance
      tol = 0.0E0_nag_wp
      maxit = 500

!      Dummy allocation for the arrays
      Allocate (a(1),rlamda(1),mult(1),wrk(1))

      pn = 0
d_lp: Do
      Read (nin,*,Iostat=ifail) n, c
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Reallocate arrays if required
      If (pn/=n) Then
         Deallocate (a,rlamda,mult,wrk)

```

```

        lwrk = n + 2*maxit
        Allocate (a(n),rlamda(n),mult(n),wrk(lwrk))
End If
pn = n

!      Read in weights, degrees of freedom and distribution parameter
Read (nin,*) a(1:n)
Read (nin,*) mult(1:n)
Read (nin,*) rlamda(1:n)

!      Calculate probability
ifail = -1
Call g01jcf(a,mult,rlamda,n,c,p,pdf,tol,maxit,wrk,ifail)
If (ifail/=0) Then
    If (ifail<4) Then
        Exit d_lp
    End If
End If

!      Display results
Write (nout,99999)(a(i),mult(i),rlamda(i),i=1,n)
Write (nout,99998) 'C = ', c, '      PROB =', p
End Do d_lp

99999 Format (1X,F10.2,I6,F9.2)
99998 Format (1X,A,F6.2,A,F7.4)
End Program g01jcf

```

10.2 Program Data

G01JCF Example Program Data

3	20.0			:N C
6.0	3.0	1.0		:A(I), I=1,N
1	1	1		:MULT(I), I=1,N
0.0	0.0	0.0		:RLAMDA(I), I=1,N
2	10.0			:N C
7.0	3.0			:A(I), I=1,N
1	1			:MULT(I), I=1,N
6.0	2.0			:RLAMDA(I), I=1,N

10.3 Program Results

G01JCF Example Program Results

	A	MULT	RLAMDA
	6.00	1	0.00
	3.00	1	0.00
	1.00	1	0.00
C =	20.00	PROB =	0.8760
	7.00	1	6.00
	3.00	1	2.00
C =	10.00	PROB =	0.0451
