

# NAG Library Routine Document

## G01MBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G01MBF returns the reciprocal of Mills' Ratio, via the routine name.

### 2 Specification

```
FUNCTION G01MBF (X)
REAL (KIND=nag_wp) G01MBF
REAL (KIND=nag_wp) X
```

### 3 Description

G01MBF calculates the reciprocal of Mills' Ratio, the hazard rate,  $\lambda(x)$ , for the standard Normal distribution. It is defined as the ratio of the ordinate to the upper tail area of the standard Normal distribution, that is,

$$\lambda(x) = \frac{Z(x)}{Q(x)} = \frac{\frac{1}{\sqrt{2\pi}}e^{-(x^2/2)}}{\frac{1}{\sqrt{2\pi}}\int_x^\infty e^{-(t^2/2)} dt}.$$

The calculation is based on a Chebyshev expansion as described in S15AGF.

### 4 References

Gross A J and Clark V A (1975) *Survival Distributions: Reliability Applications in the Biomedical Sciences* Wiley

### 5 Arguments

1: X – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $x$ , the argument of the reciprocal of Mills' Ratio.

### 6 Error Indicators and Warnings

None.

### 7 Accuracy

In the left-hand tail,  $x < 0.0$ , if  $\frac{1}{2}e^{-(1/2)x^2} \leq$  the safe range argument (X02AMF), then 0.0 is returned, which is close to the true value.

The relative accuracy is bounded by the effective ***machine precision***. See S15AGF for further discussion.

### 8 Parallelism and Performance

G01MBF is not threaded in any implementation.

## 9 Further Comments

If, before entry,  $x$  is not a standard Normal variable, it has to be standardized, and on exit, G01MBF has to be divided by the standard deviation. That is, if the Normal distribution has mean  $\mu$  and variance  $\sigma^2$ , then its hazard rate,  $\lambda(x; \mu, \sigma^2)$ , is given by

$$\lambda(x; \mu, \sigma^2) = \lambda((x - \mu)/\sigma)/\sigma.$$

## 10 Example

The hazard rate is evaluated at different values of  $x$  for Normal distributions with different means and variances. The results are then printed.

### 10.1 Program Text

```

Program g01mbfe

!      G01MBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: a00acf, g01mbf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: rm, x, xmu, xsig, z
      Integer                      :: ifail
!      .. Executable Statements ..
      Write (nout,*) 'G01MBF Example Program Results '
      Write (nout,*)

!      Check for valid licence prior to calling G01MBF
      If (.Not. a00acf()) Then
         Write (nout,*) ' ** A valid licence key was not found'

      Else
!         Skip heading in data file
         Read (nin,*)

!         Display titles
         Write (nout,*) ' Mean      Sigma      X      Reciprocal'
         Write (nout,*) '                      Mills Ratio'
         Write (nout,*)

d_lp:   Do
         Read (nin,*,Iostat=ifail) x, xmu, xsig
         If (ifail/=0) Then
            Exit d_lp
         End If

         z = (x-xmu)/xsig
         rm = g01mbf(z)/xsig

!         Display results
         Write (nout,99999) xmu, xsig, x, rm
      End Do d_lp
      End If

99999 Format (1X,4(F7.4,2X))
      End Program g01mbfe

```

## 10.2 Program Data

G01MBF Example Program Data

0.0 0.0 1.0

-2.0 1.0 2.5

10.3 9.0 1.6

## 10.3 Program Results

G01MBF Example Program Results

Mean	Sigma	X	Reciprocal Mills Ratio
0.0000	1.0000	0.0000	0.7979
1.0000	2.5000	-2.0000	0.0878
9.0000	1.6000	10.3000	0.8607

---