

# NAG Library Routine Document

## G02BWF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G02BWF calculates a matrix of Pearson product-moment correlation coefficients from sums of squares and cross-products of deviations about the mean.

### 2 Specification

```
SUBROUTINE G02BWF (M, R, IFAIL)
  INTEGER          M, IFAIL
  REAL (KIND=nag_wp) R((M*M+M)/2)
```

### 3 Description

G02BWF calculates a matrix of Pearson product-moment correlation coefficients from sums of squares and cross-products about the mean for observations on  $m$  variables which can be computed by a single call to G02BUF or a series of calls to G02BTF. The sums of squares and cross-products are stored in an array packed by column and are overwritten by the correlation coefficients.

Let  $c_{jk}$  be the cross-product of deviations from the mean, for  $j = 1, 2, \dots, m$  and  $k = j, \dots, m$ , then the product-moment correlation coefficient,  $r_{jk}$  is given by

$$r_{jk} = \frac{c_{jk}}{\sqrt{c_{jj}c_{kk}}}.$$

### 4 References

None.

### 5 Arguments

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of variables.  
*Constraint:*  $M \geq 1$ .
- 2: R((M × M + M)/2) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* contains the upper triangular part of the sums of squares and cross-products matrix of deviations from the mean. These are stored packed by column, i.e., the cross-product between variable  $j$  and  $k$ ,  $k \geq j$ , is stored in  $R((k \times (k - 1)/2 + j))$ .  
*On exit:* the Pearson product-moment correlation coefficients.  
 These are stored packed by column corresponding to the input cross-products.
- 3: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then

the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if  $IFAIL \neq 0$  on exit, the recommended value is  $-1$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of  $IFAIL$  on exit.**

*On exit:*  $IFAIL = 0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** G02BWF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $M < 1$ .

$IFAIL = 2$

A variable has a zero variance. All correlations involving the variable with zero variance will be returned as zero.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The accuracy of G02BWF is entirely dependent upon the accuracy of the elements of array R.

## 8 Parallelism and Performance

G02BWF is not threaded in any implementation.

## 9 Further Comments

G02BWF may also be used to calculate the correlations between parameter estimates from the variance-covariance matrix of the parameter estimates as is given by several routines in this chapter.

## 10 Example

A program to calculate the correlation matrix from raw data. The sum of squares and cross-products about the mean are calculated from the raw data by a call to G02BUF. The correlation matrix is then calculated from these values.

## 10.1 Program Text

```

Program g02bwfe

!      G02BWF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g02buf, g02bwf, nag_wp, x04ccf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: sw
      Integer                    :: i, ifail, ldx, lr, lwt, m, n
      Logical                    :: zero_var
      Character (1)              :: mean, weight
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: r(:), wmean(:), wt(:), x(:, :)
!      .. Executable Statements ..
      Write (nout,*) 'G02BWF Example Program Results'
      Write (nout,*)
      Flush (nout)

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) mean, weight, m, n

      If (weight=='W' .Or. weight=='w') Then
         lwt = n
      Else
         lwt = 0
      End If
      ldx = n
      lr = (m*m+m)/2
      Allocate (r(lr),wmean(m),wt(lwt),x(ldx,m))

!      Read in data
      Read (nin,*) wt(1:n)
      Read (nin,*)(x(i,1:m),i=1,n)

!      Calculate the sums of squares and cross-products matrix
      ifail = 0
      Call g02buf(mean,weight,n,m,x,ldx,wt,sw,wmean,r,ifail)

!      Calculate the correlation matrix
      ifail = -1
      Call g02bwf(m,r,ifail)
      If (ifail/=0) Then
         If (ifail==2) Then
            zero_var = .True.
         Else
            Go To 100
         End If
      Else
         zero_var = .False.
      End If

!      Display the results
      ifail = 0
      Call x04ccf('Upper','Non-unit',m,r,'Correlation matrix',ifail)
      If (zero_var) Then
         Write (nout,*) ' NOTE: some variances are zero'

```

```
        End If
100    Continue
        End Program g02bwfe
```

## 10.2 Program Data

G02BWF Example Program Data

'M'	'W'	3	3
0.1300	1.3070	0.3700	
9.1231	3.7011	4.5230	
0.9310	0.0900	0.8870	
0.0009	0.0099	0.0999	

## 10.3 Program Results

G02BWF Example Program Results

Correlation matrix

	1	2	3
1	1.0000	0.9908	0.9903
2		1.0000	0.9624
3			1.0000

---