

NAG Library Routine Document

G04DBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G04DBF computes simultaneous confidence intervals for the differences between means. It is intended for use after G04BBF or G04BCF.

2 Specification

```
SUBROUTINE G04DBF (TYP, NT, TMEAN, RDF, C, LDC, CLEVEL, CIL, CIU, ISIG,      &
                  IFAIL)
INTEGER          NT, LDC, ISIG(NT*(NT-1)/2), IFAIL
REAL (KIND=nag_wp) TMEAN(NT), RDF, C(LDC,NT), CLEVEL,      &
                  CIL(NT*(NT-1)/2), CIU(NT*(NT-1)/2)
CHARACTER(1)     TYP
```

3 Description

In the computation of analysis of a designed experiment the first stage is to compute the basic analysis of variance table, the estimate of the error variance (the residual or error mean square), $\hat{\sigma}^2$, the residual degrees of freedom, ν , and the (variance ratio) F -statistic for the t treatments. The second stage of the analysis is to compare the treatment means. If the treatments have no structure, for example the treatments are different varieties, rather than being structured, for example a set of different temperatures, then a multiple comparison procedure can be used.

A multiple comparison procedure looks at all possible pairs of means and either computes confidence intervals for the difference in means or performs a suitable test on the difference. If there are t treatments then there are $t(t-1)/2$ comparisons to be considered. In tests the type 1 error or significance level is the probability that the result is considered to be significant when there is no difference in the means. If the usual t -test is used with, say, a 6% significance level then the type 1 error for all $k = t(t-1)/2$ tests will be much higher. If the tests were independent then if each test is carried out at the 100α percent level then the overall type 1 error would be $\alpha^* = 1 - (1 - \alpha)^k \simeq k\alpha$. In order to provide an overall protection the individual tests, or confidence intervals, would have to be carried out at a value of α such that α^* is the required significance level, e.g., five percent.

The $100(1 - \alpha)$ percent confidence interval for the difference in two treatment means, $\hat{\tau}_i$ and $\hat{\tau}_j$ is given by

$$(\hat{\tau}_i - \hat{\tau}_j) \pm T_{(\alpha, \nu, t)}^* se(\hat{\tau}_i - \hat{\tau}_j),$$

where $se()$ denotes the standard error of the difference in means and $T_{(\alpha, \nu, t)}^*$ is an appropriate percentage point from a distribution. There are several possible choices for $T_{(\alpha, \nu, t)}^*$. These are:

- $\frac{1}{2}q_{(1-\alpha, \nu, t)}$, the studentized range statistic, see G01FMF. It is the appropriate statistic to compare the largest mean with the smallest mean. This is known as Tukey–Kramer method.
- $t_{(\alpha/k, \nu)}$, this is the Bonferroni method.
- $t_{(\alpha_0, \nu)}$, where $\alpha_0 = 1 - (1 - \alpha)^{1/k}$, this is known as the Dunn–Sidak method.
- $t_{(\alpha, \nu)}$, this is known as Fisher's LSD (least significant difference) method. It should only be used if the overall F -test is significant, the number of treatment comparisons is small and were planned before the analysis.

- (e) $\sqrt{(k-1)F_{1-\alpha, k-1, \nu}}$ where $F_{1-\alpha, k-1, \nu}$ is the deviate corresponding to a lower tail probability of $1 - \alpha$ from an F -distribution with $k - 1$ and ν degrees of freedom. This is Scheffe's method.

In cases (b), (c) and (d), $t_{(\alpha, \nu)}$ denotes the α two tail significance level for the Student's t -distribution with ν degrees of freedom, see G01FBB.

The Scheffe method is the most conservative, followed closely by the Dunn–Sidak and Tukey–Kramer methods.

To compute a test for the difference between two means the statistic,

$$\frac{\hat{\tau}_i - \hat{\tau}_j}{se(\hat{\tau}_i - \hat{\tau}_j)}$$

is compared with the appropriate value of $T_{(\alpha, \nu, t)}^*$.

4 References

Kotz S and Johnson N L (ed.) (1985a) Multiple range and associated test procedures *Encyclopedia of Statistical Sciences* **5** Wiley, New York

Kotz S and Johnson N L (ed.) (1985b) Multiple comparison *Encyclopedia of Statistical Sciences* **5** Wiley, New York

Winer B J (1970) *Statistical Principles in Experimental Design* McGraw–Hill

5 Arguments

- 1: TYP – CHARACTER(1) Input
On entry: indicates which method is to be used.
TYP = 'T'
The Tukey–Kramer method is used.
TYP = 'B'
The Bonferroni method is used.
TYP = 'D'
The Dunn–Sidak method is used.
TYP = 'L'
The Fisher LSD method is used.
TYP = 'S'
The Scheffé's method is used.
Constraint: TYP = 'T', 'B', 'D', 'L' or 'S'.
- 2: NT – INTEGER Input
On entry: t , the number of treatment means.
Constraint: $NT \geq 2$.
- 3: TMEAN(NT) – REAL (KIND=nag_wp) array Input
On entry: the treatment means, $\hat{\tau}_i$, for $i = 1, 2, \dots, t$.
- 4: RDF – REAL (KIND=nag_wp) Input
On entry: ν , the residual degrees of freedom.
Constraint: $RDF \geq 1.0$.

- 5: C(LDC, NT) – REAL (KIND=nag_wp) array *Input*
On entry: the strictly lower triangular part of C must contain the standard errors of the differences between the means as returned by G04BBF and G04BCF. That is $C(i, j)$, $i > j$, contains the standard error of the difference between the i th and j th mean in TMEAN.
Constraint: $C(i, j) > 0.0$, for $i = 2, 3, \dots, t$ and $j = 1, 2, \dots, i - 1$.
- 6: LDC – INTEGER *Input*
On entry: the first dimension of the array C as declared in the (sub)program from which G04DBF is called.
Constraint: $LDC \geq NT$.
- 7: CLEVEL – REAL (KIND=nag_wp) *Input*
On entry: the required confidence level for the computed intervals, $(1 - \alpha)$.
Constraint: $0.0 < CLEVEL < 1.0$.
- 8: CIL($NT \times (NT - 1)/2$) – REAL (KIND=nag_wp) array *Output*
On exit: the $((i - 1)(i - 2)/2 + j)$ th element contains the lower limit to the confidence interval for the difference between i th and j th means in TMEAN, for $i = 2, 3, \dots, t$ and $j = 1, 2, \dots, i - 1$.
- 9: CIU($NT \times (NT - 1)/2$) – REAL (KIND=nag_wp) array *Output*
On exit: the $((i - 1)(i - 2)/2 + j)$ th element contains the upper limit to the confidence interval for the difference between i th and j th means in TMEAN, for $i = 2, 3, \dots, t$ and $j = 1, 2, \dots, i - 1$.
- 10: ISIG($NT \times (NT - 1)/2$) – INTEGER array *Output*
On exit: the $((i - 1)(i - 2)/2 + j)$ th element indicates if the difference between i th and j th means in TMEAN is significant, for $i = 2, 3, \dots, t$ and $j = 1, 2, \dots, i - 1$. If the difference is significant then the returned value is 1; otherwise the returned value is 0.
- 11: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NT < 2,
or LDC < NT,
or RDF < 1.0,

or $\text{CLEVEL} \leq 0.0$,
 or $\text{CLEVEL} \geq 1.0$,
 or $\text{TYP} \neq \text{'T'}$, 'B' , 'D' , 'L' or 'S' .

IFAIL = 2

On entry, $C(i, j) \leq 0.0$ for some i, j , $i = 2, 3, \dots, t$ and $j = 1, 2, \dots, i - 1$.

IFAIL = 3

There has been a failure in the computation of the studentized range statistic. This is an unlikely error. Try using a small value of CLEVEL.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

For the accuracy of the percentage point statistics see G01FBF and G01FMF.

8 Parallelism and Performance

G04DBF is not threaded in any implementation.

9 Further Comments

If the treatments have a structure then the use of linear contrasts as computed by G04DAF may be more appropriate.

An alternative approach to one used in G04DBF is the sequential testing of the Student–Newman–Keuls procedure. This, in effect, uses the Tukey–Kramer method but first ordering the treatment means and examining only subsets of the treatment means in which the largest and smallest are significantly different. At each stage the third argument of the Studentized range statistic is the number of means in the subset rather than the total number of means.

10 Example

In the example taken from Winer (1970) a completely randomized design with unequal treatment replication is analysed using G04BBF and then confidence intervals are computed by G04DBF using the Tukey–Kramer method.

10.1 Program Text

Program g04dbfe

```
!      G04DBF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
      Use nag_library, Only: g04bbf, g04dbf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: clevel, gmean, rdf, tol
      Integer                     :: i, iblock, ifail, ij, irdf, j, ldc, &
                                   lit, n, nt
      Character (1)               :: typ
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: bmean(:), c(:, :), cil(:), ciu(:), &
                                   ef(:), r(:), tmean(:), wk(:), y(:)
      Real (Kind=nag_wp)          :: table(4,5)
      Integer, Allocatable         :: irep(:), isig(:), it(:)
      Character (1)               :: star(2)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: abs
!      .. Executable Statements ..
      Write (nout,*) 'G04DBF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) n, nt, iblock

      ldc = nt
      If (nt>1) Then
         lit = n
      Else
         lit = 1
      End If
      Allocate (y(n),bmean(abs(iblock)),tmean(nt),irep(nt),c(ldc,nt),r(n), &
               ef(nt),wk(3*nt),it(lit),cil(nt*(nt-1)/2),ciu(nt*(nt-1)/2),isig(nt*(nt- &
               1)/2))

!      Read in the data and plot information
      Read (nin,*) y(1:n)
      If (nt>1) Then
         Read (nin,*) it(1:n)
      End If

!      Read in the type of level for the CIs
      Read (nin,*) typ, clevel

!      Use default tolerance
      tol = 0.0E0_nag_wp

!      Use standard degrees of freedom
      irdf = 0

!      Calculate the ANOVA table
      ifail = 0
      Call g04bbf(n,y,iblock,nt,it,gmean,bmean,tmean,table,4,c,ldc,irep,r,ef, &
                 tol,irdf,wk,ifail)

!      Display results from G04BBF
      Write (nout,*) ' ANOVA table'
      Write (nout,*)
      Write (nout,*) '   Source          df          SS          MS          F', &
```

```

      '          Prob'
Write (nout,*)
If (iblock>1) Then
  Write (nout,99998) ' Blocks      ', table(1,1:5)
End If
Write (nout,99998) ' Treatments', table(2,1:5)
Write (nout,99998) ' Residual  ', table(3,1:3)
Write (nout,99998) ' Total     ', table(4,1:2)
Write (nout,*)
Write (nout,*) ' Treatment means'
Write (nout,*)
Write (nout,99999) tmean(1:nt)
Write (nout,*)

!      Extract the residual degrees of freedom
rdf = table(3,1)

!      Calculate simultaneous CIs
ifail = 0
Call g04dbf(typ,nt,tmean,rdf,c,ldc,clevel,cil,ciu,isig,ifail)

!      Display results from G04DBF
Write (nout,*) ' Simultaneous Confidence Intervals'
Write (nout,*)
star(2) = '*'
star(1) = ' '
ij = 0
Do i = 1, nt
  Do j = 1, i - 1
    ij = ij + 1
    Write (nout,99997) i, j, cil(ij), ciu(ij), star(isig(ij)+1)
  End Do
End Do

99999 Format (10F8.3)
99998 Format (A,3X,F3.0,2X,2(F10.1,2X),F10.3,2X,F9.4)
99997 Format (2X,2I2,3X,2(F10.3,3X),A)
End Program g04dbfe

```

10.2 Program Data

G04DBF Example Program Data

```

26 4 1                                :: N, NT, IBLOCK (G04BBF)
 3 2 4 3 1 5
 7 8 4 10 6
 3 2 1 2 4 2 3 1
10 12 8 5 12 10 9                    :: End of Y (G04BBF)
1 1 1 1 1 1
2 2 2 2 2
3 3 3 3 3 3 3
4 4 4 4 4 4 4                        :: End of IT (G04BBF)
'T' .95                             :: TYP, CLEVEL

```

10.3 Program Results

G04DBF Example Program Results

ANOVA table

Source	df	SS	MS	F	Prob
Treatments	3.	239.9	80.0	24.029	0.0000
Residual	22.	73.2	3.3		
Total	25.	313.1			

Treatment means

3.000	7.000	2.250	9.429
-------	-------	-------	-------

Simultaneous Confidence Intervals

2	1	0.933	7.067	*
3	1	-3.486	1.986	
3	2	-7.638	-1.862	*
4	1	3.610	9.247	*
4	2	-0.538	5.395	
4	3	4.557	9.800	*
