

NAG Library Routine Document

G05SPF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SPF generates a vector of pseudorandom numbers from a triangular distribution with parameters x_{\min} , x_{med} and x_{\max} .

2 Specification

```
SUBROUTINE G05SPF (N, XMIN, XMED, XMAX, STATE, X, IFAIL)
  INTEGER                N, STATE(*), IFAIL
  REAL (KIND=nag_wp) XMIN, XMED, XMAX, X(N)
```

3 Description

The triangular distribution has a PDF (probability density function) that is triangular in profile. The base of the triangle ranges from $x = x_{\min}$ to $x = x_{\max}$ and the PDF has a maximum value of $\frac{2}{x_{\max} - x_{\min}}$ at $x = x_{\text{med}}$. If $x_{\min} = x_{\text{med}} = x_{\max}$ then $x = x_{\text{med}}$ with probability 1; otherwise the triangular distribution has PDF:

$$f(x) = \frac{x - x_{\min}}{x_{\text{med}} - x_{\min}} \times \frac{2}{x_{\max} - x_{\min}} \quad \text{if } x_{\min} \leq x \leq x_{\text{med}},$$

$$f(x) = \frac{x_{\max} - x}{x_{\max} - x_{\text{med}}} \times \frac{2}{x_{\max} - x_{\min}} \quad \text{if } x_{\text{med}} < x \leq x_{\max},$$

$$f(x) = 0 \quad \text{otherwise.}$$

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SPF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: $N \geq 0$.
- 2: XMIN – REAL (KIND=nag_wp) *Input*
On entry: the end point x_{\min} of the triangular distribution.

- 3: XMED – REAL (KIND=nag_wp) *Input*
On entry: the median of the distribution x_{med} (also the location of the vertex of the triangular distribution at which the PDF reaches a maximum).
Constraint: $\text{XMED} \geq \text{XMIN}$.
- 4: XMAX – REAL (KIND=nag_wp) *Input*
On entry: the end point x_{max} of the triangular distribution.
Constraint: $\text{XMAX} \geq \text{XMED}$.
- 5: STATE(*) – INTEGER array *Communication Array*
Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 6: X(N) – REAL (KIND=nag_wp) array *Output*
On exit: the n pseudorandom numbers from the specified triangular distribution.
- 7: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N = \langle \text{value} \rangle$.
Constraint: $N \geq 0$.

IFAIL = 3

On entry, $\text{XMED} = \langle \text{value} \rangle$ and $\text{XMIN} = \langle \text{value} \rangle$.
Constraint: $\text{XMED} \geq \text{XMIN}$.

IFAIL = 4

On entry, $\text{XMAX} = \langle \text{value} \rangle$ and $\text{XMED} = \langle \text{value} \rangle$.
Constraint: $\text{XMAX} \geq \text{XMED}$.

IFAIL = 5

On entry, STATE vector has been corrupted or not initialized.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05SPF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example prints five pseudorandom numbers from a triangular distribution with parameters $x_{\min} = -1.0$, $x_{\text{med}} = 0.5$ and $x_{\max} = 1.0$, generated by a single call to G05SPF, after initialization by G05KFF.

10.1 Program Text

```

Program g05spfe

!      G05SPF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g05kff, g05spf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: xmax, xmed, xmin
      Integer                     :: genid, ifail, lstate, n, subid
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: x(:)
      Integer                       :: seed(lseed)
      Integer, Allocatable          :: state(:)
!      .. Executable Statements ..
      Write (nout,*) 'G05SPF Example Program Results'
      Write (nout,*)

```

```

!      Skip heading in data file
      Read (nin,*)

!      Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!      Initial call to initializer to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size
      Read (nin,*) n

      Allocate (x(n))

!      Read in the distribution parameters
      Read (nin,*) xmin, xmed, xmax

!      Generate the variates
      ifail = 0
      Call g05spf(n,xmin,xmed,xmax,state,x,ifail)

!      Display the variates
      Write (nout,99999) x(1:n)

99999 Format (1X,F10.4)
      End Program g05spfe

```

10.2 Program Data

G05SPF Example Program Data

```

1 1 1762543      :: GENID,SUBID,SEED(1)
5 3              :: N,NMIX
-1.0 0.5 1.0     :: XMIN,XMED,XMAX

```

10.3 Program Results

G05SPF Example Program Results

```

0.3817
-0.4348
0.4960
0.5509
-0.4398

```
