

# NAG Library Routine Document

## G05SSF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G05SSF generates a vector of pseudorandom numbers from a two parameter Weibull distribution with shape parameter  $a$  and scale parameter  $b$ .

### 2 Specification

```
SUBROUTINE G05SSF (N, A, B, STATE, X, IFAIL)
  INTEGER          N, STATE(*), IFAIL
  REAL (KIND=nag_wp) A, B, X(N)
```

### 3 Description

The distribution has PDF (probability density function)

$$f(x) = \frac{a}{b} x^{a-1} e^{-x^a/b} \quad \text{if } x > 0,$$

$$f(x) = 0 \quad \text{otherwise.}$$

G05SSF returns the value  $(-b \ln y)^{1/a}$ , where  $y$  is a pseudorandom number from a uniform distribution over  $(0, 1]$ .

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SSF.

### 4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin  
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

### 5 Arguments

- |    |  |              |
|----|--|--------------|
| 1: | N – INTEGER  | <i>Input</i> |
|    | <i>On entry:</i> $n$ , the number of pseudorandom numbers to be generated. |              |
|    | <i>Constraint:</i> $N \geq 0$ .  |              |
| 2: | A – REAL (KIND=nag_wp)   | <i>Input</i> |
|    | <i>On entry:</i> $a$ , the shape parameter of the distribution.            |              |
|    | <i>Constraint:</i> $A > 0.0$ .   |              |
| 3: | B – REAL (KIND=nag_wp)   | <i>Input</i> |
|    | <i>On entry:</i> $b$ , the scale parameter of the distribution.            |              |
|    | <i>Constraint:</i> $B > 0.0$ .   |              |

- 4: STATE(\*) – INTEGER array *Communication Array*  
**Note:** the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.  
*On entry:* contains information on the selected base generator and its current state.  
*On exit:* contains updated information on the state of the generator.
- 5: X(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the  $n$  pseudorandom numbers from the specified Weibull distribution.
- 6: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N = \langle value \rangle$ .  
 Constraint:  $N \geq 0$ .

IFAIL = 2

On entry,  $A = \langle value \rangle$ .  
 Constraint:  $A > 0.0$ .

IFAIL = 3

On entry,  $B = \langle value \rangle$ .  
 Constraint:  $B > 0.0$ .

IFAIL = 4

On entry, STATE vector has been corrupted or not initialized.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.  
 See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
 See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

G05SSF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

None.

## 10 Example

This example prints the first five pseudorandom numbers from a Weibull distribution with shape parameter 1.0 and scale parameter 2.0, generated by a single call to G05SSF, after initialization by G05KFF.

### 10.1 Program Text

```

Program g05ssf

!      G05SSF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g05kff, g05ssf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: a, b
      Integer                     :: genid, ifail, lstate, n, subid
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: x(:)
      Integer                       :: seed(lseed)
      Integer, Allocatable          :: state(:)
!      .. Executable Statements ..
      Write (nout,*) 'G05SSF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!      Initial call to initializer to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

```

```
!      Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size
      Read (nin,*) n

      Allocate (x(n))

!      Read in the distribution parameters
      Read (nin,*) a, b

!      Generate the variates
      ifail = 0
      Call g05ssf(n,a,b,state,x,ifail)

!      Display the variates
      Write (nout,99999) x(1:n)

99999 Format (1X,F10.4)
      End Program g05ssfe
```

## 10.2 Program Data

```
G05SSF Example Program Data
1 1 1762543      :: GENID,SUBID,SEED(1)
5 3              :: N,NMIX
1.0 2.0          :: A,B
```

## 10.3 Program Results

```
G05SSF Example Program Results

0.9039
4.4796
0.5860
0.4506
4.5154
```

---