

NAG Library Routine Document

G08AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G08AAF performs the Sign test on two related samples of size n .

2 Specification

```
SUBROUTINE G08AAF (X, Y, N, ISGN, N1, P, IFAIL)
  INTEGER          N, ISGN, N1, IFAIL
  REAL (KIND=nag_wp) X(N), Y(N), P
```

3 Description

The Sign test investigates the median difference between pairs of scores from two matched samples of size n , denoted by $\{x_i, y_i\}$, for $i = 1, 2, \dots, n$. The hypothesis under test, H_0 , often called the null hypothesis, is that the medians are the same, and this is to be tested against a one- or two-sided alternative H_1 (see below).

G08AAF computes:

- (a) the test statistic S , which is the number of pairs for which $x_i < y_i$;
- (b) the number n_1 of non-tied pairs ($x_i \neq y_i$);
- (c) the lower tail probability p corresponding to S (adjusted to allow the complement $(1 - p)$ to be used in an upper one tailed or a two tailed test). p is the probability of observing a value $\leq S$ if $S < \frac{1}{2}n_1$, or of observing a value $< S$ if $S > \frac{1}{2}n_1$, given that H_0 is true. If $S = \frac{1}{2}n_1$, p is set to 0.5.

Suppose that a significance test of a chosen size α is to be performed (i.e., α is the probability of rejecting H_0 when H_0 is true; typically α is a small quantity such as 0.05 or 0.01). The returned value of p can be used to perform a significance test on the median difference, against various alternative hypotheses H_1 , as follows

- (i) H_1 : median of $x \neq$ median of y . H_0 is rejected if $2 \times \min(p, 1 - p) < \alpha$.
- (ii) H_1 : median of $x >$ median of y . H_0 is rejected if $p < \alpha$.
- (iii) H_1 : median of $x <$ median of y . H_0 is rejected if $1 - p < \alpha$.

4 References

Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw–Hill

5 Arguments

- 1: X(N) – REAL (KIND=nag_wp) array *Input*
- 2: Y(N) – REAL (KIND=nag_wp) array *Input*
On entry: X(i) and Y(i) must be set to the i th pair of data values, $\{x_i, y_i\}$, for $i = 1, 2, \dots, n$.
- 3: N – INTEGER *Input*
On entry: n , the size of each sample.
Constraint: $N \geq 1$.

- 4: ISGN – INTEGER *Output*
On exit: the Sign test statistic, S .
- 5: N1 – INTEGER *Output*
On exit: the number of non-tied pairs, n_1 .
- 6: P – REAL (KIND=nag_wp) *Output*
On exit: the lower tail probability, p , corresponding to S .
- 7: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 1$.

IFAIL = 2

$N1 = 0$, i.e., the samples are identical.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The tail probability, p , is computed using the relationship between the binomial and beta distributions. For $n_1 < 120$, p should be accurate to at least 4 significant figures, assuming that the machine has a precision of 7 or more digits. For $n_1 \geq 120$, p should be computed with an absolute error of less than 0.005 . For further details see G01EEF.

8 Parallelism and Performance

G08AAF is not threaded in any implementation.

9 Further Comments

The time taken by G08AAF is small, and increases with n .

10 Example

This example is taken from page 69 of Siegel (1956). The data relates to ratings of ‘insight into paternal discipline’ for 17 sets of parents, recorded on a scale from 1 to 5.

10.1 Program Text

```

Program g08aafe

!      G08AAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g08aaf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: p
      Integer                      :: ifail, isgn, n, n1
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: x(:), y(:)
!      .. Executable Statements ..
      Write (nout,*) 'G08AAF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in problem size
      Read (nin,*) n

      Allocate (x(n),y(n))

!      Read in data
      Read (nin,*) x(1:n)
      Read (nin,*) y(1:n)

!      Display title
      Write (nout,*) 'Sign test'
      Write (nout,*)

!      Display input data
      Write (nout,*) 'Data values'
      Write (nout,*)
      Write (nout,99999) x(1:n)
      Write (nout,99999) y(1:n)

!      Perform the sign test
      ifail = 0
      Call g08aaf(x,y,n,isgn,n1,p,ifail)

!      Display results
      Write (nout,*)
      Write (nout,99998) 'Test statistic    ', isgn
      Write (nout,99998) 'Observations      ', n1
      Write (nout,99997) 'Lower tail prob.', p

```

```

99999 Format (4X,20F3.0)
99998 Format (1X,A,I5)
99997 Format (1X,A,F6.3)
      End Program g08aafe

```

10.2 Program Data

```

G08AAF Example Program Data
17                                :: N
4.0 4.0 5.0 5.0 3.0 2.0 5.0 3.0 1.0
5.0 5.0 5.0 4.0 5.0 5.0 5.0 5.0      :: End of X
2.0 3.0 3.0 3.0 3.0 3.0 3.0 3.0 2.0
3.0 2.0 2.0 5.0 2.0 5.0 3.0 1.0      :: End of Y

```

10.3 Program Results

G08AAF Example Program Results

Sign test

Data values

```

      4. 4. 5. 5. 3. 2. 5. 3. 1. 5. 5. 5. 4. 5. 5. 5. 5.
      2. 3. 3. 3. 3. 3. 3. 3. 2. 3. 2. 2. 5. 2. 5. 3. 1.

```

```

Test statistic      3
Observations       14
Lower tail prob. 0.029

```
