
NAG Library Routine Document

G08AKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

G08AKF calculates the exact tail probability for the Mann–Whitney rank sum test statistic for the case
where there are ties in the two samples pooled together.

2 Specification

SUBROUTINE G08AKF (N1, N2, TAIL, RANKS, U, P, WRK, LWRK, IWRK, IFAIL)

INTEGER N1, N2, LWRK, IWRK(2*(N1+N2+1)), IFAIL
REAL (KIND=nag_wp) RANKS(N1+N2), U, P, WRK(LWRK)
CHARACTER(1) TAIL

3 Description

G08AKF computes the exact tail probability for the Mann–Whitney U test statistic (calculated by
G08AHF and returned through the argument U) using a method based on an algorithm developed by
Neumann (1988), for the case where there are ties in the pooled sample.

The Mann–Whitney U test investigates the difference between two populations defined by the
distribution functions F xð Þ and G yð Þ respectively. The data consist of two independent samples of size
n1 and n2, denoted by x1; x2; . . . ; xn1 and y1; y2; . . . ; yn2 , taken from the two populations.

The hypothesis under test, H0, often called the null hypothesis, is that the two distributions are the
same, that is F xð Þ ¼ G xð Þ, and this is to be tested against an alternative hypothesis H1 which is

H1: F xð Þ 6¼ G yð Þ; or
H1: F xð Þ < G yð Þ, i.e., the x's tend to be greater than the y's; or

H1: F xð Þ > G yð Þ, i.e., the x's tend to be less than the y's,

using a two tailed, upper tailed or lower tailed probability respectively. You select the alternative
hypothesis by choosing the appropriate tail probability to be computed (see the description of argument
TAIL in Section 5).

Note that when using this test to test for differences in the distributions one is primarily detecting
differences in the location of the two distributions. That is to say, if we reject the null hypothesis H0 in
favour of the alternative hypothesis H1: F xð Þ > G yð Þ we have evidence to suggest that the location, of
the distribution defined by F xð Þ, is less than the location of the distribution defined by G yð Þ.
G08AKF returns the exact tail probability, p, corresponding to U , depending on the choice of alternative
hypothesis, H1.

The value of p can be used to perform a significance test on the null hypothesis H0 against the
alternative hypothesis H1. Let � be the size of the significance test (that is � is the probability of
rejecting H0 when H0 is true). If p < � then the null hypothesis is rejected. Typically � might be 0:05
or 0:01.

4 References

5 Arguments

1: N1 – INTEGER Input

On entry: the number of non-tied pairs, n1.

Constraint: N1 � 1.

2: N2 – INTEGER Input

On entry: the size of the second sample, n2.

Constraint: N2 � 1.

3: TAIL – CHARACTER(1) Input

On entry: indicates the choice of tail probability, and hence the alternative hypothesis.

TAIL ¼ T
A two tailed probability is calculated and the alternative hypothesis is H1 : F xð Þ 6¼ G yð Þ.

TAIL ¼ U
An upper tailed probability is calculated and the alternative hypothesis H1 : F xð Þ < G yð Þ,
i.e., the x's tend to be greater than the y's.

TAIL ¼ L
A lower tailed probability is calculated and the alternative hypothesis H1 : F xð Þ > G yð Þ, i.
e., the x's tend to be less than the y's.

Constraint: TAIL ¼ T , U or L .

4: RANKSðN1þ N2Þ – REAL (KIND=nag_wp) array Input

On entry: the ranks of the pooled sample. These ranks are output in the array RANKS by
G08AHF and should not be altered in any way if you are using the same n1, n2 and U as used in
G08AHF.

5: U – REAL (KIND=nag_wp) Input

On entry: U , the value of the Mann–Whitney rank sum test statistic. This is the statistic returned
through the argument U by G08AHF.

Constraint: U � 0:0.

6: P – REAL (KIND=nag_wp) Output

On exit: the tail probability, p, as specified by the argument TAIL.

7: WRKðLWRKÞ – REAL (KIND=nag_wp) array Workspace
8: LWRK – INTEGER Input

On entry: the dimension of the array WRK as declared in the (sub)program from which G08AKF
is called.

Constraint: LWRK � n þ n n þ 1ð Þ n þmð Þ � n n þ 1ð Þ 2� n þ 1ð Þ
3

þ 1 , where n ¼ min N1;N2ð Þ
and m ¼ max N1;N2ð Þ.

9: IWRKð2� N1þ N2þ 1ð ÞÞ – INTEGER array Workspace

10: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N1 < 1,
or N2 < 1.

IFAIL ¼ 2

On entry, TAIL 6¼ T , U or L .

IFAIL ¼ 3

On entry, U < 0:0.

IFAIL ¼ 4

On entry, LWRK is too small.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The exact tail probability, p, is computed to an accuracy of at least 4 significant figures.

8 Parallelism and Performance

G08AKF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The time taken by G08AKF increases with n1 and n2 and the product n1n2. Note that the amount of
workspace required becomes very large for even moderate sizes of n1 and n2.

10 Example

This example finds the Mann–Whitney test statistic, using G08AHF for two independent samples of
size 16 and 23 respectively. This is used to test the null hypothesis that the distributions of the two
populations from which the samples were taken are the same against the alternative hypothesis that the
distributions are different. The test statistic, the approximate Normal statistic and the approximate two
tail probability are printed. G08AKF is then called to obtain the exact two tailed probability. The exact
probability is also printed.

10.1 Program Text

Program g08akfe

! G08AKF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: g08ahf, g08akf, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: p, pexact, u, unor
Integer :: ifail, liwrk, lwrk, lwrk1, lwrk2, &

mn, n1, n2, nsum
Logical :: ties
Character (1) :: tail

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ranks(:), wrk(:), x(:), y(:)
Integer, Allocatable :: iwrk(:)

! .. Intrinsic Procedures ..
Intrinsic :: max, min

! .. Executable Statements ..
Write (nout,*) ’G08AKF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)

! Read in problem size
Read (nin,*) n1, n2, tail

! Calculate sizes of various workspaces
nsum = n1 + n2
mn = min(n1,n2)

! Workspace for G08AHF
lwrk1 = nsum

! Workspace for G08AKF

lwrk2 = mn + mn*(mn+1)*nsum - mn*(mn+1)*(2*mn+1)/3 + 1
liwrk = 2*nsum + 2

lwrk = max(lwrk1,lwrk2)
Allocate (x(n1),y(n2),ranks(nsum),wrk(lwrk),iwrk(liwrk))

! Read in data
Read (nin,*) x(1:n1)
Read (nin,*) y(1:n2)

! Display title
Write (nout,*) ’Mann-Whitney U test’
Write (nout,*)

! Display input data
Write (nout,99999) ’Sample size of group 1 = ’, n1
Write (nout,99999) ’Sample size of group 2 = ’, n2
Write (nout,*)
Write (nout,*) ’Data values’
Write (nout,*)
Write (nout,99998) ’ Group 1 ’, x(1:n1)
Write (nout,*)
Write (nout,99998) ’ Group 2 ’, y(1:n2)

! Perform test
ifail = 0
Call g08ahf(n1,x,n2,y,tail,u,unor,p,ties,ranks,wrk,ifail)

! Calculate exact probabilities
If (ties) Then

ifail = 0
Call g08akf(n1,n2,tail,ranks,u,pexact,wrk,lwrk,iwrk,ifail)

End If

! Display results
Write (nout,*)
Write (nout,99997) ’Test statistic = ’, u
Write (nout,99997) ’Normal statistic = ’, unor
Write (nout,99997) ’Tail probability = ’, p
Write (nout,*)
If (ties) Then

Write (nout,*)
Write (nout,*) ’Ranks’
Write (nout,*)
Write (nout,99998) ’ Group 1 ’, ranks(1:n1)
Write (nout,*)
Write (nout,99998) ’ Group 2 ’, ranks((n1+1):nsum)
Write (nout,*)
Write (nout,*)
Write (nout,99997) ’Exact tail probability = ’, pexact

Else
Write (nout,*) &

’There are no ties in the pooled sample so G08AKF was not called.’
End If

99999 Format (1X,A,I5)
99998 Format (1X,A,8F5.1,2(/,14X,8F5.1))
99997 Format (1X,A,F10.4)

End Program g08akfe

10.2 Program Data

G08AKF Example Program Data
16 23 ’L’ :: N1,N2,TAIL
13.0 6.0 12.0 7.0 12.0 7.0 10.0 7.0
10.0 7.0 16.0 7.0 10.0 8.0 9.0 8.0 :: End of X
17.0 6.0 10.0 8.0 15.0 8.0 15.0 10.0
15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
13.0 12.0 13.0 12.0 13.0 12.0 12.0 :: End of Y

10.3 Program Results

G08AKF Example Program Results

Mann-Whitney U test

Sample size of group 1 = 16
Sample size of group 2 = 23

Data values

Group 1 13.0 6.0 12.0 7.0 12.0 7.0 10.0 7.0
10.0 7.0 16.0 7.0 10.0 8.0 9.0 8.0

Group 2 17.0 6.0 10.0 8.0 15.0 8.0 15.0 10.0
15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
13.0 12.0 13.0 12.0 13.0 12.0 12.0

Test statistic = 86.0000
Normal statistic = -2.8039
Tail probability = 0.0025

Ranks

Group 1 29.5 1.5 24.5 5.0 24.5 5.0 16.0 5.0
16.0 5.0 38.0 5.0 16.0 9.5 12.0 9.5

Group 2 39.0 1.5 16.0 9.5 36.0 9.5 36.0 16.0
36.0 16.0 33.0 16.0 33.0 20.5 33.0 20.5
29.5 24.5 29.5 24.5 29.5 24.5 24.5

Exact tail probability = 0.0020

