NAG Library Routine Document

G08CKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G08CKF calculates the Anderson–Darling goodness-of-fit test statistic and its probability for the case of a fully-unspecified Normal distribution.

2 Specification

```
SUBROUTINE G08CKF (N, ISSORT, Y, YBAR, YVAR, A2, AA2, P, IFAIL)
INTEGER            N, IFAIL
REAL (KIND=nag_wp) Y(N), YBAR, YVAR, A2, AA2, P
LOGICAL            ISSORT
```

3 Description

Calculates the Anderson–Darling test statistic A^2 (see G08CHF) and its upper tail probability for the small sample correction:

$$\text{Adjusted } A^2 = A^2\left(1 + 0.75/n + 2.25/n^2\right),$$

for n observations.

4 References

Anderson T W and Darling D A (1952) Asymptotic theory of certain 'goodness-of-fit' criteria based on stochastic processes *Annals of Mathematical Statistics* **23** 193–212

Stephens M A and D'Agostino R B (1986) *Goodness-of-Fit Techniques* Marcel Dekker, New York

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of observations.

Constraint: N > 1.

2: ISSORT – LOGICAL *Input*

On entry: set ISSORT = .TRUE. if the observations are sorted in ascending order; otherwise the routine will sort the observations.

3: Y(N) – REAL (KIND=nag_wp) array *Input*

On entry: y_i, for $i = 1, 2, \ldots, n$, the n observations.

Constraint: if ISSORT = .TRUE., the values must be sorted in ascending order.

4: YBAR – REAL (KIND=nag_wp) *Output*

On exit: the maximum likelihood estimate of mean.

5: YVAR – REAL (KIND=nag_wp) *Output*

On exit: the maximum likelihood estimate of variance.

6: A2 – REAL (KIND=nag_wp) *Output*

On exit: A^2, the Anderson–Darling test statistic.

7: AA2 – REAL (KIND=nag_wp) *Output*

On exit: the adjusted A^2.

8: P – REAL (KIND=nag_wp) *Output*

On exit: p, the upper tail probability for the adjusted A^2.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $= \langle value \rangle$.
Constraint: N > 1.

IFAIL $= 3$

ISSORT $=$ .TRUE. and the data in Y is not sorted in ascending order.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Probabilities are calculated using piecewise polynomial approximations to values estimated by simulation.

8 Parallelism and Performance

G08CKF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example calculates the A^2 statistics for data assumed to arise from a fully-unspecified Normal distribution and the p-value.

10.1 Program Text

```
    Program g08ckfe

!     G08CKF Example Program Text
!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g08ckf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: a2, aa2, p, ybar, yvar
      Integer                          :: i, ifail, n
      Logical                          :: issort
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: y(:)
!     .. Executable Statements ..
      Write (nout,*) 'G08CKF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read number of observations
      Read (nin,*) n

!     Memory allocation
      Allocate (y(n))

!     Read observations
      Read (nin,*)(y(i),i=1,n)

!     Let g08ckf sort the data
      issort = .False.

!     Calculate A-squared and probability
      ifail = 0
      Call g08ckf(n,issort,y,ybar,yvar,a2,aa2,p,ifail)

!     Results
      Write (nout,'(1X,A,E11.4,1X,A,E11.4)')                                &
        'H0: data from Normal distribution with mean', ybar, 'and variance',  &
        yvar
      Write (nout,'(1X,A,1X,F8.4)') 'Test statistic, A-squared: ', a2
      Write (nout,'(1X,A,1X,F8.4)') 'Adjusted A-squared:        ', aa2
      Write (nout,'(1X,A,1X,F8.4)') 'Upper tail probability:    ', p

    End Program g08ckfe
```

10.2 Program Data

```
G08CKF Example Program Data
26 :: n
 0.3131132  0.2520412  1.5788841  1.4416712 -0.8246043 -1.6466685
 0.7943184  1.2874915 -0.8347250  0.3352505  0.9434467  2.1099520
-0.2801654 -0.7843009  0.6218187  2.0963809  1.7170403 -0.1350142
 0.7982763 -0.2980977  1.2283043  1.5576090 -0.4828757  2.6070754
 0.1213996  0.1431621 :: end of observations
```

10.3 Program Results

```
 G08CKF Example Program Results

 H0: data from Normal distribution with mean 0.5639E+00 and variance 0.1139E+01
 Test statistic, A-squared:    0.1660
 Adjusted A-squared:           0.1713
 Upper tail probability:       0.9312
```