

NAG Library Routine Document

G08DAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G08DAF calculates Kendall's coefficient of concordance on k independent rankings of n objects or individuals.

2 Specification

```
SUBROUTINE G08DAF (X, LDX, K, N, RNK, W, P, IFAIL)
  INTEGER          LDX, K, N, IFAIL
  REAL (KIND=nag_wp) X(LDX,N), RNK(LDX,N), W, P
```

3 Description

Kendall's coefficient of concordance measures the degree of agreement between k comparisons of n objects, the scores in the i th comparison being denoted by

$$x_{i1}, x_{i2}, \dots, x_{in}.$$

The hypothesis under test, H_0 , often called the null hypothesis, is that there is no agreement between the comparisons, and this is to be tested against the alternative hypothesis, H_1 , that there is some agreement.

The n scores for each comparison are ranked, the rank r_{ij} denoting the rank of object j in comparison i , and all ranks lying between 1 and n . Average ranks are assigned to tied scores.

For each of the n objects, the k ranks are totalled, giving rank sums R_j , for $j = 1, 2, \dots, n$. Under H_0 , all the R_j would be approximately equal to the average rank sum $k(n+1)/2$. The total squared deviation of the R_j from this average value is therefore a measure of the departure from H_0 exhibited by the data. If there were complete agreement between the comparisons, the rank sums R_j would have the values $k, 2k, \dots, nk$ (or some permutation thereof). The total squared deviation of these values is $k^2(n^3 - n)/12$.

Kendall's coefficient of concordance is the ratio

$$W = \frac{\sum_{j=1}^n (R_j - \frac{1}{2}k(n+1))^2}{\frac{1}{12}k^2(n^3 - n)}$$

and lies between 0 and 1, the value 0 indicating complete disagreement, and 1 indicating complete agreement.

If there are tied rankings within comparisons, W is corrected by subtracting $k \sum T$ from the denominator, where $T = \sum (t^3 - t)/12$, each t being the number of occurrences of each tied rank within a comparison, and the summation of T being over all comparisons containing ties.

G08DAF returns the value of W , and also an approximation, p , of the significance of the observed W . (For $n > 7$, $k(n-1)W$ approximately follows a χ^2_{n-1} distribution, so large values of W imply rejection of H_0 .) H_0 is rejected by a test of chosen size α if $p < \alpha$. If $n \leq 7$, tables should be used to establish the significance of W (e.g., Table R of Siegel (1956)).

4 References

Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw–Hill

5 Arguments

- 1: X(LDX,N) – REAL (KIND=nag_wp) array *Input*
On entry: $X(i,j)$ must be set to the value x_{ij} of object j in comparison i , for $i = 1, 2, \dots, k$ and $j = 1, 2, \dots, n$.

- 2: LDX – INTEGER *Input*
On entry: the first dimension of the arrays X and RNK as declared in the (sub)program from which G08DAF is called.
Constraint: $LDX \geq K$.

- 3: K – INTEGER *Input*
On entry: k , the number of comparisons.
Constraint: $K \geq 2$.

- 4: N – INTEGER *Input*
On entry: n , the number of objects.
Constraint: $N \geq 2$.

- 5: RNK(LDX,N) – REAL (KIND=nag_wp) array *Workspace*

- 6: W – REAL (KIND=nag_wp) *Output*
On exit: the value of Kendall's coefficient of concordance, W .

- 7: P – REAL (KIND=nag_wp) *Output*
On exit: the approximate significance, p , of W .

- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 2$.

IFAIL = 2

On entry, LDX < K.

IFAIL = 3

On entry, $K \leq 1$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

All computations are believed to be stable. The statistic W should be accurate enough for all practical uses.

8 Parallelism and Performance

G08DAF is not threaded in any implementation.

9 Further Comments

The time taken by G08DAF is approximately proportional to the product nk .

10 Example

This example is taken from page 234 of Siegel (1956). The data consists of 10 objects ranked on three different variables: X , Y and Z . The computed values of Kendall's coefficient is significant at the 1% level of significance ($p = 0.008 < 0.01$), indicating that the null hypothesis of there being no agreement between the three rankings X , Y , Z may be rejected with reasonably high confidence.

10.1 Program Text

```

Program g08dafa

!      G08DAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g08daf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: p, w
      Integer                     :: i, ifail, k, ldx, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: rnk(:,,:), x(:,,:)

```

```

!      .. Executable Statements ..
      Write (nout,*) 'G08DAF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) k, n

      ldx = k
      Allocate (x(ldx,n),rnk(ldx,n))

!      Read in data
      Read (nin,*)(x(i,1:n),i=1,k)

!      Display title
      Write (nout,*) 'Kendall''s coefficient of concordance'
      Write (nout,*)

!      Display data
      Write (nout,*) 'Data values'
      Write (nout,*)
      Write (nout,99999)('Comparison ',i,' scores ',x(i,1:n),i=1,k)

      ifail = 0
      Call g08daf(x,ldx,k,n,rnk,w,p,ifail)

!      Display results
      Write (nout,*)
      Write (nout,99998) 'Kendall''s coefficient =', w
      Write (nout,99998) '          Significance =', p

99999 Format (1X,A,I1,A,10F5.1)
99998 Format (1X,A,F8.3)
      End Program g08dafa

```

10.2 Program Data

```

G08DAF Example Program Data
3  10                                     : K,N
  1.0  4.5  2.0  4.5  3.0  7.5  6.0  9.0  7.5 10.0
  2.5  1.0  2.5  4.5  4.5  8.0  9.0  6.5 10.0  6.5
  2.0  1.0  4.5  4.5  4.5  4.5  8.0  8.0  8.0 10.0 : End of X

```

10.3 Program Results

G08DAF Example Program Results

Kendall's coefficient of concordance

Data values

Comparison 1 scores	1.0	4.5	2.0	4.5	3.0	7.5	6.0	9.0	7.5	10.0
Comparison 2 scores	2.5	1.0	2.5	4.5	4.5	8.0	9.0	6.5	10.0	6.5
Comparison 3 scores	2.0	1.0	4.5	4.5	4.5	4.5	8.0	8.0	8.0	10.0

Kendall's coefficient = 0.828
 Significance = 0.008
