

NAG Library Routine Document

G13BGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13BGF accepts a series of new observations of an output time series and any associated input time series, for which a multi-input model is already fully specified, and updates the ‘state set’ information for use in constructing further forecasts.

The previous specification of the multi-input model will normally have been obtained by using G13BEF to estimate the relevant transfer function and ARIMA parameters. The supplied state set will originally have been produced by G13BEF (or possibly G13BJF), but may since have been updated by G13BGF.

2 Specification

```
SUBROUTINE G13BGF (STTF, NSTTF, MR, NSER, MT, PARA, NPARA, NNV, XXYN,      &
                  LDXXYN, KZEF, RES, WA, IWA, IFAIL)
INTEGER          NSTTF, MR(7), NSER, MT(4,NSER), NPARA, NNV, LDXXYN,      &
                  KZEF, IWA, IFAIL
REAL (KIND=nag_wp) STTF(NSTTF), PARA(NPARA), XXYN(LDXXYN,NSER),      &
                  RES(NNV), WA(IWA)
```

3 Description

The multi-input model is specified in Section 3 in G13BEF. The form of these equations required to update the state set is as follows:

$$z_t = \delta_1 z_{t-1} + \delta_2 z_{t-2} + \cdots + \delta_p z_{t-p} + \omega_0 x_{t-b} - \omega_1 x_{t-b-1} - \cdots - \omega_q x_{t-b-q}$$

the transfer models which generate input component values $z_{i,t}$ from one or more inputs $x_{i,t}$,

$$n_t = y_t - z_{1,t} - z_{2,t} - \cdots - z_{m,t}$$

which generates the output noise component from the output y_t and the input components, and

$$\begin{aligned} w_t &= \nabla^d \nabla_s^D n_t - c \\ e_t &= w_t - \Phi_1 w_{t-s} - \Phi_2 w_{t-2 \times s} - \cdots - \Phi_P w_{t-P \times s} + \Theta_1 e_{t-s} + \Theta_2 e_{t-2 \times s} + \cdots + \Theta_Q e_{t-Q \times s} \\ a_t &= e_t - \phi_1 e_{t-1} - \phi_2 e_{t-2} - \cdots - \phi_p e_{t-p} + \theta_1 a_{t-1} + \theta_2 a_{t-2} + \cdots + \theta_q a_{t-q} \end{aligned}$$

the ARIMA model for the output noise which generates the residuals a_t .

The state set (as also given in Section 3 in G13BEF) is the collection of terms

$$z_{n+1-k}, x_{n+1-k}, n_{n+1-k}, w_{n+1-k}, e_{n+1-k} \quad \text{and} \quad a_{n+1-k}$$

for $k = 1$ up to the maximum lag associated with each of these series respectively, in the above model equations. n is the latest time point of the series from which the state set has been generated.

The routine accepts further values of the series $y_t, x_{1,t}, x_{2,t}, \dots, x_{m,t}$, for $t = n+1, \dots, n+l$, and applies the above model equations over this time range, to generate new values of the various model components, noise series and residuals. The state set is reconstructed, corresponding to the latest time point $n+l$, the earlier values being discarded.

The set of residuals corresponding to the new observations may be of use in checking that the new observations conform to the previously fitted model. The components of the new observations of the output series which are due to the various inputs, and the noise component, are also optionally returned.

The parameters of the model are not changed in this routine.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

5 Arguments

- 1: STTF(NSTTF) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the NSTTF values in the state set before updating as returned by G13BEF or G13BJF, or a previous call to G13BGF.
On exit: the state set values after updating.
- 2: NSTTF – INTEGER *Input*
On entry: the exact number of values in the state set array STTF as returned by G13BEF or G13BJF.
- 3: MR(7) – INTEGER array *Input*
On entry: the orders vector (p, d, q, P, D, Q, s) of the ARIMA model for the output noise component.
 p, q, P and Q refer respectively to the number of autoregressive (ϕ), moving average (θ), seasonal autoregressive (Φ) and seasonal moving average (Θ) parameters.
 d, D and s refer respectively to the order of non-seasonal differencing, the order of seasonal differencing, and the seasonal period.
Constraints:

$$\begin{aligned} p, d, q, P, D, Q, s &\geq 0; \\ p + q + P + Q &> 0; \\ s &\neq 1; \\ \text{if } s = 0, P + D + Q &= 0; \\ \text{if } s > 1, P + D + Q &> 0. \end{aligned}$$
- 4: NSER – INTEGER *Input*
On entry: the total number of input and output series. There may be any number of input series (including none), but only one output series.
- 5: MT(4,NSER) – INTEGER array *Input*
On entry: the transfer function model orders b, p and q of each of the input series. The data for input series i are held in column i . Row 1 holds the value b_i , row 2 holds the value q_i and row 3 holds the value p_i . For a simple input, $b_i = q_i = p_i = 0$.
Row 4 holds the value r_i , where $r_i = 1$ for a simple input and $r_i = 2$ or 3 for a transfer function input. When $r_i = 1$ any nonzero contents of rows 1, 2 and 3 of column i are ignored. The choice of $r_i = 2$ or $r_i = 3$ is an option for use in model estimation and does not affect the operation of G13BGF.
Constraint: $\text{MT}(4, i) = 1, 2 \text{ or } 3$, for $i = 1, 2, \dots, \text{NSER} - 1$.
- 6: PARA(NPARA) – REAL (KIND=nag_wp) array *Input*
On entry: estimates of the multi-input model parameters as returned by G13BEF. These are in order, firstly the ARIMA model parameters: p values of ϕ parameters, q values of θ parameters, P values of Φ parameters and Q values of Θ parameters. These are followed by the transfer function model parameter values $\omega_0, \omega_1, \dots, \omega_{q_1}, \delta_1, \delta_2, \dots, \delta_{p_1}$ for the first of any input series and similarly for each subsequent input series. The final component of PARA is the value of the constant c .

- 7: NPARA – INTEGER *Input*
On entry: the exact number of ϕ , θ , Φ , Θ , ω , δ and c parameters. (c must be included whether its value was previously estimated or was set fixed.)
- 8: NNV – INTEGER *Input*
On entry: the number of new observation sets being used to update the state set, each observation set consisting of a value of the output series and the associated values of each of the input series at a particular time point.
- 9: XXYN(LDXXYN, NSER) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the NNV new observation sets being used to update the state set. Column i contains the values of input series i , for $i = 1, 2, \dots, \text{NSER} - 1$. Column NSER contains the values of the output series. Consecutive rows correspond to increasing time sequence.
On exit: if $\text{KZEF} = 0$, XXYN remains unchanged.
If $\text{KZEF} \neq 0$, the columns of XXYN hold the corresponding values of the input component series z_t and the output noise component n_t in that order.
- 10: LDXXYN – INTEGER *Input*
On entry: the first dimension of the array XXYN as declared in the (sub)program from which G13BGF is called.
Constraint: $\text{LDXXYN} \geq \text{NNV}$.
- 11: KZEF – INTEGER *Input*
On entry: must not be set to 0, if the values of the input component series z_t and the values of the output noise component n_t are to overwrite the contents of XXYN on exit, and must be set to 0 if XXYN is to remain unchanged on exit.
- 12: RES(NNV) – REAL (KIND=nag_wp) array *Output*
On exit: the values of the residual series a_t corresponding to the new observations of the output series.
- 13: WA(IWA) – REAL (KIND=nag_wp) array *Workspace*
14: IWA – INTEGER *Input*
On entry: the dimension of the array WA as declared in the (sub)program from which G13BGF is called.
Constraint: $\text{IWA} \geq \text{NNV} + 2 \times \text{NSTTF} + \max(\text{NNV}, \text{NSTTF}) + \max(\text{NNV}, ncc)$, where $ncc = 4 \times (p + q + P + Q)$.
- 15: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, NSTTF is not consistent with the orders in arrays MR and MT.

$IFAIL = 2$

On entry, NPARA is not consistent with the orders in arrays MR and MT.

$IFAIL = 3$

On entry, LDXXYN is too small.

$IFAIL = 4$

On entry, IWA is too small.

$IFAIL = 5$

On entry, one of the r_i , stored in $MT(4, i)$, for $i = 1, 2, \dots, NSER - 1$ does not equal 1, 2 or 3.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computations are believed to be stable.

8 Parallelism and Performance

G13BGF is not threaded in any implementation.

9 Further Comments

The time taken by G13BGF is approximately proportional to $NNV \times NPARA$.

10 Example

This example uses the data described in G13BEF in which 40 observations of an output time series and a single input series were processed. In this example a model which included seasonal differencing of order 1 was used. The 10 values of the state set and the 5 final values of PARA then obtained are used as input to this program, together with the values of 4 new observations and the transfer function orders

of the input series. The model used is $\phi_1 = 0.5158$, $\Theta_1 = 0.9994$, $\omega_0 = 8.6343$, $\delta_1 = 0.6726$, $c = -0.3172$.

The following are computed and printed out: the updated state set, the residuals a_t and the values of the components z_t and the output noise component n_t corresponding to the new observations.

10.1 Program Text

```

Program gl3bgfe

!      G13BGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: gl3bgf, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, iwa, kzef, ldxxyn, ncc,      &
                                   nis, nnv, npara, nser, nsttf
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: para(:), res(:), sttf(:), wa(:),      &
                                   xxyn(:, :)
      Integer                         :: mr(7)
      Integer, Allocatable            :: mt(:, :)
!      .. Intrinsic Procedures ..
      Intrinsic                      :: max
!      .. Executable Statements ..
      Write (nout,*) 'G13BGF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) nsttf, nser, nnv, kzef

!      Number of input series
      nis = nser - 1

!      Read in orders
      Read (nin,*) mr(1:7)

      ldxxyn = nnv
      Allocate (mt(4,nser))

!      Read in transfer function
      Do i = 1, nis
         Read (nin,*) mt(1:4,i)
      End Do

!      Calculate NPARA
      npara = 0
      Do i = 1, nis
         npara = npara + mt(2,i) + mt(3,i)
      End Do
      npara = npara + mr(1) + mr(3) + mr(4) + mr(6) + nser

      ldxxyn = nnv
      ncc = 4*(mr(1)+mr(3)+mr(4)+mr(6))
      iwa = nnv + 2*nsttf + max(nnv,nsttf) + max(nnv,ncc)
      Allocate (sttf(nsttf),xxyn(ldxxyn,nser),res(nnv),para(npara),wa(iwa))

!      Read in state set
      Read (nin,*) sttf(1:nsttf)

!      Read in parameters

```

```

      Read (nin,*) para(1:npara)

!      Read in new observations
      Read (nin,*)(xxyn(i,1:nser),i=1,nnv)

      ifail = 0
      Call gl3bgf(sttf,nsttf,mr,nser,mt,para,npara,nnv,xxyn,ldxxyn,kzef,res, &
        wa,iwa,ifail)

!      Display results
      Write (nout,*) 'The updated state set'
      Write (nout,99999) sttf(1:nsttf)
      Write (nout,*)
      Write (nout,*) 'The residuals (after differencing)'
      Write (nout,99998)(i,res(i),i=1,nnv)
      Write (nout,*)
      Flush (nout)
      ifail = 0
      Call x04caf('General',' ',nnv,nser,xxyn,ldxxyn, &
        'The values of z(t) and n(t)',ifail)
      Write (nout,99997) 'The first ', nis, &
        ' columns hold the z(t) and the last column the n(t)'

99999 Format (1X,6F10.4)
99998 Format (1X,I4,F10.4)
99997 Format (1X,A,I0,A)
      End Program gl3bgfe

```

10.2 Program Data

G13BGF Example Program Data

```

10  2  4  1      :: NSTTF,NSER,NNV,KZEF
  1  0  0  0  1  1  4      :: MR
  1  0  1  3      :: Transfer fun. for series 1, MT(:,1)
    6.0530 184.4749 -80.0885 -75.1704
-76.9481 -81.4749  0.7776 -2.6190
-2.3054 -1.1963
    0.5158  0.9994  8.6343  0.6726
-0.3172      :: STTF
    5.9410  96.0000
    5.3860  95.0000
    5.8110  80.0000
    6.7160  88.0000      :: PARA
                                :: End of XXYN

```

10.3 Program Results

G13BGF Example Program Results

The updated state set

6.7160	158.3155	-80.3412	-74.9035	-80.7814	-70.3155
0.8416	-2.0333	-5.8201	10.2810		

The residuals (after differencing)

1	1.4586
2	-2.4674
3	-4.7714
4	13.2830

The values of $z(t)$ and $n(t)$

	1	2
1	176.3412	-80.3412
2	169.9035	-74.9035
3	160.7814	-80.7814
4	158.3155	-70.3155

The first 1 columns hold the $z(t)$ and the last column the $n(t)$