NAG Library Routine Document

G13BHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13BHF produces forecasts of a time series (the output series) which depends on one or more other (input) series via a multi-input model which will usually have been fitted using G13BEF. The future values of the input series must be supplied. The original observations are not required. G13BHF uses as input either the original state set obtained from G13BEF, or the state set updated by a series of new observations from G13BGF. Standard errors of the forecasts are produced. If future values of some of the input series have been obtained as forecasts using ARIMA models for those series, this may be allowed for in the calculation of the standard errors.

2 Specification

```
SUBROUTINE G13BHF (STTF, NSTTF, MR, NSER, MT, PARA, NPARA, NFV, XXYN,      &
                   LDXXYN, MRX, PARX, LDPARX, RMSXY, KZEF, FVA, FSD, WA,   &
                   IWA, IFAIL)
INTEGER            NSTTF, MR(7), NSER, MT(4,NSER), NPARA, NFV, LDXXYN,     &
                   MRX(7,NSER), LDPARX, KZEF, IWA, IFAIL
REAL (KIND=nag_wp) STTF(NSTTF), PARA(NPARA), XXYN(LDXXYN,NSER),            &
                   PARX(LDPARX,NSER), RMSXY(NSER), FVA(NFV), FSD(NFV),     &
                   WA(IWA)
```

3 Description

The forecasts of the output series y_t are calculated, for $t = n+1, \ldots, n+L$, where n is the latest time point of the observations used to produce the state set and L is the maximum lead time of the forecasts.

First the new input series values x_t are used to form the input components z_t, for $t = n+1, \ldots, n+L$, using the transfer function models:

(a) $z_t = \delta_1 z_{t-1} + \delta_2 z_{t-2} + \cdots + \delta_p z_{t-p} + \omega_0 x_{t-b} - \omega_1 x_{t-b-1} - \cdots - \omega_q x_{t-b-q}$.

The output noise component n_t is then forecast by setting $a_t = 0$, for $t = n+1, \ldots, n+L$, and using the ARIMA model equations:

(b) $e_t = \phi_1 e_{t-1} + \phi_2 e_{t-2} + \cdots + \phi_p e_{t-p} + a_t - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \cdots - \theta_1 a_{t-q}$

(c) $w_t = \Phi_1 w_{t-s} + \Phi_2 w_{t-2\times s} + \cdots + \Phi_P w_{t-P\times s} + e_t - \Theta_1 e_{t-s} - \Theta_2 e_{t-2\times s} - \cdots - \Theta_Q e_{t-Q\times s}$

(d) $n_t = \left(\nabla^d \nabla_s^D\right)^{-1} (w_t + c)$.

This last step of 'integration' reverses the process of differencing. Finally the output forecasts are calculated as

$$y_t = z_{1,t} + z_{2,t} + \cdots + z_{m,t} + n_t.$$

The forecast error variance of y_{t+l} (i.e., at lead time l) is S_l^2, which is the sum of parts which arise from the various input series, and the output noise component. That part due to the output noise is

$$sn_l^2 = V_n \times \left(\psi_0^2 + \psi_1^2 + \cdots + \psi_{l-1}^2\right),$$

where V_n is the estimated residual variance of the output noise ARIMA model, and $\psi_0, \psi_1, \ldots$ are the 'psi-weights' of this model as defined in Box and Jenkins (1976). They are calculated by applying the equations (b), (c) and (d) above, for $t = 0, 1, \ldots, L$, but with artificial values for the various series and with the constant c set to 0. Thus all values of a_t, e_t, w_t and n_t are taken as zero, for $t < 0$; a_t is taken to be 1, for $t = 0$ and 0, for $t > 0$. The resulting values of n_t, for $t = 0, 1, \ldots, L$, are precisely $\psi_0, \psi_1, \ldots, \psi_L$ as required.

Further contributions to S_l^2 come only from those input series, for which future values are forecasts which have been obtained by applying input series ARIMA models. For such a series the contribution is

$$sz_l^2 = V_x \times \left(\nu_0^2 + \nu_1^2 + \cdots + \nu_{l-1}^2\right),$$

where V_x is the estimated residual variance of the input series ARIMA model. The coefficients $\nu_0, \nu_1, \ldots$ are calculated by applying the transfer function model equation (a) above, for $t = 0, 1, \ldots, L$, but again with artificial values of the series. Thus all values of z_t and x_t, for $t < 0$, are taken to be zero, and $x_0, x_1, \ldots$ are taken to be the psi-weight sequence $\psi_0, \psi_1, \ldots$ for the **input series** ARIMA model. The resulting values of z_t, for $t = 0, 1, \ldots, L$, are precisely $\nu_0, \nu_1, \ldots, \nu_L$ as required.

In adding such contributions sz_l^2 to sn_l^2 to make up the total forecast error variance S_l^2, it is assumed that the various input series with which these contributions are associated are statistically independent of each other.

When using the routine in practice an ARIMA model is required for all the input series. In the case of those inputs for which no such ARIMA model is available (or its effects are to be excluded), the corresponding orders and parameters and the estimated residual variance should be set to zero.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden–Day

5 Arguments

1: STTF(NSTTF) – REAL (KIND=nag_wp) array *Input*

On entry: the NSTTF values in the state set as returned by G13BEF or G13BGF.

2: NSTTF – INTEGER *Input*

On entry: the exact number of values in the state set array STTF as returned by G13BEF or G13BGF.

3: MR(7) – INTEGER array *Input*

On entry: the orders vector (p, d, q, P, D, Q, s) of the ARIMA model for the output noise component.

p, q, P and Q give respectively the number of autoregressive (ϕ), moving average (θ), seasonal autoregressive (Φ) and seasonal moving average (Θ) parameters.

d, D and s refer respectively to the order of non-seasonal differencing, the order of seasonal differencing, and the seasonal period.

Constraints:

p, d, q, P, D, Q, $s \geq 0$;
$p + q + P + Q > 0$;
$s \neq 1$;
if $s = 0$, $P + D + Q = 0$;
if $s > 1$, $P + D + Q > 0$.

4: NSER – INTEGER *Input*

On entry: the total number of input and output series. There may be any number of input series (including none), but only one output series.

5: MT(4, NSER) – INTEGER array *Input*

On entry: the transfer function orders b, p and q of each of the input series. The data for input series i are held in column i. Row 1 holds the value b_i, row 2 holds the value q_i and row 3 holds the value p_i. For a simple input, $b_i = q_i = p_i = 0$.

Row 4 holds the value r_i, where $r_i = 1$ for a simple input, $r_i = 2$ or 3 for a transfer function input. When $r_i = 1$, any nonzero contents of rows 1, 2 and 3 of column i are ignored. The choice of $r_i = 2$ or $r_i = 3$ is an option for use in model estimation and does not affect the operation of G13BHF.

Constraint: $\mathrm{MT}(4, i) = 1$, 2 or 3, for $i = 1, 2, \ldots, \mathrm{NSER} - 1$.

6: PARA(NPARA) – REAL (KIND=nag_wp) array *Input*

On entry: estimates of the multi-input model parameters as returned by G13BEF. These are in order, firstly the ARIMA model parameters: p values of ϕ parameters, q values of θ parameters, P values of Φ parameters and Q values of Θ parameters. These are followed by the transfer function model parameter values $\omega_0, \omega_1, \ldots, \omega_{q_1}$, $\delta_1, \delta_2, \ldots, \delta_{p_1}$ for the first of any input series and similar sets of values for any subsequent input series. The final component of PARA is the constant c.

7: NPARA – INTEGER *Input*

On entry: the exact number of ϕ, θ, Φ, Θ, ω, δ and c parameters. (c must be included, whether its value was previously estimated or was set fixed).

8: NFV – INTEGER *Input*

On entry: the number of forecast values required.

9: XXYN(LDXXYN, NSER) – REAL (KIND=nag_wp) array *Input/Output*

On entry: the supplied NFV values for each of the input series required to produce the NFV output series forecasts. Column i contains the values for input series i. Column NSER need not be supplied.

On exit: if $\mathrm{KZEF} = 0$, then column NSER of XXYN contains the output series forecast values (as does FVA), but XXYN is otherwise unchanged.

If $\mathrm{KZEF} \neq 0$, then the columns of XXYN hold the corresponding values of the forecast components z_t for each of the input series and the values of the output noise component n_t in that order.

10: LDXXYN – INTEGER *Input*

On entry: the first dimension of the array XXYN as declared in the (sub)program from which G13BHF is called.

Constraint: $\mathrm{LDXXYN} \geq \mathrm{NFV}$.

11: MRX(7, NSER) – INTEGER array *Input/Output*

On entry: the orders array for each of the input series ARIMA models. Thus, column i contains values of p, d, q, P, D, Q, s for input series i. In the case of those inputs for which no ARIMA model is available, the corresponding orders should be set to 0. (The model for any input series only affects the standard errors of the forecast values.)

On exit: unchanged, apart from column NSER which is used for workspace.

12: PARX(LDPARX, NSER) – REAL (KIND=nag_wp) array *Input*

On entry: values of the parameters (ϕ, θ, Φ and Θ) for each of the input series ARIMA models. Thus column i contains $\mathrm{MRX}(1, i)$ values of ϕ parameters, $\mathrm{MRX}(3, i)$ values of θ parameters, $\mathrm{MRX}(4, i)$ values of Φ parameters and $\mathrm{MRX}(6, i)$ values of Θ parameters – in that order.

Values in the columns relating to those input series for which no ARIMA model is available are ignored. (The model for any input series only affects the standard errors of the forecast values.)

13: LDPARX – INTEGER *Input*

On entry: the first dimension of the array PARX as declared in the (sub)program from which G13BHF is called.

Constraint: LDPARX $\geq ncd$, where ncd is the maximum number of parameters in any of the input series ARIMA models. If there are no input series, LDPARX ≥ 1.

14: RMSXY(NSER) – REAL (KIND=nag_wp) array *Input*

On entry: the estimated residual variances for each input series ARIMA model followed by that for the output noise ARIMA model. In the case of those inputs for which no ARIMA model is available, or when its effects are to be excluded in the calculation of forecast standard errors, the corresponding entry of RMSXY should be set to 0.

15: KZEF – INTEGER *Input*

On entry: must not be set to 0, if the values of the input component series z_t and the values of the output noise component n_t are to overwrite the contents of XXYN on exit, and must be set to 0 if XXYN is to remain unchanged on exit, apart from the appearance of the forecast values in column NSER.

16: FVA(NFV) – REAL (KIND=nag_wp) array *Output*

On exit: the required forecast values for the output series.

17: FSD(NFV) – REAL (KIND=nag_wp) array *Output*

On exit: the standard errors for each of the forecast values.

18: WA(IWA) – REAL (KIND=nag_wp) array *Workspace*
19: IWA – INTEGER *Input*

On entry: the dimension of the array WA as declared in the (sub)program from which G13BHF is called.

A good, slightly conservative approximation to the required size of IWA is given by

$$\mathrm{IWA} \geq 4 \times (\mathrm{NSTTF} + \mathrm{NFV} + ncf)$$

where ncf is the largest number of ARIMA parameters in any one of the input or output series.

An exact value for the required size of IWA can be calculated as follows:

Let $ncg = \max(p_i)$,

$nch = \max(b_i + q_i)$,

$nci = \max(b_i + q_i + p_i)$,

over each of the transfer function input series for which $r_i > 1$, where b_i, q_i, p_i are the orders held in rows 1 to 3 of array MT.

Let $ncj = 1 + nci$,

$nck = \mathrm{NFV} + \max(ncg, nch)$,

$ncl = \max(\mathrm{NSTTF}, ncf, ncj, nck)$,

$ncm = \max(\mathrm{NSTTF} + 4 \times ncf, ncl)$,

then $\mathrm{IWA} \geq ncm + 3 \times ncl + \mathrm{NFV}$.

20: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, NSTTF is not consistent with the orders in arrays MR and MT.

IFAIL $= 2$

On entry, NPARA is not consistent with the orders in arrays MR and MT.

IFAIL $= 3$

On entry, LDXXYN is too small.

IFAIL $= 4$

On entry, IWA is too small.

IFAIL $= 5$

On entry, LDPARX is too small.

IFAIL $= 6$

On entry, one of the r_i, stored in MT$(4, i)$, for $i = 1, 2, \ldots,$NSER – 1, does not equal 1, 2 or 3.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computations are believed to be stable.

8 Parallelism and Performance

G13BHF is not threaded in any implementation.

9 Further Comments

The time taken by G13BHF is approximately proportional to NFV $\times$ NPARA.

10 Example

This example follows up that described in G13BGF and makes use of its data. These consist of output series orders and parameter values, input series transfer function orders and the updated state set.

Four new values of the input series are supplied, as are the orders and parameter values for the single input series ARIMA model (which has 2 values of ϕ, 2 values of θ, 1 value of Θ, single seasonal differencing and a seasonal period of 4), and the estimated residual variances for the input series ARIMA model and the output noise ARIMA model.

Four forecast values and their standard errors are computed and printed; also the values of the components z_t and the output noise component n_t corresponding to the forecasts.

10.1 Program Text

```
    Program g13bhfe

!     G13BHF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g13bhf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, iwa, kzef, ldparx, ldxxyn, &
                                          ncf, ncg, nch, nci, ncj, nck, ncl,   &
                                          ncm, nfv, nis, npara, nparax, nser,  &
                                          nsttf
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: fsd(:), fva(:), para(:), parx(:,:),  &
                                          rmsxy(:), sttf(:), wa(:), xxyn(:,:)
      Integer                          :: mr(7)
      Integer, Allocatable             :: mrx(:,:), mt(:,:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max
!     .. Executable Statements ..
      Write (nout,*) 'G13BHF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in problem size
      Read (nin,*) nsttf, nser, nfv, kzef

!     Number of input series
      nis = nser - 1

!     Read in the orders
      Read (nin,*)(mr(i),i=1,7)

      Allocate (mt(4,nser))

!     Read in transfer function
      Do i = 1, nis
```

```
          Read (nin,*) mt(1:4,i)
        End Do

!       Calculate NPARA
        npara = 0
        Do i = 1, nis
          npara = npara + mt(2,i) + mt(3,i)
        End Do
        npara = npara + mr(1) + mr(3) + mr(4) + mr(6) + nser

        ldxxyn = nfv
        ldparx = npara
        Allocate (para(npara),sttf(nsttf),xxyn(ldxxyn,nser),mrx(7,nser),         &
          parx(ldparx,nser),rmsxy(nser),fva(nfv),fsd(nfv))

!       Read in rest of data
        Read (nin,*) sttf(1:nsttf)
        Read (nin,*) para(1:npara)
        Read (nin,*)(xxyn(i,1:nis),i=1,nfv)
        ncf = mr(1) + mr(3) + mr(4) + mr(6)
        Do i = 1, nis
          Read (nin,*) mrx(1:7,i)
          nparax = mrx(1,i) + mrx(3,i) + mrx(4,i) + mrx(6,i)
          ncf = max(ncf,nparax)
          Read (nin,*) parx(1:nparax,i)
        End Do
        Read (nin,*) rmsxy(1:nser)

!       Calculate size of workspace array
        ncg = 0
        nch = 0
        nci = 0
        Do i = 1, nis
          If (mt(4,i)>1) Then
            ncg = max(ncg,mrx(1,i))
            nch = max(nch,mt(1,i)+mrx(3,i))
            nci = max(nci,mt(1,i)+mrx(3,i)+mrx(1,i))
          End If
          ncf = max(ncf,mrx(1,i)+mrx(3,i)+mrx(4,i)+mrx(6,i))
        End Do
        ncj = nci + 1
        nck = nfv + max(ncg,nch)
        ncl = max(nsttf,ncf,ncj,nck)
        ncm = max(nsttf+4*ncf,ncl)
        iwa = ncm + 3*ncl + nfv
        Allocate (wa(iwa))

!       Produce forecasts
        ifail = 0
        Call g13bhf(sttf,nsttf,mr,nser,mt,para,npara,nfv,xxyn,ldxxyn,mrx,parx,   &
          ldparx,rmsxy,kzef,fva,fsd,wa,iwa,ifail)

!       Display results
        Write (nout,*) 'The forecast values and their standard errors'
        Write (nout,*)
        Write (nout,*) '   I     FVA       FSD'
        Write (nout,*)
        Write (nout,99999)(i,fva(i),fsd(i),i=1,nfv)
        Write (nout,*)
        Flush (nout)
        ifail = 0
        Call x04caf('General',' ',nfv,nser,xxyn,ldxxyn,                          &
          'The values of z(t) and n(t)',ifail)
        Write (nout,99998) 'The first ', nis,                                    &
          ' columns hold the z(t) and the last column the n(t)'

99999 Format (1X,I4,2F10.4)
99998 Format (1X,A,I0,A)
    End Program g13bhfe
```

10.2 Program Data

```
G13BHF Example Program Data
   10    2    4    1                       :: NSTTF,NSER,NFV,KZEF
    1    0    0    0    1    1    4        :: MR
1 0 1 3                                    :: Transfer fun. for series 1, MT(:,1)
  6.7160 158.3022 -80.3352 -74.8937
-80.7694 -70.3022   0.8476  -2.0234
 -5.8080  10.2943                          :: STTF
  0.5158   0.9994   8.6343   0.6726
 -0.3172                                   :: PARA
   6.923
   6.939
   6.705
   6.914                                   :: Input series, XXYN(:,1:(NSER-1))
2 0 2 0 1 1 4                              :: Order for input series 1, MRX(:,1)
1.6743 -0.9505 1.4605 -0.4862 0.8993       :: Params for input series 1, PARX(:,1)
0.1720 22.9256                             :: RMSXY
```

10.3 Program Results

```
 G13BHF Example Program Results

 The forecast values and their standard errors

   I     FVA       FSD

   1   88.2723    4.7881
   2   99.9425    6.4690
   3  100.6499    7.3175
   4   95.0958    7.5534

 The values of z(t) and n(t)
            1          2
 1    164.4620   -76.1897
 2    170.3924   -70.4499
 3    174.5193   -73.8694
 4    175.2747   -80.1789
 The first 1 columns hold the z(t) and the last column the n(t)
```