

NAG Library Routine Document

H03BBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

H03BBF calculates an approximate solution to a symmetric travelling salesman problem using simulated annealing via a configuration free interface.

2 Specification

```
SUBROUTINE H03BBF (NC, DM, BOUND, TARGC, PATH, COST, TMODE, ALG_STATS,      &
                  STATE, IFAIL)
INTEGER          NC, PATH(NC), TMODE, STATE(*), IFAIL
REAL (KIND=nag_wp) DM(NC,NC), BOUND, TARGC, COST, ALG_STATS(6)
```

3 Description

H03BBF provides a probabilistic strategy for the calculation of a near optimal path through a symmetric and fully connected distance matrix; that is, a matrix for which element (i, j) is the pairwise distance (also called the cost, or weight) between nodes (cities) i and j . This problem is better known as the Travelling Salesman Problem (TSP), and symmetric means that the distance to travel between two cities is independent of which is the destination city.

In the classical TSP, which this routine addresses, a salesman wishes to visit a given set of cities once only by starting and finishing in a home city and travelling the minimum total distance possible. It is one of the most intensively studied problems in computational mathematics and, as a result, has developed some fairly sophisticated techniques for getting near-optimal solutions for large numbers of cities. H03BBF adopts a very simple approach to try to find a reasonable solution, for moderately large problems. The routine uses simulated annealing: a stochastic mechanical process in which the heating and controlled cooling of a material is used to optimally refine its molecular structure.

The material in the TSP is the distance matrix and a given state is represented by the order in which each city is visited—the path. This system can move from one state to a neighbouring state by selecting two cities on the current path at random and switching their places; the order of the cities in the path between the switched cities is then reversed. The cost of a state is the total cost of traversing its path; the resulting difference in cost between the current state and this new proposed state is called the delta; a negative delta indicates the proposal creates a more optimal path and a positive delta a less optimal path. The random selection of cities to switch uses random number generators (RNGs) from Chapter G05; it is thus necessary to initialize a state array for the RNG of choice (by a call to G05KFF or G05KGF) prior to calling H03BBF.

The simulation itself is executed in two stages. In the first stage, a series of sample searches through the distance matrix is conducted where each proposed new state is accepted, regardless of the change in cost (delta) incurred by applying the switches, and statistics on the set of deltas are recorded. These metrics are updated after each such sample search; the number of these searches and the number of switches applied in each search is dependent on the number of cities. The final collated set of metrics for the deltas obtained by the first stage are used as control parameters for the second stage. If no single improvement in cost is found during the first stage, the algorithm is terminated.

In the second stage, as before, neighbouring states are proposed. If the resulting delta is negative or causes no change the proposal is accepted and the path updated; otherwise moves are accepted based on a probabilistic criterion, a modified version of the Metropolis–Hastings algorithm.

The acceptance of some positive deltas (increased cost) reduces the probability of a solution getting trapped at a non-optimal solution where any single switch causes an increase in cost. Initially the

acceptance criteria allow for relatively large positive deltas, but as the number of proposed changes increases, the criteria become more stringent, allowing fewer positive deltas of smaller size to be accepted; this process is, within the realm of the simulated annealing algorithm, referred to as ‘cooling’. Further exploration of the system is initially encouraged by accepting non-optimal routes, but is increasingly discouraged as the process continues.

The second stage will terminate when:

- a solution is obtained that is deemed acceptable (as defined by supplied values);
- the algorithm will accept no further positive deltas and a set of proposed changes have resulted in no improvements (has cooled);
- a number of consecutive sets of proposed changes has resulted in no improvement.

4 References

Applegate D L, Bixby R E, Chvátal V and Cook W J (2006) *The Traveling Salesman Problem: A Computational Study* Princeton University Press

Cook W J (2012) *In Pursuit of the Traveling Salesman* Princeton University Press

Johnson D S and McGeoch L A The traveling salesman problem: A case study in local optimization *Local search in combinatorial optimization* (1997) 215–310

Press W H, Teukolsky S A, Vetterling W T and Flannery B P (2007) *Numerical Recipes The Art of Scientific Computing (3rd Edition)*

Rego C, Gamboa D, Glover F and Osterman C (2011) Traveling salesman problem heuristics: leading methods, implementations and latest advances *European Journal of Operational Research* **211** (3) 427–441

Reinelt G (1994) *The Travelling Salesman. Computational Solutions for TSP Applications, Volume 840 of Lecture Notes in Computer Science* Springer–Verlag, Berlin Heidelberg New York

5 Arguments

- 1: NC – INTEGER *Input*
On entry: the number of cities. In the trivial cases $NC = 1, 2$ or 3 , the routine returns the optimal solution immediately with $TMODE = 0$ (provided the relevant distance matrix entries are not negative).
Constraint: $NC \geq 1$.
- 2: DM(NC,NC) – REAL (KIND=nag_wp) array *Input*
On entry: the distance matrix; each $DM(i, j)$ is the effective cost or weight between nodes i and j . Only the strictly upper half of the matrix is referenced.
Constraint: $DM(i, j) \geq 0.0$, for $j = 2, 3, \dots, NC$ and $i = 1, 2, \dots, j - 1$.
- 3: BOUND – REAL (KIND=nag_wp) *Input*
On entry: a lower bound on the solution. If the optimum is unknown set BOUND to zero or a negative value; the routine will then calculate the minimum spanning tree for DM and use this as a lower bound (returned in ALG_STATS(6)). If an optimal value for the cost is known then this should be used for the lower bound. A detailed discussion of relaxations for lower bounds, including the minimal spanning tree, can be found in Reinelt (1994).
- 4: TARGC – REAL (KIND=nag_wp) *Input*
On entry: a measure of how close an approximation needs to be to the lower bound. The routine terminates when a cost is found less than or equal to $BOUND + TARGC$. This argument is useful when an optimal value for the cost is known and supplied in BOUND. It may be sufficient

to obtain a path that is close enough (in terms of cost) to the optimal path; this allows the algorithm to terminate at that point and avoid further computation in attempting to find a better path.

If $TARGC < 0$, $TARGC = 0$ is assumed.

- 5: **PATH(NC)** – INTEGER array *Output*
On exit: the best path discovered by the simulation. That is, PATH contains the city indices in path order. If $IFAIL \neq 0$ on exit, PATH contains the indices 1 to NC.

- 6: **COST** – REAL (KIND=nag_wp) *Output*
On exit: the cost or weight of PATH. If $IFAIL \neq 0$ on exit, COST contains the largest model real number (see X02BLF).

- 7: **TMODE** – INTEGER *Output*
On exit: the termination mode of the routine (if $IFAIL \neq 0$ on exit, TMODE is set to -1):
 TMODE = 0
 Optimal solution found, $COST = BOUND$.
 TMODE = 1
 System temperature cooled. The algorithm returns a PATH and associated COST that does not attain, nor lie within TARGC of, the BOUND. This could be a sufficiently good approximation to the optimal PATH, particularly when $BOUND + TARGC$ lies below the optimal COST.
 TMODE = 2
 Halted by COST falling within the desired TARGC range of the BOUND.
 TMODE = 3
 System stalled following lack of improvement.
 TMODE = 4
 Initial search failed to find a single improvement (the solution could be optimal).

- 8: **ALG_STATS(6)** – REAL (KIND=nag_wp) array *Output*
On exit: an array of metrics collected during the initial search. These could be used as a basis for future optimization. If $IFAIL \neq 0$ on exit, the elements of ALG_STATS are set to zero; the first five elements are also set to zero in the trivial cases $NC = 1, 2$ or 3 .
 ALG_STATS(1)
 Mean delta.
 ALG_STATS(2)
 Standard deviation of deltas.
 ALG_STATS(3)
 Cost at end of initial search phase.
 ALG_STATS(4)
 Best cost encountered during search phase.
 ALG_STATS(5)
 Initial system temperature. At the end of stage 1 of the algorithm, this is a function of the mean and variance of the deltas, and of the distance from best cost to the lower bound. It is a measure of the initial acceptance criteria for stage 2. The larger this value, the more iterations it will take to geometrically reduce it during stage 2 until the system is cooled (below a threshold value).

ALG_STATS(6)

The lower bound used, which will be that computed internally when $\text{BOUND} \leq 0$ on input. Subsequent calls with different random states can set BOUND to the value returned in $\text{ALG_STATS}(6)$ to avoid recomputation of the minimal spanning tree.

9: $\text{STATE}(\ast)$ – INTEGER array

Communication Array

Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF .

On entry: a valid RNG state initialized by G05KFF or G05KGF . Since the algorithm used is stochastic, a random number generator is employed; if the generator is initialized to a non-repeatable sequence (G05KGF) then different solution paths will be taken on successive runs, returning possibly different final approximate solutions.

On exit: contains updated information on the state of the generator.

10: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{NC} = \langle \text{value} \rangle$.
Constraint: $\text{NC} \geq 1$.

$\text{IFAIL} = 2$

On entry, the strictly upper triangle of DM had a negative element.

$\text{IFAIL} = 9$

On entry, STATE vector has been corrupted or not initialized.

$\text{IFAIL} = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$\text{IFAIL} = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The routine will not perform well when the average change in cost caused by switching two cities is small relative to the cost; this can happen when many of the values in the distance matrix are relatively close to each other.

The quality of results from this routine can vary quite markedly when different initial random states are used. It is therefore advisable to compute a number of approximations using different initial random states. The best cost and path can then be taken from the set of approximations obtained. If no change in results is obtained after 10 such trials then it is unlikely that any further improvement can be made by this routine.

8 Parallelism and Performance

Running many instances of the routine in parallel with independent random number generator states can yield a set of possible solutions from which a best approximate solution may be chosen.

9 Further Comments

Memory is internally allocated for $3 \times NC - 2$ integers and $NC - 1$ real values.

In the case of two cities that are not connected, a suitably large number should be used as the distance (cost) between them so as to deter solution paths which directly connect the two cities. Solutions which contain an artificial link (i.e., a connection with a large distance between them to indicate no actual link) may be patched, using the shortest path algorithm H03ADF.

If a city is to be visited more than once (or more than twice for the home city) then the distance matrix should contain multiple entries for that city (on rows and columns i_1, i_2, \dots) with zero entries for distances to itself and identical distances to other cities.

10 Example

An approximation to the best path through 21 cities in the United Kingdom and Ireland, beginning and ending in Oxford, is sought. A lower bound is calculated internally.

10.1 Program Text

Program h03bbfe

```
!      H03BBF Example Program Text
!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g05kff, h03bbf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: lseed = 4, nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: bound, cost, targc
      Integer                     :: genid, i, i2, ib, ifail, j, l,      &
                                   lstate, nb, nc, subid, tmode
!      .. Local Arrays ..
      Real (Kind=nag_wp)          :: alg_stats(6)
      Real (Kind=nag_wp), Allocatable :: dm(:, :)
      Integer, Allocatable         :: path(:), state(:)
      Integer                     :: seed(lseed)
      Character (20), Allocatable :: cities(:)
```

```

!      .. Intrinsic Procedures ..
      Intrinsic                                :: len_trim, max, min, repeat, trim
!      .. Executable Statements ..

      Write (nout,*) 'H03BBF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Number of cities
      Read (nin,*) nc

!      Allocate distance matrix and path
      Allocate (path(nc),dm(nc,nc))
!      Read distance matrix 10 columns at a time
      nb = (nc+8)/10
      Do ib = 1, nb
         Read (nin,*)
         Read (nin,*)
         i2 = min(10*ib,nc-1)
         Do i = 1, i2
            Read (nin,*)(dm(i,j),j=max(i+1,10*ib-8),i2+1)
         End Do
      End Do

      Allocate (cities(nc))
      Do i = 1, nc
         Read (nin,*) cities(i)
      End Do

!      Calculate a lower bound internally and try to find lowest cost path.
      bound = -1.0_nag_wp
      targc = -1.0_nag_wp

!      Initialize the random number state array.
!      Use the query mechanism to find the required lstate.
      genid = 2
      subid = 53
      seed(:) = (/304950,889934,209094,23423990/)
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)
      Deallocate (state)
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Find low cost return path through all cities
      ifail = 0
      Call h03bbf(nc,dm,bound,targc,path,cost,tmode,alg_stats,state,ifail)

      Write (nout,99999) 'Initial search end cost', alg_stats(3)
      Write (nout,99999) 'Search best cost', alg_stats(4)
      Write (nout,99999) 'Initial temperature', alg_stats(5)
      Write (nout,99999) 'Lower bound', alg_stats(6)
      Write (nout,99998) 'Termination mode', tmode
      Write (nout,*)
      Write (nout,99999) 'Final cost', cost

      Write (nout,*)
      Write (nout,*) 'Final Path:'
      Write (nout,99997) trim(cities(path(1))), trim(cities(path(2)))
      l = len_trim(cities(path(1)))
      Write (nout,99997)(repeat(' ',l),trim(cities(path(i+1))),i=2,nc-1)
      Write (nout,99997) repeat(' ',l), trim(cities(path(1)))
99999 Format (1X,A,':',F12.2)
99998 Format (1X,A,':',I12)
99997 Format (1X,A,'--> ',A)
      End Program h03bbfe

```

10.2 Program Data

H03BBF Example Program Data

```

21                                     : number of cities

      2      3      4      5      6      7      8      9      10      11
23961  7112  21331  9050  22548  20667  13227  11617  14292  9455  : 1
      25998      4724  27936  2014  3997  20826  30488  21891  28327  : 2
              23108  2871  24325  22444  15004  8664  16359  6503  : 3
              25203  3444  3379  18093  27755  19158  25593  : 4
                  26434  24553  15169  10773  16033  8612  : 5
                      2668  19496  29159  20562  26997  : 6
                      17550  27212  18615  25051  : 7
                          19516  1895  17354  : 8
                          20649  3135  : 9
                              18537 :10

      12      13      14      15      16      17      18      19      20      21
19634  6394  29483  14068  28136  11052  7228  13771  4752  24111 : 1
      5403  25281  9312  31882  4751  18651  24909  25448  20113  25289 : 2
      21411  1263  31260  7889  29913  12829  12517  8941  7038  26178 : 3
      3598  22547  10592  29149  8868  15918  21956  22715  17380  23484 : 4
      23519  3372  33368  5988  32022  13917  14626  6916  9147  25852 : 5
      4074  23951  7766  30553  6075  17322  23580  24119  18784  23960 : 6
      2127  22005  9586  28606  8239  15375  21634  22172  16837  22013 : 7
      16200  14308  26049  15136  24703  2447  14727  8446  9140  11714 : 8
      25990  7981  35839  15655  34493  17409  17103  15937  11618  30467 : 9
      17383  15491  7232  16033  25886  3630  15910  9343  10323  9866 :10
      23819  5810  33668  13484  32321  15237  14931  13766  9446  28296 :11
              21026  10985  27628  9638  14397  20655  21193  15858  20188 :12
                  30598  8276  29252  12168  11856  9064  6377  25227 :13
                  37538  9425  24307  30565  31103  25769  30945 :14
                      35803  14744  19628  6869  14149  26227 :15
                      22962  29220  29758  24423  29599 :16
                          12712  8242  7126  13457 :17
                          15366  6300  25639 :18
                              9465  18936 :19
                                  20048 :20, dm

Oxford                                     : 1
Dundee                                     : 2
Cardiff                                   : 3
Edinburgh                               : 4
Swansea                                 : 5
Perth                                   : 6
Stirling                                 : 7
Bangor                                  : 8
Plymouth                                : 9
Holyhead                               :10
Exeter                                  :11
Glasgow                                :12
Newport                                :13
Inverness                               :14
St.Davids                              :15
Aberdeen                               :16
St.Asaph                               :17
Cambridge                              :18
Aberystwyth                            :19
Birmingham                             :20
Dublin                                 :21, cities

```

10.3 Program Results

H03BBF Example Program Results

```

Initial search end cost:  432459.00
Search best cost       :  237068.00
Initial temperature    :  598481.00
Lower bound            :  106350.00

```

Termination mode : 3

Final cost : 131580.00

Final Path:

Oxford --> Cambridge
--> Birmingham
--> Glasgow
--> Stirling
--> Edinburgh
--> Perth
--> Dundee
--> Aberdeen
--> Inverness
--> Holyhead
--> Dublin
--> Bangor
--> St.Asaph
--> Aberystwyth
--> St.Davids
--> Swansea
--> Cardiff
--> Newport
--> Exeter
--> Plymouth
--> Oxford
