

NAG Library Routine Document

M01NAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

M01NAF searches an ordered vector of real numbers and returns the index of the first value equal to the sought-after item.

2 Specification

```
FUNCTION M01NAF (VALID, RV, M1, M2, ITEM, IFAIL)
  INTEGER M01NAF
  INTEGER          M1, M2, IFAIL
  REAL (KIND=nag_wp) RV(M2), ITEM
  LOGICAL          VALID
```

3 Description

M01NAF is based on Professor Niklaus Wirth's implementation of the Binary Search algorithm (see Wirth (2004)), but with two modifications. First, if the sought-after item is less than the value of the first element of the array to be searched, 0 is returned. Second, if a value equal to the sought-after item is not found, the index of the immediate lower value is returned.

4 References

Wirth N (2004) *Algorithms and Data Structures* 35–36 Prentice Hall

5 Arguments

- | | | |
|----|--|--------------|
| 1: | VALID – LOGICAL | <i>Input</i> |
| | <i>On entry:</i> if VALID is set to .TRUE. argument checking will be performed. If VALID is set to .FALSE. M01NAF will be called without argument checking (which includes checking that array RV is sorted in ascending order) and the routine will return with IFAIL = 0. See Section 9 for further details. | |
| 2: | RV(M2) – REAL (KIND=nag_wp) array | <i>Input</i> |
| | <i>On entry:</i> elements M1 to M2 contain real values to be searched. | |
| | <i>Constraint:</i> elements M1 to M2 of RV must be sorted in ascending order. | |
| 3: | M1 – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the index of the first element of RV to be searched. | |
| | <i>Constraint:</i> $M1 \geq 1$. | |
| 4: | M2 – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the index of the last element of RV to be searched. | |
| | <i>Constraint:</i> $M2 \geq M1$. | |

- 5: ITEM – REAL (KIND=nag_wp) *Input*
On entry: the sought-after item.
- 6: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

(**Note:** these errors will only be returned if VALID = .TRUE..)

IFAIL = 2

On entry, RV must be sorted in ascending order: RV element $\langle value \rangle >$ element $\langle value \rangle$.

IFAIL = 3

On entry, M1 = $\langle value \rangle$.

Constraint: M1 \geq 1.

IFAIL = 4

On entry, M1 = $\langle value \rangle$, M2 = $\langle value \rangle$.

Constraint: M1 \leq M2.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

M01NAF is not threaded in any implementation.

9 Further Comments

The argument VALID should be used with caution. Set it to .FALSE. only if you are confident that the other arguments are correct, in particular that array RV is in fact arranged in ascending order. If you wish to search the same array RV many times, you are recommended to set VALID to .TRUE. on first call of M01NAF and to .FALSE. on subsequent calls, in order to minimize the amount of time spent checking RV, which may be significant if RV is large.

The time taken by M01NAF is $O(\log(n))$, where $n = M2 - M1 + 1$, when VALID = .FALSE..

10 Example

This example reads a list of double precision numbers and sought-after items and performs the search for these items.

10.1 Program Text

```

Program m01naf

!      M01NAF Example Program Text
!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: m01naf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: item
      Integer                     :: i, ifail, index, ioerr, m1, m2
      Logical                     :: first
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: rv(:)
!      .. Executable Statements ..
      Write (nout,*) 'M01NAF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Read (nin,*) m2
      Allocate (rv(m2))

      m1 = 1

      Read (nin,*)(rv(i),i=m1,m2)

      first = .True.

data: Do
      Read (nin,*,Iostat=ioerr) item

      If (ioerr<0) Then
         Exit data
      End If

      ifail = 0
      index = m01naf(first,rv,m1,m2,item,ifail)

      If (first) Then
         Write (nout,*)
         Write (nout,*) 'Reference Vector is:'
         Write (nout,99999)(rv(i),i=m1,m2)

```

```

        first = .False.
      End If

      Write (nout,*)
      Write (nout,99998) item, index
    End Do data

99999 Format (1X,8F7.1)
99998 Format (1X,'Search for item ',F7.1,' returned index: ',I4)
      End Program m01nafe

```

10.2 Program Data

M01NAF Example Program Data

16	: M2
0.5 0.6 1.1 1.2 1.3 1.3 2.1 2.3	
2.3 4.1 5.8 5.9 6.5 6.5 8.6 9.9	: RV
2.1	: Item 1
0.4	: Item 2
7.1	: Item 3
10.0	: Item 4

10.3 Program Results

M01NAF Example Program Results

Reference Vector is:

0.5	0.6	1.1	1.2	1.3	1.3	2.1	2.3
2.3	4.1	5.8	5.9	6.5	6.5	8.6	9.9

Search for item 2.1 returned index: 7

Search for item 0.4 returned index: 0

Search for item 7.1 returned index: 14

Search for item 10.0 returned index: 16
