NAG Library Routine Document

S09AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S09AAF returns the value of the inverse circular sine, $\arcsin x$, via the function name. The value is in the principal range $(-\pi/2, \pi/2)$.

2 Specification

```
FUNCTION S09AAF (X, IFAIL)
REAL (KIND=nag_wp) S09AAF
INTEGER            IFAIL
REAL (KIND=nag_wp) X
```

3 Description

S09AAF calculates an approximate value for the inverse circular sine, $\arcsin x$. It is based on the Chebyshev expansion

$$\arcsin x = x \times y(x) = x \sum_{r=0} a_r T_r(t)$$

where $-\dfrac{1}{\sqrt{2}} \leq x \leq \dfrac{1}{\sqrt{2}}$ and $t = 4x^2 - 1$.

For $x^2 \leq \dfrac{1}{2}$, $\quad \arcsin x = x \times y(x)$.

For $\frac{1}{2} < x^2 \leq 1$, $\quad \arcsin x = \text{sign}\, x \left\{ \dfrac{\pi}{2} - \arcsin \sqrt{1 - x^2} \right\}$.

For $x^2 > 1$, $\quad \arcsin x$ is undefined and the routine fails.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Arguments

1: X – REAL (KIND=nag_wp) *Input*

On entry: the argument x of the function.

Constraint: $|\mathrm{X}| \leq 1.0$.

2: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The routine has been called with an argument greater than 1.0 in absolute value; $\arcsin x$ is undefined and the routine returns zero.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

If δ and ϵ are the relative errors in the argument and result, respectively, then in principle

$$|\epsilon| \simeq \left| \frac{x}{\arcsin x \sqrt{1-x^2}} \times \delta \right|.$$

That is, a relative error in the argument x is amplified by at least a factor $\dfrac{x}{\arcsin x \sqrt{1-x^2}}$ in the result.

The equality should hold if δ is greater than the ***machine precision*** (δ is a result of data errors etc.) but if δ is produced simply by round-off error in the machine it is possible that rounding in internal calculations may lose an extra figure in the result.

This factor stays close to one except near $|x| = 1$ where its behaviour is shown in the following graph.



Figure 1

For $|x|$ close to unity, $1 - |x| \sim \delta$, the above analysis is no longer applicable owing to the fact that both argument and result are subject to finite bounds, ($|x| \leq 1$ and $|\arcsin x| \leq \frac{1}{2}\pi$). In this region $\epsilon \sim \sqrt{\delta}$; that is the result will have approximately half as many correct significant figures as the argument.

For $|x| = 1$ the result will be correct to full ***machine precision***.

8 Parallelism and Performance

S09AAF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example reads values of the argument x from a file, evaluates the function at each value of x and prints the results.

10.1 Program Text

```
    Program s09aafe

!     S09AAF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s09aaf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: x, y
      Integer                          :: ifail, ioerr
!     .. Executable Statements ..
```

```
      Write (nout,*) 'S09AAF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) '     X           Y'
      Write (nout,*)

data: Do
        Read (nin,*,Iostat=ioerr) x

        If (ioerr<0) Then
          Exit data
        End If

        ifail = -1
        y = s09aaf(x,ifail)

        If (ifail<0) Then
          Exit data
        End If

        Write (nout,99999) x, y
      End Do data

99999 Format (1X,1P,2E12.3)
    End Program s09aafe
```

10.2 Program Data

```
S09AAF Example Program Data
            -0.5
             0.1
             0.9
```

10.3 Program Results

```
 S09AAF Example Program Results

      X           Y

   -5.000E-01  -5.236E-01
    1.000E-01   1.002E-01
    9.000E-01   1.120E+00
```