

NAG Library Routine Document

S15DDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S15DDF computes values of the function $w(z) = e^{-z^2} \operatorname{erfc}(-iz)$, for complex z .

2 Specification

```
FUNCTION S15DDF (Z, IFAIL)
COMPLEX (KIND=nag_wp) S15DDF
INTEGER IFAIL
COMPLEX (KIND=nag_wp) Z
```

3 Description

S15DDF computes values of the function $w(z) = e^{-z^2} \operatorname{erfc}(-iz)$, where $\operatorname{erfc} z$ is the complementary error function

$$\operatorname{erfc} z = \frac{2}{\sqrt{\pi}} \int_z^\infty e^{-t^2} dt,$$

for complex z . The method used is that in Gautschi (1970) for z in the first quadrant of the complex plane, and is extended for z in other quadrants via the relations $w(-z) = 2e^{-z^2} - w(z)$ and $w(\bar{z}) = \overline{w(-z)}$. Following advice in Gautschi (1970) and van der Laan and Temme (1984), the code in Gautschi (1969) has been adapted to work in various precisions up to 18 decimal places. The real part of $w(z)$ is sometimes known as the Voigt function.

4 References

Gautschi W (1969) Algorithm 363: Complex error function *Comm. ACM* **12** 635

Gautschi W (1970) Efficient computation of the complex error function *SIAM J. Numer. Anal.* **7** 187–198

van der Laan C G and Temme N M (1984) Calculation of special functions: the gamma function, the exponential integrals and error-like functions *CWI Tract 10* Centre for Mathematics and Computer Science, Amsterdam

5 Arguments

1: Z – COMPLEX (KIND=nag_wp) *Input*
On entry: the argument z of the function.

2: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The real part of the result overflows, and is set to the largest safe number with the correct sign. The imaginary part of the result is meaningful.

IFAIL = 2

The imaginary part of the result overflows, and is set to the largest safe number with the correct sign. The real part of the result is meaningful.

IFAIL = 3

Both real and imaginary parts of the result overflow, and are set to the largest safe number with the correct signs.

IFAIL = 4

The result returned is accurate to less than half precision, due to the size of an intermediate result.

IFAIL = 5

The result returned has no precision, due to the size of an intermediate result, and is set to zero.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the returned result depends on the argument z . If z lies in the first or second quadrant of the complex plane (i.e., $\text{Im}(z)$ is greater than or equal to zero), the result should be accurate almost to **machine precision**, except that there is a limit of about 18 decimal places on the achievable accuracy because constants in the routine are given to this precision. With such arguments, IFAIL can only return as IFAIL = 0.

If however $\text{Im}(z)$ is less than zero, accuracy may be lost in two ways; firstly, in the evaluation of e^{-z^2} , if $\text{Im}(-z^2)$ is large, in which case a warning will be issued through IFAIL = 4 or 5; and secondly, near the zeros of the required function, where precision is lost due to cancellation, in which case no warning

is given – the result has absolute accuracy rather than relative accuracy. Note also that in this half-plane, one or both parts of the result may overflow – this is signalled through IFAIL = 1, 2 or 3.

8 Parallelism and Performance

S15DDF is not threaded in any implementation.

9 Further Comments

The time taken for a call of S15DDF depends on the argument z , the time increasing as $|z| \rightarrow 0.0$.

S15DDF may be used to compute values of $\operatorname{erfc} z$ and $\operatorname{erf} z$ for complex z by the relations $\operatorname{erfc} z = e^{-z^2} w(iz)$, $\operatorname{erf} z = 1 - \operatorname{erfc} z$. (For real arguments, S15ADF and S15AEF should be used.)

10 Example

This example reads values of the argument z from a file, evaluates the function at each value of z and prints the results.

10.1 Program Text

```

Program s15ddfe

!      S15DDF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, s15ddf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Complex (Kind=nag_wp)      :: w, z
      Integer                    :: ifail, ioerr
!      .. Executable Statements ..
      Write (nout,*) 'S15DDF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) '              Z                      W'

data: Do
      Read (nin,*,Iostat=ioerr) z

      If (ioerr<0) Then
         Exit data
      End If

      ifail = -1
      w = s15ddf(z,ifail)

      If (ifail<0) Then
         Exit data
      End If

      Write (nout,99999) z, w
End Do data

99999 Format (1X, '(' ,F12.4, ',' ,F12.4, ')'          (' ,F12.4, ',' ,F12.4, ')')
End Program s15ddfe

```

10.2 Program Data

S15DDF Example Program Data

```
( 1.00E0,  0.00E0)      - Values for Z.
(-3.01E0,  0.75E0)
( 2.75E0, -1.52E0)
(-1.33E0, -0.54E0)
```

10.3 Program Results

S15DDF Example Program Results

Z		W	
(1.0000,	0.0000)	(0.3679,	0.6072)
(-3.0100,	0.7500)	(0.0522,	-0.1838)
(2.7500,	-1.5200)	(-0.1015,	0.1654)
(-1.3300,	-0.5400)	(-0.1839,	-0.7891)
