

NAG Library Routine Document

S17AUF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S17AUF returns an array of values for the Airy function, $\text{Ai}(x)$.

2 Specification

```
SUBROUTINE S17AUF (N, X, F, IVALID, IFAIL)
  INTEGER          N, IVALID(N), IFAIL
  REAL (KIND=nag_wp) X(N), F(N)
```

3 Description

S17AUF evaluates an approximation to the Airy function, $\text{Ai}(x_i)$ for an array of arguments x_i , for $i = 1, 2, \dots, n$. It is based on a number of Chebyshev expansions:

For $x < -5$,

$$\text{Ai}(x) = \frac{a(t) \sin z - b(t) \cos z}{(-x)^{1/4}}$$

where $z = \frac{\pi}{4} + \frac{2}{3}\sqrt{-x^3}$, and $a(t)$ and $b(t)$ are expansions in the variable $t = -2\left(\frac{5}{x}\right)^3 - 1$.

For $-5 \leq x \leq 0$,

$$\text{Ai}(x) = f(t) - xg(t),$$

where f and g are expansions in $t = -2\left(\frac{x}{5}\right)^3 - 1$.

For $0 < x < 4.5$,

$$\text{Ai}(x) = e^{-3x/2}y(t),$$

where y is an expansion in $t = 4x/9 - 1$.

For $4.5 \leq x < 9$,

$$\text{Ai}(x) = e^{-5x/2}u(t),$$

where u is an expansion in $t = 4x/9 - 3$.

For $x \geq 9$,

$$\text{Ai}(x) = \frac{e^{-z}v(t)}{x^{1/4}},$$

where $z = \frac{2}{3}\sqrt{x^3}$ and v is an expansion in $t = 2\left(\frac{18}{z}\right) - 1$.

For $|x| < \textit{machine precision}$, the result is set directly to $\text{Ai}(0)$. This both saves time and guards against underflow in intermediate calculations.

For large negative arguments, it becomes impossible to calculate the phase of the oscillatory function with any precision and so the routine must fail. This occurs if $x < -\left(\frac{3}{2\epsilon}\right)^{2/3}$, where ϵ is the **machine precision**.

For large positive arguments, where Ai decays in an essentially exponential manner, there is a danger of underflow so the routine must fail.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the number of points.
Constraint: $N \geq 0$.
- 2: X(N) – REAL (KIND=nag_wp) array *Input*
On entry: the argument x_i of the function, for $i = 1, 2, \dots, N$.
- 3: F(N) – REAL (KIND=nag_wp) array *Output*
On exit: $Ai(x_i)$, the function values.
- 4: IVALID(N) – INTEGER array *Output*
On exit: IVALID(i) contains the error code for x_i , for $i = 1, 2, \dots, N$.
 IVALID(i) = 0
 No error.
 IVALID(i) = 1
 x_i is too large and positive. F(i) contains zero. The threshold value is the same as for IFAIL = 1 in S17AGF, as defined in the Users' Note for your implementation.
 IVALID(i) = 2
 x_i is too large and negative. F(i) contains zero. The threshold value is the same as for IFAIL = 2 in S17AGF, as defined in the Users' Note for your implementation.
- 5: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, at least one value of X was invalid.
Check IVALID for more information.

IFAIL = 2

On entry, N = $\langle value \rangle$.
Constraint: $N \geq 0$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

For negative arguments the function is oscillatory and hence absolute error is the appropriate measure. In the positive region the function is essentially exponential-like and here relative error is appropriate. The absolute error, E , and the relative error, ϵ , are related in principle to the relative error in the argument, δ , by

$$E \simeq |x \text{Ai}'(x)|\delta, \epsilon \simeq \left| \frac{x \text{Ai}'(x)}{\text{Ai}(x)} \right| \delta.$$

In practice, approximate equality is the best that can be expected. When δ , ϵ or E is of the order of the **machine precision**, the errors in the result will be somewhat larger.

For small x , errors are strongly damped by the function and hence will be bounded by the **machine precision**.

For moderate negative x , the error behaviour is oscillatory but the amplitude of the error grows like

$$\text{amplitude}\left(\frac{E}{\delta}\right) \sim \frac{|x|^{5/4}}{\sqrt{\pi}}.$$

However the phase error will be growing roughly like $\frac{2}{3}\sqrt{|x|^3}$ and hence all accuracy will be lost for large negative arguments due to the impossibility of calculating sin and cos to any accuracy if $\frac{2}{3}\sqrt{|x|^3} > \frac{1}{\delta}$.

For large positive arguments, the relative error amplification is considerable:

$$\frac{\epsilon}{\delta} \sim \sqrt{x^3}.$$

This means a loss of roughly two decimal places accuracy for arguments in the region of 20. However very large arguments are not possible due to the danger of setting underflow and so the errors are limited in practice.

8 Parallelism and Performance

S17AUF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example reads values of X from a file, evaluates the function at each value of x_i and prints the results.

10.1 Program Text

```

Program s17aufe

!      S17AUF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, s17auf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: f(:), x(:)
      Integer, Allocatable        :: ivalid(:)
!      .. Executable Statements ..
      Write (nout,*) 'S17AUF Example Program Results'

!      Skip heading in data file
      Read (nin,*)

      Write (nout,*)
      Write (nout,*) '      X      F      IVALID'
      Write (nout,*)

      Read (nin,*) n

      Allocate (x(n),f(n),ivalid(n))

      Read (nin,*) x(1:n)

      ifail = 0
      Call s17auf(n,x,f,ivalid,ifail)

      Do i = 1, n
         Write (nout,99999) x(i), f(i), ivalid(i)
      End Do

99999 Format (1X,1P,2E12.3,I5)
End Program s17aufe

```

10.2 Program Data

S17AUF Example Program Data

7

-10.0 -1.0 0.0 1.0 5.0 10.0 20.0

10.3 Program Results

S17AUF Example Program Results

X	F	INVALID
-1.000E+01	4.024E-02	0
-1.000E+00	5.356E-01	0
0.000E+00	3.550E-01	0
1.000E+00	1.353E-01	0
5.000E+00	1.083E-04	0
1.000E+01	1.105E-10	0
2.000E+01	1.692E-27	0
