

NAG Library Routine Document

S22BBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S22BBF returns a value for the confluent hypergeometric function ${}_1F_1(a; b; x)$, with real parameters a and b and real argument x . The solution is returned in the scaled form ${}_1F_1(a; b; x) = m_f \times 2^{m_s}$. This function is sometimes also known as Kummer's function $M(a, b, x)$.

2 Specification

```
SUBROUTINE S22BBF (ANI, ADR, BNI, BDR, X, FRM, SCM, IFAIL)
  INTEGER          SCM, IFAIL
  REAL (KIND=nag_wp) ANI, ADR, BNI, BDR, X, FRM
```

3 Description

S22BBF returns a value for the confluent hypergeometric function ${}_1F_1(a; b; x)$, with real parameters a and b and real argument x , in the scaled form ${}_1F_1(a; b; x) = m_f \times 2^{m_s}$, where m_f is the real scaled component and m_s is the integer power of two scaling. This function is unbounded or not uniquely defined for b equal to zero or a negative integer.

The confluent hypergeometric function is defined by the confluent series,

$${}_1F_1(a; b; x) = M(a, b, x) = \sum_{s=0}^{\infty} \frac{(a)_s x^s}{(b)_s s!} = 1 + \frac{a}{b}x + \frac{a(a+1)}{b(b+1)2!}x^2 + \dots$$

where $(a)_s = 1(a)(a+1)(a+2)\dots(a+s-1)$ is the rising factorial of a . $M(a, b, x)$ is a solution to the second order ODE (Kummer's Equation):

$$x \frac{d^2 M}{dx^2} + (b-x) \frac{dM}{dx} - aM = 0. \quad (1)$$

Given the parameters and argument (a, b, x) , this routine determines a set of safe values $\{(\alpha_i, \beta_i, \zeta_i) \mid i \leq 2\}$ and selects an appropriate algorithm to accurately evaluate the functions $M_i(\alpha_i, \beta_i, \zeta_i)$. The result is then used to construct the solution to the original problem $M(a, b, x)$ using, where necessary, recurrence relations and/or continuation.

For improved precision in the final result, this routine accepts a and b split into an integral and a decimal fractional component. Specifically $a = a_i + a_r$, where $|a_r| \leq 0.5$ and $a_i = a - a_r$ is integral. b is similarly deconstructed.

Additionally, an artificial bound, $arwnd$ is placed on the magnitudes of a_i , b_i and x to minimize the occurrence of overflow in internal calculations. $arwnd = 0.0001 \times I_{\max}$, where $I_{\max} = X02BBF$. It should, however, not be assumed that this routine will produce an accurate result for all values of a_i , b_i and x satisfying this criterion.

Please consult the NIST Digital Library of Mathematical Functions or the companion (2010) for a detailed discussion of the confluent hypergeometric function including special cases, transformations, relations and asymptotic approximations.

4 References

NIST Handbook of Mathematical Functions (2010) (eds F W J Olver, D W Lozier, R F Boisvert, C W Clark) Cambridge University Press

Pearson J (2009) Computation of hypergeometric functions *MSc Dissertation, Mathematical Institute, University of Oxford*

5 Arguments

- 1: ANI – REAL (KIND=nag_wp) *Input*
On entry: a_i , the nearest integer to a , satisfying $a_i = a - a_r$.
Constraints:

$$\text{ANI} = \lfloor \text{ANI} \rfloor;$$

$$|\text{ANI}| \leq \text{arbn}d.$$
- 2: ADR – REAL (KIND=nag_wp) *Input*
On entry: a_r , the signed decimal remainder satisfying $a_r = a - a_i$ and $|a_r| \leq 0.5$.
Constraint: $|\text{ADR}| \leq 0.5$.
Note: if $|\text{ADR}| < 100.0\epsilon$, $a_r = 0.0$ will be used, where ϵ is the **machine precision** as returned by X02AJF.
- 3: BNI – REAL (KIND=nag_wp) *Input*
On entry: b_i , the nearest integer to b , satisfying $b_i = b - b_r$.
Constraints:

$$\text{BNI} = \lfloor \text{BNI} \rfloor;$$

$$|\text{BNI}| \leq \text{arbn}d;$$
 if $\text{BDR} = 0.0$, $\text{BNI} > 0$.
- 4: BDR – REAL (KIND=nag_wp) *Input*
On entry: b_r , the signed decimal remainder satisfying $b_r = b - b_i$ and $|b_r| \leq 0.5$.
Constraint: $|\text{BDR}| \leq 0.5$.
Note: if $|\text{BDR} - \text{ADR}| < 100.0\epsilon$, $a_r = b_r$ will be used, where ϵ is the **machine precision** as returned by X02AJF.
- 5: X – REAL (KIND=nag_wp) *Input*
On entry: the argument x of the function.
Constraint: $|X| \leq \text{arbn}d$.
- 6: FRM – REAL (KIND=nag_wp) *Output*
On exit: m_f , the scaled real component of the solution satisfying $m_f = M(a, b, x) \times 2^{-m_s}$.
Note: if overflow occurs upon completion, as indicated by IFAIL = 2, the value of m_f returned may still be correct. If overflow occurs in a subcalculation, as indicated by IFAIL = 5, this should not be assumed.
- 7: SCM – INTEGER *Output*
On exit: m_s , the scaling power of two, satisfying $m_s = \log_2 \left(\frac{M(a, b, x)}{m_f} \right)$.
Note: if overflow occurs upon completion, as indicated by IFAIL = 2, then $m_s \geq I_{\max}$, where I_{\max} is the largest representable integer (see X02BBF). If overflow occurs during a

subcalculation, as indicated by $\text{IFAIL} = 5$, m_s may or may not be greater than I_{\max} . In either case, $\text{SCM} = \text{X02BBF}$ will have been returned.

8: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

Underflow occurred during the evaluation of $M(a, b, x)$.
The returned value may be inaccurate.

$\text{IFAIL} = 2$

On completion, overflow occurred in the evaluation of $M(a, b, x)$.

$\text{IFAIL} = 3$

All approximations have completed, and the final residual estimate indicates some precision may have been lost.
Relative residual = $\langle \text{value} \rangle$.

$\text{IFAIL} = 4$

All approximations have completed, and the final residual estimate indicates no accuracy can be guaranteed.
Relative residual = $\langle \text{value} \rangle$.

$\text{IFAIL} = 5$

Overflow occurred in a subcalculation of $M(a, b, x)$.
The answer may be completely incorrect.

$\text{IFAIL} = 11$

On entry, $\text{ANI} = \langle \text{value} \rangle$.
Constraint: $|\text{ANI}| \leq \text{arwnd} = \langle \text{value} \rangle$.

$\text{IFAIL} = 13$

ANI is non-integral.
On entry, $\text{ANI} = \langle \text{value} \rangle$.
Constraint: $\text{ANI} = \lfloor \text{ANI} \rfloor$.

IFAIL = 21

On entry, $ADR = \langle value \rangle$.
 Constraint: $|ADR| \leq 0.5$.

IFAIL = 31

On entry, $BNI = \langle value \rangle$.
 Constraint: $|BNI| \leq arwnd = \langle value \rangle$.

IFAIL = 32

On entry, $b = BNI + BDR = \langle value \rangle$.
 $M(a, b, x)$ is undefined when b is zero or a negative integer.

IFAIL = 33

BNI is non-integral.
 On entry, $BNI = \langle value \rangle$.
 Constraint: $BNI = \lfloor BNI \rfloor$.

IFAIL = 41

On entry, $BDR = \langle value \rangle$.
 Constraint: $|BDR| \leq 0.5$.

IFAIL = 51

On entry, $X = \langle value \rangle$.
 Constraint: $|X| \leq arwnd = \langle value \rangle$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
 See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
 See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.
 See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

In general, if IFAIL = 0, the value of M may be assumed accurate, with the possible loss of one or two decimal places. Assuming the result does not under or overflow, an error estimate res is made internally using equation (1). If the magnitude of res is sufficiently large a nonzero IFAIL will be returned. Specifically,

IFAIL = 0 $res \leq 1000\epsilon$
 IFAIL = 3 $1000\epsilon < res \leq 0.1$
 IFAIL = 4 $res > 0.1$

A further estimate of the residual can be constructed using equation (1), and the differential identity,

$$\begin{aligned}\frac{dM(a, b, x)}{dx} &= \frac{a}{b}M(a+1, b+1, x), \\ \frac{d^2M(a, b, x)}{dx^2} &= \frac{a(a+1)}{b(b+1)}M(a+2, b+2, x).\end{aligned}$$

This estimate is however dependent upon the error involved in approximating $M(a+1, b+1, x)$ and $M(a+2, b+2, x)$.

8 Parallelism and Performance

S22BBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

S22BBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The values of m_f and m_s are implementation dependent. In most cases, if ${}_1F_1(a; b; x) = 0$, $m_f = 0$ and $m_s = 0$ will be returned, and if ${}_1F_1(a; b; x) = 0$ is finite, the fractional component will be bound by $0.5 \leq |m_f| < 1$, with m_s chosen accordingly.

The values returned in FRM (m_f) and SCM (m_s) may be used to explicitly evaluate $M(a, b, x)$, and may also be used to evaluate products and ratios of multiple values of M as follows,

$$\begin{aligned}M(a, b, x) &= m_f \times 2^{m_s} \\ M(a_1, b_1, x_1) \times M(a_2, b_2, x_2) &= (m_{f1} \times m_{f2}) \times 2^{(m_{s1} + m_{s2})} \\ \frac{M(a_1, b_1, x_1)}{M(a_2, b_2, x_2)} &= \frac{m_{f1}}{m_{f2}} \times 2^{(m_{s1} - m_{s2})} \\ \ln|M(a, b, x)| &= \ln|m_f| + m_s \times \ln(2)\end{aligned}$$

10 Example

This example evaluates the confluent hypergeometric function at two points in scaled form using S22BBF, and subsequently calculates their product and ratio without having to explicitly construct M .

10.1 Program Text

```
Program s22bbfe

!      S22BBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, s22bbf, x02bhf, x02blf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: ai, ar, bi, br, delta, frm, scale, x
      Integer                     :: ifail, k, scm
```

```

!      .. Local Arrays ..
      Real (Kind=nag_wp)                :: frmv(2)
      Integer                                :: scmv(2)
!      .. Intrinsic Procedures ..
      Intrinsic                          :: real
!      .. Executable Statements ..
      Write (nout,*) 'S22BBF Example Program Results'

      ai = -10.0_nag_wp
      bi = 30.0_nag_wp
      delta = 1.0E-4_nag_wp
      ar = delta
      br = -delta
      x = 25.0_nag_wp

      Write (nout,99999) 'a', 'b', 'x', 'frm', 'scm', 'M(a,b,x)'

      Do k = 1, 2
        If (k==2) Then
          ar = -ar
          br = -br
        End If

        ifail = -1
        Call s22bbf(ai,ar,bi,br,x,frm,scm,ifail)
        If (ifail==2 .Or. ifail>3) Then
!          Either the result has overflowed, no accuracy may be assumed,
!          or an input error has been detected.
          Write (nout,99996) ai + ar, bi + br, x, 'FAILED'
          Go To 100
        Else If (scm<x02blf()) Then
          scale = frm*real(x02bhf(),kind=nag_wp)**scm
          Write (nout,99998) ai + ar, bi + br, x, frm, scm, scale
        Else
          Write (nout,99997) ai + ar, bi + br, x, frm, scm,
            'Not representable' &
          End If
          frmv(k) = frm
          scmv(k) = scm
        End Do

!      Calculate the product M1*M2
      frm = frmv(1)*frmv(2)
      scm = scmv(1) + scmv(2)
      Write (nout,*)
      If (scm<x02blf()) Then
        scale = frm*real(x02bhf(),kind=nag_wp)**scm
        Write (nout,99995) 'Solution product', frm, scm, scale
      Else
        Write (nout,99994) 'Solution product', frm, scm, 'Not representable'
      End If

!      Calculate the ratio M1/M2
      If (frmv(2)/=0.0_nag_wp) Then
        frm = frmv(1)/frmv(2)
        scm = scmv(1) - scmv(2)
        Write (nout,*)
        If (scm<x02blf()) Then
          scale = frm*real(x02bhf(),kind=nag_wp)**scm
          Write (nout,99995) 'Solution ratio ', frm, scm, scale
        Else
          Write (nout,99994) 'Solution ratio ', frm, scm, 'Not representable'
        End If
      End If

100    Continue

99999 Format (/ ,1X,3(A10,1X),A12,1X,A6,1X,A12)
99998 Format (1X,3(F10.4,1X),E12.4,1X,I6,1X,E12.4)

```

```

99997 Format (1X,3(F10.4,1X),Es12.4,1X,I6,1X,A17)
99996 Format (1X,3(F10.4,1X),20X,A17)
99995 Format (1X,A16,17X,Es12.4,1X,I6,1X,Es12.4)
99994 Format (1X,A16,17X,Es12.4,1X,I6,1X,A17)
      End Program s22bbfe

```

10.2 Program Data

None.

10.3 Program Results

S22BBF Example Program Results

a	b	x	frm	scm	M(a,b,x)
-9.9999	29.9999	25.0000	-7.7329E-01	-15	-2.3599E-05
-10.0001	30.0001	25.0000	-7.7318E-01	-15	-2.3596E-05
Solution product			5.9789E-01	-30	5.5683E-10
Solution ratio			1.0001E+00	0	1.0001E+00
